
内容简介

Verilog-A/AMS 不是Verilog！数字设计工程师到此留步，模拟和混合信号电路设计工程师请往下看。

知道Verilog-A是在十几年前，作者还在学校读书的时候。沉醉于几个器件的方寸之间，追求着电路优化和创新，满心为了发表论文的作者，和很多人一样不觉得这个东西多么的有用。

实际上，限于经验和系统规模，那个阶段的确无法了解Verilog-A能够做什么。

十年之后，摆在面前的实际问题早已不是用几十个管子搭建电路，仿真出个结果这么简单了。模拟系统本身已经越长越大，越长越臃肿，我们还要把它和数字系统仿真一起，大老板们说这叫SOC。

这时候，SPICE模型已经没法帮你完成自顶向下的系统设计了。

Verilog-A/AMS可以帮助你描述模拟与混合信号电路。它可以替代SPICE书写复杂的模型，也可以使用抽象的数学算式描述精确的系统函数。SPICE模型需要运行3天的顶层，Verilog-A/AMS可以在30分钟内给你结果。

本文介绍Verilog-A/AMS的详细语法和在几种仿真器中的使用。如果你没有书写代码的习惯，没有关系，Verilog-A并不复杂，而且文中会有详细的例子。

除此之外，作者还探讨了使用Verilog-A/AMS进行系统建模的步骤，以及一些作者在使用Verilog-A中的经验和技巧。

使用Verilog-A/AMS，你才能感受什么是模拟和混合信号系统建模和验证。

如果你有兴趣，请往下看。





第1章 硬件描述语言与Verilog-A/AMS

硬件描述语言是用来描述硬件工作方式的程序语言。

硬件描述语言已经有了一段不小的历史，早在1995年，当大多数人还不知道电脑和程序是什么的时候，Cadence Design System公司就已经提交并通过了IEEE1364标准。这就是VHDL描述语言。VHDL语言和后来的Verilog-HDL语言主要用于描述数字电路的行为。它们描述的对象是离散、时间上不连续和逻辑的信号。

大约同一时间，一个名为Analog Verilog International的组织成立了。它希望通过扩展Verilog-HDL语言来支持模拟和混合信号仿真。在1996年，该组织公布了它的第一个成果就是Verilog-A。Verilog-A的工作方式和Verilog或者VHDL等描述数字电路行为的语言不相同。它是一个描述连续信号（也就是我们通常称为模拟系统）的语言。它兼容SPICE程序[1]的描述语言。

很快Cadence Design System 公司就在他们的spectre仿真器中支持了Verilog-A语言。

随着实际应用中模拟系统规模的不断扩大，支持Verilog-A的工具也越来越多，同时也为了支持混合信号的仿真, 这个组织在2000年发布了Verilog-AMS标准。Verilog-AMS混合了Verilog和Verilog-A的语法，并且增加了多个混合信号结构，使得Verilog-AMS成为了一个支持模拟、数字和混合信号描述和仿真的语言。

而后，Verilog-AMS还在继续进化中。2003年1月发布的Verilog-AMS的2.1版本标准加入了对直接使用Verilog的支持。同时他们也在努力将Verilog-AMS合并入System-Verilog。


注：SPICE, 全称Simulation Program with Integrated Circuit Emphasis, SPICE。是一种用于电路描述与仿真的语言与仿真器软件，用于检测电路的连接和功能的完整性，以及用于预测电路的行为。一般将使用SPICE仿真程序的代码均称为SPICE模型或SPICE代码。下文中提到SPICE都指此仿真程序。

Verilog，或称为Verilog-HDL，下文中提到这两者都是指这种用于描述数字电路行为的语言。



1.1 Verilog-AMS

Verilog-AMS是Verilog硬件描述语言的一个衍生物。类似Verilog-HDL和VHDL语言，Verilog-AMS被设计来创建系统模型，并进行仿真验证。 与Verilog和VHDL不同的是，我们还没有办法直接从Verilog-AMS代码“综合”出完整的模拟电路模块。也因为如此，Verilog-AMS并没有从一开始就进入模拟电路设计师的设计流程中。现在，随着模拟系统集成度的提高，仿真和验证的难度的加大，Verilog-AMS才越来越多地被人们采用。

Verilog-AMS包含了模拟和混合信号扩展模块。它扩展了Verilog、SystemVerilog等的事件驱动仿真器的回路，通过使用一个连续时间仿真器，可以在模拟域上求解微分方程。在Verilog-AMS里模拟事件可以触发数字行为，反之亦可。Verilog-AMS可以像SPICE模型一样描述非常细节的工作情况，但更多的时候它被用于描述高层的，更加抽象的功能，用于对大规模系统进行前期的，快速高效的仿真验证。

1.2 Verilog-A

Verilog-A是Verilog-AMS的连续时间子集。当我们使用Verilog-AMS描述模拟领域的功能，我们可以说Verilog-A和Verilog-AMS是一样的。

[image: 图像说明文字]


图 1.1 Verilog-AMS的组成 (摘自Cadence用户手册)

1.2.1 Verilog-AMS能做什么

使用模拟电路的描述语言，我们可以实现： 1.自顶向下的设计流程

传统的模拟电路设计方法，一切都要从电路开始。当单元电路没有完成的时候，顶层的仿真验证无从谈起，也无法实现自顶向下的设计方法。

2.项目研究阶段的可行性分析

项目开始阶段，对一些关键模块和参数的要求，决定了项目是否能够实现。使用模型进行验证，确定需求，能避免大量时间和人力的浪费。

3.系统级模型搭建和仿真

使用Verilog-A/AMS可以从高层较为简单地描述系统模块和行为。比起传统描述语言，能够更加简洁高效的进行大规模系统级别的建模和仿真。

4.快捷高效的验证

比起传统的SPICE模型，Verilog-A/AMS语言更加简单高效。使用Verilog-A/AMS可以大大提高系统级别的仿真验证速度。

5.提供IP模型

对于很多IP的厂商和开发者，提供一个简单的模型，让客户在真正购买之前能够试用你的IP，是一种非常好的事情。以往，需要搭建SPICE模型，其描述难度大大高于Verilog-AMS描述语言。同时对规模比较大的IP，也很难简化描述复杂程度。

6.提供系统级别模型

从IP到芯片，到板级系统，再到整个系统，每一个层次都需要验证。以往缺乏统一的描述方式，或者根本无法描述和验证模拟子系统。现在每一个层次都可以用Verilog-A/AMS描述，每个供应商都可以提供自己的模型。

1.2.2 Verilog-A/AMS的实际使用

Verilog-AMS与Verilog-A的差别很大程度上即是允许同一个模块既描述数字行为又描述模拟行为。我们要不要使用Verilog-AMS的数字描述功能？我们何时需要将模拟的描述内容和数字的描述内容放在一起？

数模混合仿真本身没有问题。对于大型设计，将数字和模拟分开建模和仿真已经是一个必然的流程。但是Verilog-AMS支持的在一个模块内部同时使用数字和模拟描述则会带来一些困扰。

以下，让我们看看现在面临的问题。

第一种情况，我们要实现一个复杂的模拟系统，同时里面包含了很少一些逻辑单元，总数不超过100个门。我们需要使用Verilog-A搭建模型来进行模拟部分的仿真。既然是一个只有不超过100个的数字门单元的系统，通常会让模拟电路工程师自己来做这些逻辑模块吧。既然这样，最好不用模拟电路工程师们不熟悉的Verilog数字语法，因为100个门的逻辑可以很容易的可以实现，而且不太会发生错误。

第二种情况，我们要实现一个复杂的混合信号系统。里面有复杂的模拟模块，也有复杂的数字模块，多于1000个的门电路，甚至更多。我们有一个或多个专职的数字工程师。通常情况下，数字工程师继续使用他熟悉的Verilog，模拟工程师使用Verilog-A，合并之后使用混合信号仿真工具进行验证。因为，数字工程师通常不熟悉Verilog-AMS的混合信号书写方法。而且很多数字的仿真器，综合布局布线工具，都不支持Verilog-AMS。因此使用Verilog-AMS基本上对数字设计无用。

第三种情况，我们有一个比较复杂的模块需要建模，它既带有数字信号，又带有模拟信号。即使在这种情况下，我们也应该慎重考虑是否使用混合信号模型。

建立模型的目的是加快仿真，但在后面的章节可以看到，数字和模拟混合描述并不是没有代价的，在简化了数字域仿真的情况下，也带来了额外的数模转换连接模块的仿真时间。因此，分开描述是否能够简化模型，加快仿真是一个问题。其次，数模混合仿真简化数字域仿真的同时，也丢失了数字域的模拟特性。数字域的延时、噪声、摆率等是否会对模块的性能有影响，是否能被忽略，也是需要考虑的一个问题。最后，数模混合的书写，仿真验证平台的搭建，都需要一个额外的学习时间，也都可能带来一些错误和修正的时间。

EDA厂商们一直强调使用混合信号仿真必定有其道理。相信随着EDA技术的进步，有一天混合信号描述甚至可以快捷地生成包含模拟部分的系统原型电路。目前而言，作者对于使用Verilog-AMS混合信号模块描述的准则是：

1.对于数字功能模块，不使用混合信号描述。数字模块直接使用Verilog或者VHDL是最佳选择。

2.不在高层次的模块描述中使用混合信号描述。在比较大而复杂的模块中使用混合信号描述，往往容易造成错误和精度缺失。

3.在一个模块内部进行混合信号描述时，需要认真考虑那些网线和端口可以被定义为数字信号，并且分析这样做带来的好处。

1.3 SPICE模型

现在模拟设计中广泛使用的是SPICE模型。SPICE模型其实就是SPICE仿真器原生可以识别的网表代码。由于SPICE仿真器在模拟设计中的重要地位，SPICE模型也同样是事实标准。绝大多数厂商都为自己的器件和模块提供SPICE模型以供用户仿真。

SPICE模型的最大优点是精确。它可以用一个带有多达400+个参数的模块来描述某个器件。SPICE模型详细描述各个工艺参数下的各种行为变化，精确的保证仿真结果和实际物理产品高度一致。我们通常所说的对电路进行仿真，一般都是指使用带有SPICE模型的器件搭建电路并进行仿真。

随着高精度而来的是SPICE模型书写复杂、仿真速度慢并且灵活性差。对于描述高层次的功能和行为，SPICE模型会比Verilog-A/AMS代码复杂得多。同时其灵活性差也使得一个SPICE模型的修改相当麻烦，非常不利于模型的复用。

所以，在搭建系统模型，或者提供用户一个简单的产品模型这些任务上，SPICE模型往往不是一个很好的选择。这也是为什么在这些地方，我们需要引入Veirlog-A/AMS语言。

1.4 WREAL

WREAL 是Verilog-AMS支持的一种新的数值模型。WREAL的特殊之处在于它使用有限的浮点数值的点来模拟一条电路工作曲线。而SPICE和Verilog-A的计算结果是一条理论上可以无限精度，包含无限点的的曲线。从某种程度上，WREAL的实现方式类似于Fast-Spice的查表点工作模式，其目的是进一步简化仿真，从而支持更大规模的模拟系统仿真。

使用WREAL的最大好处是速度快。使用WREAL变量的模型在计算的时候无需使用SPICE迭代运算。它使用的是比较简单的，直接推导的函数来模仿模拟电路真正的工作情况。

相比SPICE和Verilog-A的模型必须使用SPICE仿真器的迭代运算，WREAL仿真器使用离散事件触发，就像数字仿真器那样。这给WREAL的计算上带来极大的速度优势。但同时，离散的计算模式使得WREA模型在含有反馈的电路中无法给出准确的结果。在普通电路中它也需要牺牲输出的精度。所以，WREAL并不适用于需要精确度量的模拟电路的模型中。

下图是一张多个描述语言的仿真速度的对比。

[image: 图像说明文字]


图 1.2 描述语言的精度与仿真速度（上图摘自Cadence用户手册）


汤姆·贝克利 -- Cadence设计系统公司高级副总裁，在CDNLive EMEA2012的主题演讲中说：如果你能够将所有的SPICE仿真计算（包括Verilog-AMS）替代为事件驱动和REAL数值仿真，仿真速度能够提高大约300倍。因此，WREAL拥有可观的速度优势。



WREAL的一个问题是，它需要对真正的模拟电路的行为有一个非常好的预测。因为，WREAL的所有计算都是前向的，我们想要用这些前向计算来模拟SPICE仿真器迭代结算的结果，就需要对实际电路工作情况有个很好的了解。然而，模型通常在设计阶段的前期实现，而此时通常不会有很好的对真实电路的预测。如果WREAL模型无法很好的体现SPICE仿真器对真实电路的仿真结果，那模型的意义就不大了。亦或需要在设计后期随着真实电路的开发修改模型，这样就会牺牲一些研发的时间。

有一些相关的研究正在进行中，比如利用简单的电路生成WREAL模型，这样可以作为未来新项目系统的起始WREAL模型。也有一些研究包括迭代WREAL和真实电路仿真，这样可以自动修改WREAL模型等等。但不管怎么，对于WREAL在大规模需要精度的验证中的使用，还是需要一定考虑。

本文对WREAL不做详细介绍。

1.5 仿真工具

目前主流的商用仿真器都已经全面支持Verilog-A/AMS了。

Cadence Design System -- 作为世界上第一个支持Verilog-A标准的EDA厂商以及本身就是Verilog-AMS等标准组织重要成员，Cadence对Verilog-A/AMS语言的支持是全方位的。 参见： http://www.cadence.com/products/cic/Pages/default.aspx

Mentor Graphic -- ELDO 和新的ADMS 都能够支持Verilog-A仿真。 参见： http://www.mentor.com/products/ic_nanometer_design/analog-mixed-signal-verification/

Synopsis -- 新思还在推进新的Hspice，当然它也支持多种模拟与混合信号描述语言。 参见： http://www.synopsys.com/ Hspice默认配置并不支持Verilog-A/AMS，你需要购买Hspice_va这个License。

对于无法使用商用仿真的人来说，现在有很多开源的免费仿真器。它们同样支持SPICE和Verilog-A仿真。

Symica -- 一个商业的仿真平台。但是它的免费版本可以在一定限制之下进行完整的仿真和后端设计。 参见： http://www.symica.com/

Ngspice -- 一个开源软件，本身只是一个命令行软件，但支持一些外挂的图形界面以方便电路设计。 参见： http://ngspice.sourceforge.net/

Qucs- 另一个开源的免费的仿真器，它包含有各种外挂。包括支持图形化的电路和代码输入，也支持Verilog和VHDL的仿真。 参见： http://qucs.sourceforge.net/index.html

Ngspice和Qucs支持Verilog-A/AMS都使用了一个ADMS的软件包。ADMS可以将Verilog-A/AMS代码转换成为C代码而给类SPICE仿真器使用。有兴趣研究ADMS可以到这里下载：http://sourceforge.net/projects/mot-adms/

免费开源的软件，很适合业余学习和开发。Qucs正在成长为一个可以进行模型设计，电路实现，系统仿真，PCB设计等一整套流程的完整开发平台。而Ngspice灵活的接口，被使用在很多在线仿真和计算上面。本文会在附录里对开源的和免费的EDA软件如何支持Verilog-A/AMS的使用方式做一些介绍。

本文要求读者有一些模拟和数字设计基础，能够进行电路绘制和进行简单仿真。对下文的一些基础语句如有疑问，可参考相应的基础书籍。 本文主要示例使用Cadence环境。不同的仿真器环境差异较大，代码不能保证直接在其他仿真器上100%兼容运行。





第2章 构建简单模块

这一章介绍如何搭建简单的模拟单元模块。我们将简述Verilog-A所使用的数据类型，语句语法，以及如何进行仿真。同时我们也会给出一些简单的常用器件和模块的模型。通过这一章，我们将能够使用Verilog-A描述简单的系统。

2.1 Verilog-A代码示例

一个简单的Verilog-A的文件，如下所示：


// Verilog-A for LIB_NAME, CELL_NAME
`include "constants.vams"
`include "disciplines.vams"
module Module_Name(P,N,GND);
input P;
electrical P;
output N;
electrical N;
inout GND;
electrical GND;

parameter real R=100;
parameter integer Debug=0;
real i;
integer j;

analog begin
V(P ,N) <+I(P, N)*R;
end
endmodule





这个例子包含了所有Verilog-A代码的必备元素，我们下面逐句分析Verilog-A最基础的代码格式。

2.1.1 注释

第一句的开始有这样一个符号：


//




和其他很多程序语言一样，双斜杠在Verilog-A/AMS中表示注释。也就意味着这一行是不会被执行的注释代码。 你也可以使用：


/* */




来进行多行注释，但多行注释有时候会在特殊字符上面引起奇怪的问题，一般来说双斜杠用的是最多的。

2.1.2 包含外部文件

接下来，示例代码使用了


`include




（注意符号`是键盘上TAB键上面的那个键，而不是单引号）来引用两个外部文件。


`include "constants.vams"
`include "disciplines.vams"





在Cadence环境下使用Verilog-A，默认会导入上面这两个文件。 在其他的一些平台上，也可能使用的是这样两个文件： include "constants.h"
 include "disciplines.h" 文件后缀不同，但是基本内容是一样的。

第一个文件是constants.vams。这个文件内部定义了很多常用的数值。比如： `define M_PI 3.14159265358979323846 有了这样的定义，我们可以在自己的代码里任何地方直接使用M_PI作为π。

第二个是disciplines.vams。单词discipline本身是约束的意思。看这个文件内部，包含有一大堆的nature、potential、flow等等的定义。其实我们只需要了解这样几个属性就可以：


discipline electrical
potential Voltage ;
flow Current ;
enddiscipline




我们后面每次定义一个模拟节点都会用到electrical这个限定词。这里的定义表示：任何一个electrical的端口，都包含两个属性，一个是电压，一个是电流。更具体一点地讲，电压被认为是一个表示“势”（potential）的属性，而电流是一个表示“流”（flow）的属性。 你可以自己定义自己的nature，也可以自己定义自己的discipline。但是，作为一般的电路设计应用，其实不知道这些基本上不会影响我们使用Verilog-A。我们只需要记住electrical、V和I是电路的属性即可。

2.1.3 模块声明


module Module_Name(port, port);




Verilog-A中使用module来声明一个模块。可以认为每一个Verilog-A module对应一块实际电路。

在module的声明列表里，你需要加上这个模块的输入和输入端口。一个真实的模块声明可能是这样的：


module Module_Name(VDD, inp, inn, out, GND);




值得注意的是在Verilog-A中V和G是保留关键字，所以不能使用V或者G作为端口名称。它们很多时候被其他文件默认的用作电源和地电压的端口名，这样会造成一些问题。当一个模块既有Verilog-A代码，又有真实电路的时候，一些器件的内部会因为默认使用全局的V和G表示电源和地电压而产生问题。这种情况为了使用Verilog-A/AMS就只能修改那些器件的默认连接。


input ; 
electrical ; 
output ; 
inout ;




我们在module声明中列出来的所有端口，都必须在随后的端口声明中明确表示端口的属性。和我们在电路中使用的一样，端口有三种类型：输入，输出和双向，分别对应input，output和inout。模拟的节点和端口都需要被声明为electrical。

electrical限定词就是我们前面提到的电路中使用的discipline。它用于表示这个端口（或节点）是一个模拟端口（或节点）。

如果我们需要声明一个总线端口，我们可以这样写。 input [1:12] bus; electrical [1:12] bus; 请注意端口数量和端口名称的先后顺序。在代码里，我们可以用bus[1]这样的方式来访问总线里的任意一个。

我们的这第一段代码是一个很糟糕的端口声明的例子。为了覆盖三种端口类型，我们将一个类似电阻的东西的端口定义的乱七八糟。在实际应用中，应该清楚地了解每个端口大致的信号流方向，以便于确定其端口类型。

2.1.4 数值变量

数值变量通常被用来保存中间计算结果。Verilog-A支持几种数值变量。 integer 整数类型，其数值只能是整数，并且数值范围在-231 ~ 231之间。


real




浮点数类型，数值范围在10-37 ~ 1037之间。 由于计算精度，浮点数会更加常用于保存中间结果，整数则可以用来做一些控制逻辑的计算。 这两种数值变量也都可以声明为数组变量。


integer i[1:10];
real j[1:10];




注意和总线声明不同，这里的空格在前而矢量声明在后。


parameter integer/real parameter_name;




参数是用户可以在模块实例化的时候修改的数值。比如一个比较器的Verilog-A模型，你可以设置一个参数作为这个比较器的延时。这样，在你的整个系统中使用多个这种比较器的时候，可以根据系统的需求为每个比较器设置不同的延时，从而获得每个具体电路的参数需求。 参数分为整数型和浮点数型。参数带有一个默认值。如果你直接使用而不做外部设置，这个默认值会起作用。关于参数更加详细的描述可以参见第3章。

2.1.5 模拟过程


analog begin
end





所有的模拟模块的功能都要在这两个词之间描述，比如这里描述了一个电阻：


analog begin
    V(P ,N) <+I(P, N)*R;
end




顾名思义，V表示了电压，而I表示了电流。V可以指定两个端点，表示两个点之间的电压差；也可以只指定一个端点，那就表示那一点对0电位的电压，或称绝对电位。I同样可以这样，使用两个端点表示两端点之间的电流，使用一个端点表示流入这点的全部电流。I的方向是流入为正，流出为负。

上面的语句也可以这样写：


analog begin
    I(P ,N) <+V(P, N)/R;
end



大多数情况下，这两种写法是一样的。但是，R是一个参数，如果你在实际使用的时候忘记了具体的写法（当然会忘记）而将R设置为0，对很多仿真器来说是会有问题的（一些‘聪明’的仿真器也许会直接忽视你的这种单元）。所以我们的结论是：


大多数情况下，使用乘法比使用除法好。



2.1.6 赋值

在Verilog-A语句中，常常能见到如下语句：


V(P ,N) <+I(P, N)*R;




也能见到如下语句：


R=(R1+R2)/2;




我们大致也能猜到这两个符号“<+”和“=”都是赋值操作。那么不同的写法之间的差别是什么呢？

等号（=）在Verilog-A中用作为数值赋值，也就是说等号左侧的变量要么是integer要么是real。而（<+）符号用于对electrical量的赋值，也就是类似V和I的量。

其本质差别是，等号式子的计算是不需要SPICE进行迭代计算的。在每一步仿真中，等号的式子是直接出结果的。而<+则并不是单纯的赋值，它代表的是一个需要SPICE仿真器进行迭代计算的方程。所以，等号基本上可以用在所有位置，而<+必须被包含在analog begin 语句中。

另一重要的差别是赋值的叠加性，如下面的例子：


R=R1;
R=R2;




计算的结果是R=R2; 而对于:


V(P ,N) <+V1;
V(P ,N) <+V2;




计算的结果是V(P, N)=V1+V2; 这也是因为，所有带<+的等式都是SPICE仿真器要处理的一个方程式，不存在计算的先后。所以对于模拟量的赋值，都采用叠加方式计算。

2.1.7 仿真示例

了解了上面这些，我们就可以开始尝试仿真了。如果在Cadence环境下，用Cadence自带的编辑器编辑Verilog-A/AMS代码。你可以自己设置喜欢的文本编辑器作为Verilog-A/AMS代码文件的编辑器。通常默认的编辑器是vi。你也可以将它设置为gedit或者kate这样的图形化的文本编辑器，这样书写起来更加直观一些。

保存退出后（必须完全关闭编辑器而不是关闭你在编辑的文件），cadence就会识别你书写的代码，询问是否需要帮你生成一个symbol。 你可以把这 个symbol加入到你的仿真电路里。

如图2-1所示，代码书写完毕后，关闭编辑器，系统询问是否生成图标：

[image: 图像说明文字]


图 2-1 系统询问生成图标

如图2-2所示，生成图标时需要定义端口： [image: 图像说明文字]


图 2-2 生成端口列表

如图2-3所示：将生成的图标加入到测试电路中： [image: 图像说明文字]


图 2-3 系统生成图标

上图中这个symbol很难看，我们稍加修改，放入测试电路中直接运行仿真就可以看见各点的电压电流曲线，就像任意一个普通的电路那样。如图2-4所示： 如何进行最基础的仿真请读者参阅Cadence或其他仿真器的手册。 [image: 图像说明文字]


图 2-4 仿真结果

2.2 简单模块描述

前面我们了解了一个基本的Verilog-A模块的书写以及如何在仿真软件中进行仿真。接下来，我们学习更多的语句，并开始书写自己的模块。

2.2.1 输入输出的关系

所有的Verilog-A模块本质上都是代表了一个模拟电路。对于一个电路模块，如果我们把它看作一个黑匣子，我们关注的只是输入和输出端。也就是说，最基本的代码只需要描述输入和输出的关系即可。对于逻辑电路，输入输出只考虑绝对电平高低，而模拟电路的输出输出则包含：


	电压

	电流

	阻抗



所有的功能的描述实际上都归结于定义上面三个属性。大多数情况下，我们通过定义其中的两者来规范第三个属性。比如我们写如下语句： V(P ,N) <+I(P, N)*R; 实际上我们就是用阻抗R和电流I确定了第三个属性V的大小。

对于大多数无源器件，书写的方式是直观的，只需要根据器件属性定义即可。而对于有源单元和比较大的模块，我们一般从输入输出的关系来定义。 简单的输入输出的关系大致分为以下几种：


电压控制电压输出
V(OUT) <+V(IP, IN)*GAIN;
电压控制电流输出
I(OUT) <+V(IP, IN)*GM;
电流控制电压输出
V(OUT) <+I(IP, IN)*R;
电流控制电流输出
I(OUT) <+I(IP, IN)*RATIO;





电压输入最常见，比如一个放大器，比较器。

电压输出包括普通的电压信号输出--放大器；也包括逻辑电平输出--AD转换器。

电流输入比如说一个电流镜单元，电流检测单元。

电流输出包括一些电压电流转换器，gm单元等。

我们可以根据上面四种情况将我们要描述的输入输出关系表示为这四类变量：gain,，gm，r和ratio。这些关系可以只是个简单的数值，比如对于一个电压放大器；也可能是频率上的某种关系，比如对于一个滤波器；也可能是更加复杂的转换，比如对于一个AD转换器。

选择哪种类型取决于你对系统结构的设计，取决于在系统中这个模块的输入输出的要求。

2.2.2 输入输出阻抗

除了定义输入输出的关系，模拟系统也需要关注每个节点（端口）的阻抗。阻抗定义了这个节点（端口）上自身的电压电流关系。

为了保持Verilog-A模块的模拟特性，我们一般不会定义阻抗无穷大的输入端，也不会定义供电流能力无穷的输出。通常我们比较习惯于定义输出和输出阻抗，来确定V或者I的值。

常见的阻抗包含阻性，容性和感性。前面就是一个电阻性阻抗的的描述。


V(P ,N) <+ddt(I(P, N))*L;




这是一个感性阻抗的描述。在Verilog-A里我们用ddt()来表示微分。上面的式子可以简单的理解为：电流变化的速度乘以L等于电压的大小。这也就是一个标准的理想电感的描述。


I(P ,N) <+ddt(V(P, N))*C;




如果这样写，就变成了容性阻抗的描述。

顺带介绍一下，Verilog-A中也支持积分操作。


idt( input[ , ic[ , assert[ , nature] ] ] )




input指需要做积分的计算项。积分上限总是当前仿真时间。ic表示初始值。assert默认为1。在DC仿真中，如果assert=1，计算返回ic值；如果assert=0，初始值由DC计算产生。在transient仿真中，当assert变为0的时候，我们称t0，则积分时间为从当前时间至t0。

2.2.3 内部节点

现在我们有了输入输出端口，有了输入输出的阻抗，但有时候他们的关系并不那么直接，我们还需要定义内部节点。内部节点一般用于保存中间结果。


electrical net1;




定义内部节点的方式很简单，只需要定义一个节点名称即可使用。这个名称在这个模块里必须是唯一的。 在实际应用中，由于内部节点只存在于计算过程中，我们可以尽可能的用数值变量存储内部节点的值，来替代一个真正的内部节点。这样的好处是更少的节点意味着更少的SPICE仿真器计算量。

2.2.4 并联、串联与Branch

并联和串联是电路里常见的连接方式。在电路里，串联和并联的关系可以写成如下形式：


	串联 = 流过两者（或多者）的电流大小相同，总电压等于各自电压的和

	并联 = 电压相同，总电流等于各自电流之和



最简单的串联方式的定义就是使用内部节点。比如，如果模块内部的节点有串联关系，我们可以这样写：


I(ip1, net1)<+I(net1, ip2);




这样就表示ip1和ip2通过ne1连接。

并联的关系可以这样写：


I(ip1, in1) <+V(ip1, in1)/R1;
I(ip1, in1) <+V(ip1, in1)/R2;
I(ip1, in1) <+V(ip1, in1)/R3;




由于前面提到的叠加赋值，这就表示流过两点的电流来源于三个部分。电流的叠加，也就意味三个直路的并联特性。

在VerilogA中我们也可以为并联分支定义branch。


branch (ip1, in1) resistor1;
branch (ip1, in1) resistor2;
branch (ip1, in1) resistor3;




这样，我们也可以把前面的式子写成：


I(resistor1) <+ V(ip1, in1)/R1;
I(resistor2) <+ V(ip1, in1)/R2;
I(resistor3) <+ V(ip1, in1)/R3;




branch主要的作用就是使代码更加易读。

2.2.5 示例

结合上面基本上所有的语法，我们来描述如下图这样一个负载。

[image: 图像说明文字]


图 2-5 RLC电路

代码如下：


`include "constants.vams"
`include "disciplines.vams"

module series_rlc1 (p, n);
parameter real r=100;            // resistance (Ohms)
parameter real l=1n;            // inductance (H)
parameter real c=1p exclude 0;    // capacitance (F)
inout p, n;
electrical p, n;

analog begin
        V(p,n) <+ r*I(p,n);
        V(p,n) <+ l*ddt(I(p,n));
        V(p,n) <+ idt(I(p,n))/c;
end
endmodule 




我们把两个端口都定义为inout，表示这个模型并没有特定的入端和出端。接着我们定义了三个参数，分别用于表示电阻，电感和电容的大小。 在模拟行为描述中，可以看到电阻的电压电流关系直接是电阻值，电感的描述使用了微分操作，而电容的描述使用了积分操作。 注意叠加赋值的描述方法。 我们也可以这样写：


module series_rlc2 (p, n);
parameter real r=100;            // resistance (Ohms)
parameter real l=1n;            // inductance (H)
parameter real c=1p exclude 0;    // capacitance (F)
inout p, n;
electrical p, n, i;
branch (p, n) rl, (p, n) cap;

analog begin
        V(rl) <+ r*I(rl);
        V(rl) <+ l*ddt(I(rl));
        I(cap) <+ c*ddt(V(cap));
end
endmodule 





用V和I描述多个branch是等效的。

2.3 条件与循环语句

使用前面的语句，我们已经可以描述一般的功能模块了。但是要实现复杂的功能，还需要更多的语句，特别是流程执行控制语句。最常见的包括条件语句和循环语句。

2.3.1 条件语句

条件语句用于控制代码的执行流程。通过判断条件是否成立，条件语句可以使代码在不同的情况下执行不同的部分。 主要的条件语句包括if语句和case语句。

2.3.1.1 if语句

最常见的条件判断语句是if语句。其基本格式如下：


if (X)
    A;
else if (Y)
    B;
else  
    C;




if语句几乎出现在所有的计算机语言里，用法也基本都相同。当判断语句为真的时候，执行相应的语句。在Verilog-A中，用begin-end对标志执行体。 如果执行单元含有多个语句，我们这样写。


if (X)  begin
A;
B;
  end  
else  begin
C;
  end




需要注意的是，仿真器为每个else匹配举例它最近的那个if。


If (X)
       If (Y)
    A;  
else
B;




在这里，else跟随的是if(Y)的条件。 所以，如果使用了else，为了避免出错，我们最好为附近的每个if都匹配一个else。如果匹配的else不需要执行任何实际操作，我们可以使用一个带‘;’ 的空行。

2.3.1.2 case语句

另一种常见的条件语句是case语句。


case(value)
(1) : A;
(2) : B;
(3) : C;
default : D;
endcase




这样，当value等于1，2，3的时候分别执行语句A, B, C。 很多时候，case也可以当作多个if来执行，比如这样写。


case(1)
(X) : A;
(Y) : B;
(Z) : C;
default : D;
endcase




当X, Y, Z语句为真的时候，会分别执行A, B, C。

2.3.2 循环语句

循环语句同样用于控制代码的执行流程。在不少实际问题中有许多具有规律性的重复操作，因此在程序中就需要重复执行某些语句。当符合某个条件的情况下，某个步骤需要被执行多次，我们就可以使用循环语句。 主要的循环语句包括：


	for

	while

	repeat



2.3.2.1for语句

for 语句格式非常类似C语言中的使用方法。 for (i = 1; i < 10; i = i + 1 ) total = total + i; Verilog-A不支持类似i++的写法。这是一个作者经常犯的错误。作者也好奇，为什么Verilog-A/AMS不提供这么简洁而且优美的语句呢。

2.3.2.2while语句

同样，while语句也非常类似C语言的用法。具体语法分析请参见第三章。


i = 1;
while (i<10) begin
i= i+ 1 ;
total= total + i ;
end




2.3.2.3repeat语句

repeat 语句是更加简化的for语句。对于熟悉常见编程语言的人来说，习惯使用for或者while，也许不需要使用repeat。


i = 1 ;
total = 0 ;
repeat (9) begin
i = i + 1 ;
total = total + i ;
end




请注意，为了让三个例子等效，这里repeat只执行9次。 由于repeat的变量（这里的9）必须是一个常数，所以它不能被用在模拟变量的操作中，这也限制了repeat语句的适用范围。

2.3.3 genvar

前面的几种循环语句的例子都是真对某个数值的多次操作。在Verilog-A模型中，我们常常会需要用到总线操作。比如一个16bit的ADC或DAC，又或者一个32路的检测信号。我们不会想重复写16遍赋值或者判断语句。我们第一个念头肯定是用循环语句。我们可能会这样写：


integer i;
for (i = 1; i < 10; i = i + 1 ) 
V(out[i]) <+ V(in[i]);




哦，对不起，上面这段代码是是错的。因为i是一个整数型变量。整数型变量无法在仿真器内展开指定每一个electrical端口。 在Verilog-A中，我们有一种专门的变量，叫做genvar。它最大用处就是动态的指定总线中的端口。


genvar i;
for (i = 1; i < 10; i = i + 1 ) 
V(out[i]) <+ V(in[i]);




genvar变量的使用是受限制的。首先，它只能在analog行为模块内使用。其次，对它赋值和修改操作的变量只能是常数或者另一个genvar。


integer i;
genvar j;
j = i+1;




上面这样的代码也是错误的。我们不能用一个整数型变量去修改一个genvar变量。

2.4 模拟事件

在Verilog-A中，很多时候我们不仅仅要检测电流和电压，我们还希望检测一些电压和电流的特殊行为，比如电压的跳变等等。这些信号的特定行为，被定义为模拟信号的某种事件。在Verilog-A中我们有一些语句是专门用来检测这些模拟事件的。

2.4.1 initial_step 和 final_step


@(initial_step) begin
end




initial_step 事件发生在仿真的开始阶段，就类似与仿真中的initial condition设置。Initial_step里的代码会是整个模块第一个被SPICE仿真器计算的代码。通常我们会在initial_step执行一些初始化的工作，或者输出一些设置信息。


@(final_step) begin
end




final_step只在仿真结束的时候被执行。如果手动结束仿真，比如一个100uS的tran仿真在90uS被手动结束，将不会触发final_step。

2.4.2 cross


@(cross(expr1[ ,direction [ ,time_tol [ ,expr_tol ] ] ] )) begin
end





cross事件表示当信号的值穿过0值（也就是从正值变为负值，或者反过来）的时候，其中的语句将会被执行。

参数expr1 表示检测的信号。参数direction 表示穿越方向，1表示正向（从负值变为正值），0表示双向，-1表示负向。direction 默认为0。参数time_tol和expr_tol分别表示变化的时间和数值下限。比如将time_tol设置为1u，则所有持续时间小于1uS的毛刺将被忽略。即使这些毛刺的尖峰越0，也不会触发cross事件。

可以想象，cross大量使用在transient仿真中，比如数字逻辑信号的变化，比较器的输入等等。

由于cross检测的是穿越0值这个动态的事件，在DC等静态仿真中它是不会被触发的。

2.4.3 above


@(above (expr1[ ,time_tol [ ,expr_tol ] ] )) begin
end




above事件表示当信号值大于0的时候，其中的语句将会被执行。

参数expr1 表示检测的信号。参数time_tol和expr_tol分别表示变化的时间和数值下限。

和cross事件不同的在于，above事件是一个静态的检测。它在DC，initial_step等静态点仿真中也会被触发。

2.4.4 timer


@(timer ( start_time[ , period[ , timetol]] )) begin
end




timer被用于周期性的事件，比如生成时钟信号。通常真正生成时钟的方式都需要大量的仿真时间，而直接使用timer事件可以大大减少仿真器的计算。或者timer语句也可以被放置与每一个响应时钟的单元中，从而简化掉整个时钟信号线。

参数start_time表示开始时间。参数period表示周期时间。参数timetol表示在仿真中，最小允许的时间误差（也就是说，如果定义事件发生的时间是每10uS，timetol为1nS，则要求仿真器在每个10uS+/-1nS的时间点内必须触发事件）。

2.4.5 or


@(cross(V(clk)-thresh, 1) or initial_step) begin
end





这里的or不是一个计算符，它表示当两种事件发生时都执行内部的语句。它的作用是当同样的计算被不同事件触发的时候简化代码。

2.4.6 示例

下面是一个ADC的例子，它综合使用了我们现在用到的各种语法。


`include "constants.vams"
`include "disciplines.vams"

module adc(out, in, clk);
parameter real fullscale = 1.0;    // input range is from 0 to fullscale (V)
parameter real td = 0;        // delay from clock edge to output (s)
parameter real tt = 0;        // transition time of output (s)
parameter real vdd = 5.0;        // voltage level of logic 1 (V)
parameter real thresh = vdd/2;    // logic threshold level (V)

input in, clk;
output [0:7] out;
voltage in, clk;
voltage [0:7] out;
real sample, midpoint;
integer result[0:7];
integer i;
genvar j;

analog begin
@(cross(V(clk)-thresh, 1) or initial_step) begin
sample = V(in);
midpoint = fullscale/2.0;
for (i = 7; i >= 0; i = i - 1) begin
    if (sample > midpoint) begin
    result[i] = 1;
    sample = sample - midpoint;
    end 
else begin
    result[i] = 0;
    end
    sample = 2.0*sample;
end
end
for(j=0;j<8;j=j+1)
V(out[j]) <+ transition(result[j] ? vdd : 0.0, td, tt);
end
end
endmodule





首先，当初始化和时钟跳变的时候，ADC采样输入信号。在一个循环结构里，计算后的sample电压和基准电压进行比较（这里是fullscle的一半），输出至内部数组result。同时，在这个genvar的循环里，随着判断位数的升高，每次都将sample电压翻倍，表示这是一个2进位编码的AD转换器。 接下来，在另一个genvar循环里，用result数组对每一个输出赋值。transition语句是用于产生平滑的瞬态响应，我们会在后面进行介绍。 这种描述结构的思路是，先根据输入计算出输出的数值量值。再将这些数值量值直接地或者经过处理的赋给输出端口。这样书写结构的好处是，计算输出的数值量的时候可以最大地避免模拟量互相之间的影响，减少内部的循环和带给仿真器的迭代计算。

2.5 复用模块

当我们有了一个Verilog-A/AMS模块，我们可以向把它像一般电路模块一样，放置在电路里，互相连接成为更大的电路模块。这是一种最常见的、显式的使用现有模块的方式。

我们也可以在别的Verilog-A/AMS模块里，直接用语句调用已有的模块。这样，我们隐含的使用了这个模块，对上层设计者来说，这个模块是不可见 的，也可以被称为隐式的调用模块。


module submoduleA(in, out) ;
input in ;
output out ;
electrical in, out ;
analog begin
V(out) <+ V(in) ;
end
endmodule

module highermodue(qin, qout) ;
input qin ;
output qout ;
electrical qin, qout ;
electrical aa1 ;
submoduleA vd1 (qin, aa1) ;
submoduleA vd2 (aa1, qout) ;
endmodule




这个例子里，模块highermodule内部使用了两个上面定义的子模块submoduleA。而子模块的详细实现，对父模块的使用者来说是不可见的。 模块实例化的具体语法请参见第3章。 通常的做法是将这两个module要放在两个不同的文件里。如果是在cadence环境下，只要这两个module在同一个库里，直接可以调用。其他仿真器请查阅对应文档。

2.6 库和高级工具的使用

如果你使用的的是商业的混合信号开发平台，通常EDA厂商会提供不少预设的Verilog-A和Verilog-AMS库。这里以Cadence平台为例，讲解一下一些库和高级复用工具的使用方法。

2.6.1 ADHL库

Cadence环境下提供了多种现成的库，比如我们常用的analogLib库，bms库。ahdl库是一个包含了很多常用Verilog-A模型的库。每个ahdl库里的单元和一般的器件一样，你只需要把它的图标加入到电路中，设定好你需要的参数，就直接可以进行仿真了。 下面以从ahdl库中生成一个比较器模型为例。

一般使用现成的ahdl库时，可以象使用其他库一样，直接在你自己的电路中插入单元的图标。如下图2-6中，我们选中一个ahdl库中的比较器，添加到电路中。

图 2-6 单元的属性列表

可以看见，这个ahdl库中的Verilog-A比较器模型支持修改的参数包括：输出高电平，输出低电平，输入失调和斜率。我们将输出高和低分别设置为5V和0V，然后将输入失调设置为1mV。 在图2-7中，我们为测试电路加上输入电压源和地电平，可以直接进行仿真。

图 2-7 仿真电路

图2-8显示了仿真的结果。可以看见，输出翻转点位于输入相差1mV处。

图 2-8 仿真结果

图 2-9 仿真结果

在图2-9中，我们修改斜率参数为500，重新进行仿真，可以看见输出曲线（黄色）变得更加平缓。 这一切操作和平时使用电路完全一样 。 对于简单的使用，用上面的方法就可以满足你的要求了。如果要更进一步，你也可以参考ahdl库的内容，来设计一个自己的单元。

ahdl库中的所有代码文件都是直接可见的。你可以直接打开ahld库对应单元的veriloga文件。上面使用的比较器代码如下：


`include "discipline.h"
`include "constants.h"

// $Date: 1997/08/28 05:48:45 $
// $Revision: 1.1 $
//
//
// Based on the OVI Verilog-A Language Reference Manual, version 1.0 1996
//
//
//--------------------
// comparator
//
// -  comparator
//
// sigin:    (val,flow)
// sigref:    reference to which 'sigin' is compared (val,flow)
// sigout:    comparator output (val,flow)
//
// INSTANCE parameters
//sigout_high  = maximum output of the comparator  (val)
//sigout_low   = minimum output of the comparator  (val)
//sigin_offset = subtracted from 'sigin' before comparason to sigref (val)
//comp_slope   = determines the sensitivity of the comparator []
//
// MODEL parameters
//{none}
//
// Compares ('sigin'-'sigin_offset') to 'sigref' - the output is related to
// their difference by a tanh relationship.
//
// If the difference >>> 'sigref', 'sigout' is 'sigout_high'.
// If the difference = 'sigref', 'sigout' is ('sigout_high' + 'sigout_low')/2.
// If the difference <<< 'sigref', 'sigout' is 'sigout_low'.
// Intermediate points are fitting to a tanh scaled by 'comp_slope'.
//

module comparator(sigin, sigref, sigout);
input sigin, sigref;
output sigout;
electrical sigin, sigref, sigout;
parameter real sigout_high = 10;
parameter real sigout_low = -10;
parameter real sigin_offset = 0;
parameter real comp_slope = 1000;


   analog begin

  @ ( initial_step ) begin
     if (sigout_high <= sigout_low) begin
    $display("Range specification error.  sigout_high = (%E) less than sigout_low = (%E).\n", sigout_high, sigout_low );
    $finish;
     end
  end

  V(sigout) <+ 0.5 * (sigout_high - sigout_low)
* tanh(comp_slope*(V(sigin, sigref)- sigin_offset))
  + (sigout_high + sigout_low)/2;
   end
endmodule
（摘自Cadence的ahdl库）




ahld库的代码通常有很不错的书写习惯。可以看见，代码开头是一大片注释。包括版本信息、书写日期、参数介绍以及简单的工作原理的介绍。如果我们书写Verilog-A的代码，也应该如此，以便将来修改和维护。

由于实现的是一个比较简单的比较器，可以看见它的实现代码也很简单。在一开始initial_step中，进行了参数的校验工作。假如用户将两个输出参数的大小写反了，这里将直接结束代码。

在实际的计算中，它直接使用了一个数学函数tanh。tanh函数模拟了一个随着输入变化、并且在曲线两端缓慢饱和。关于tanh的详细描述，读者可以查阅第三章。

假如我们现在需要的是一个带有磁滞的比较器，我们就可以从这段代码开始入手修改。我们只需要将以下语句：


 V(sigout) <+ 0.5 * (sigout_high - sigout_low)
* tanh(comp_slope*(V(sigin, sigref)- sigin_offset))
  + (sigout_high + sigout_low)/2;




修改为如下形式：


real real_offset;

real_offset = sigin_offset + (V(sigout)>(sigout_high + sigout_low)/2)?(1e-3):(-1e-3);
V(sigout) <+ 0.5 * (sigout_high - sigout_low)
* tanh(comp_slope*(V(sigin, sigref)- real_offset))
  + (sigout_high + sigout_low)/2;
   end




我们通过添加了一个内部数值real_offset，再判断输出是否高于电源电压的一半来给这个比较器添加了2mV的磁滞效果。当然这里的写法比较简单，还可以做将磁滞值作为参数，将磁滞改变范围作为参数等等细化工作。

即使你不使用Cadence开发环境，你同样可以利用ahdl库的里丰富的代码，来创建你自己的Verilog-A库。

2.6.2 Model Writer

在Cadence开发环境中，有一个可以更加快捷的生成简单Verilog-A模型的工具：Model Writer。我们在Cadence中新建单元（creat new cell），选择类型为model writer，即可调出Model Writer的界面。主界面如图2-10所示。

图 2-10 可使用的单元列表

Model Writer左侧列表显示了现在可以用Model Writer 生成的模块列表。它只支持Verilog-A。在Verilog-A下面显示的是各种单元类型。 如图2-11，展开列表，点击其中的某个具体单元，我们能看到右侧显示了一些描述信息、版本号等。

图 2-11 单元细节

同样的我们也找到了类似上面使用过的这个比较器模型。选中比较器模型，如图2-12所示。

图 2-12 比较器的基础模型

选中这个比较器，点击Next，进入修改模型默认属性节目。Model Writer提供了更加多的参数供选择，如图2-13所示。即使在这里不确定你需要的默认值，你依然可以在最后结果的代码里修改参数默认值。

图 2-13 可选择的属性

再次点击Next，直接可以看见Model Writer为你生成的代码。Model Writer会把你输入的内容，比如参数默认值和模型名称用红色高亮。如图2-14所示，代码窗口显示了生成的代码，被修改过的内容已经显示为红色。

图 2-14 生成的代码

这里点击Finish，Candence会继续为你生成模型的图标（symbol）。接下来就类似前面使用一般单元那样可以使用这个模型了。不再赘述。

接下来可以看一下Model Writer为我们生成的Verilog-A代码。


// FUNCTION: Tanh Comparator
//  VERSION: $Revision: 2.8 $
//   AUTHOR: Cadence Design Systems, Inc.
//
// GENERATED BY: Cadence Modelwriter 2.31
//   ON: Fri Jul 18 17:56:26 CST 2014
//
// Description: Comparator
//   Comparator with discontinuity removed by
//using a tanh() transfer function.
//   Has hysteresis and optional slew rate limiter.
//                            
//This model is an example, provided "as is" without express or
//implied warranty and with no claim as to its suitability for
//any purpose.
// 
// PARAMETERS:
//   slope = Gain of the comparator's input stage
// 

`include "discipline.h"
`include "constants.h"

// model compt_test4 - Comparator
// INSTANCE PARAMETERS:
//
module compt_test4 (vin_p, vin_n, vout) ;
 input vin_p, vin_n;
 output vout;
 electrical vin_p, vin_n, vout;

 parameter real   hys = 2m  from (0:inf);
 parameter real p_off = 10.0u;
 parameter real n_off = -10.0u;
 parameter real slope = 1000.0 from (0:inf);
 parameter real   one = 5.0;
 parameter real  zero = 0.0;

 parameter real slewp = 10.0M  from (0:inf);
 parameter real slewn = -10.0M  from (-inf:0);

 parameter integer  traceflag = 0;
real vin, mag, offset, halfhys, outv;
analog begin
  @(initial_step("ac","dc","tran","xf"))  begin
halfhys =  hys/2.0;
        mag = (one - zero ) / 2.0;
    offset = (one + zero ) / 2.0;

if (traceflag)  begin
   $strobe("%M  Initialization: vout: %g  hys/2:%g", V(vout),  halfhys);
end
  end

  vin =V(vin_p) + p_off - V(vin_n) + n_off;

  @(cross(vin - hys, +1))  begin
  halfhys = hys /2.0;
  if(traceflag)
 $strobe("%M at %g sec. output going high vin(p-n): %g  hys/2:%g", 
   $abstime, vin, halfhys);
  end
  @(cross(vin + hys, -1))  begin
  halfhys = -hys /2.0;
  if(traceflag)
 $strobe("%M at %g sec. output going low vin(p-n): %g  hys/2:%g", 
   $abstime, vin, halfhys);
  end
outv = mag * tanh(slope*(vin-halfhys)) + offset;
   V(vout) <+  slew( outv, slewp, slewn);
end
endmodule





虽然由于增加了很多参数，代码的长度增加了不少，但是这里的实现方式实际上和我们上面的并无本质差别。

这里vin是输入的差分电压，减去一些我们设置的输入失调电压，也就是相当于一个理想的比较器会看见的输入电压。然后，通过检测输入电压vin的自高而低，以及自低而高的穿过零点的事件，来生成不同的磁滞电压。

在输出语句上，它依旧使用tanh函数来生成渐变饱和的曲线。最后它使用了slew函数，来保证在瞬态仿真时，输出电压会按照有限的转换率变化。 这里有一个概念：tanh函数生成了对应输入的渐变饱和输出曲线，也就是说输入为0和输入很小-例如为1pV时，输出的差别不会直接从电源电压跳到地电压。这样可以保证在DC上连续的输入点有连续的输出，而没有突变。而slew函数则是在瞬态仿真的时候，保证当输入突然从负电压跳到正电压的时候，输入也会随着时间连续变化，而不会出现时间上的突变点。

另外值得注意的是，这段代码设置了一个参数traceflag。当用户设置这个参数为不为零的数值的时候，代码会在仿真初始阶段以及每次输入跳变的时候，输出各种调试信息。这种做法对于复杂的系统仿真中的调试是很有好处的。通常我们也可以用类似“assert”，“debug”之类的词作为调试开关参数。

2.6.3 Schematic Model Generator

如果你觉得Model Writer还是不能满足你自动生成Verilog-A代码的期望，那也许可以试试Schematic Model Generator（简称SMG）。SMG是Candence公司推出的新的自动模拟单元模型生成工具。它主要基于两个库：


	smgBasicMSLib

	smgAdvancedMSLib

	顾名思义，smBasicMSLib是比较基础的库，而smgAdvancedMSLib是高级的库。



这两个库里面的单元是比起Model Writer所使用的更加高度可定制的代码，所以打开库你会发现它里面的文件并不是一般的Verilog-A或者Verilog-AMS代码，而是一些名为bbt的XML文件。

使用smgBasicMSLib库只需要一般的Cadence Licence，而使用smgAdvancedMSLib需要额外的License。

如果你已经安装了SMG这个工具，打开库列表可以看见这两个库，如图2-15所示。

[image: 图像说明文字]


图 2-15 SMG库 不同于ahdl库，你不能直接使用或者查看这两个库的代码，我们需要使用菜单下的SMG生成工具来生成Verilog-AMS模型文件。

如图2-16，打开SMG界面，首先进入模型电路页面。用户可以使用现有的模型电路（Create New Initial Model Schematic），也可以选用已经存在的模型电路（Location of Existing Model Schematic）。

[image: 图像说明文字]


图 2-16 使用现有模块

SMG工具的大致流程是，用户调用SMG库中的单元，用画电路的方法把这些模块连接成为用户想要定制功能的模块。然后SMG工具将这个电路，结合里面的每个模块的代码，形成这个新模块的Verilog-A/AMS代码。

如果只是想要原样复用SMG库中的单元，则只需要在电路中加入这一个模块即可。

在主界面下选择“Create New Initial Model Schematic [NEW USER START HERE]”可以创建新的模型电路。如果选择“Location of Existing Model Schematic”，则选择以前使用过的电路。

如图2-17所示，出现选择新建一个模型电路的界面。

[image: 图像说明文字]


图 2-17 新建单元界面

界面上方添加输入输出端口。 Domain用于表示这个端口是数字域、模拟域或使用Wreal。在界面下方选择单元所在的库和单元名称。视图名称（View）必须是modelSchematic。

返回到主界面，打开电路界面，刚刚新建的端口已经被添加进来了，如图2-18所示。

图 2-18 创建单元

在电路编辑界面，这里右侧窗口列出了可以使用的单元。我们加入一个新创建的运放模型单元。

如图2-19所示。编辑新加入的单元，设置我们期望的属性值。在这里，SMG的子模块的使用和一般电路一样。这个例子里，作者只添加一个子模块。如果生成更加复杂的模型，可以用同样的方式加入更多的子模块。

[image: 图像说明文字]


图 2-19 编辑单元属性

回到SMG界面，如图2-20所示，可以根据需求修改一下生成代码的一些属性，比如文件格式、端口默认类型等。

[image: 图像说明文字]


图 2-20 编辑代码属性

首先，在Model Target页面上可以选择Verilog-AMS，Verilog-AMS Stub，Wreal或者Wreal Stub。顾名思义，选择带有Verilog-AMS的会生成Verilog-AMS代码，而选择带有Wreal会生成Wreal代码。Stub表示不生成具体的代码，只生成模块和端口的声明。

其他还可以选择单独保存为某个文件，或者（最常见的）保存为库中的某个view。

点击生成，得到一些信息，如图2-21所示。正常的话，应该没有出现Error信息。可以看到单元的ams代码已经生成了。

图 2-21 代码生成

可以看到生成的代码如下：


// SMG Model created by swu on Oct 13 16:29:41 2014
// From Model Schematic OZ04_S18/SMG1/modelSchematic
// From smgSettings OZ04_S18/SMG1/verilogams
// Using @(#)$CDS: virtuoso version 6.1.5-64b 06/11/2013 20:05 (sjfnl006) $
// In /design/sp_c/oz11/users/shunmin.wu+oz11+oz11a+27

`timescale 1ns/1ps
`define SMGTIMEUNIT 1e-9
`define SMGTIMEPRECISION 1e-12
`include "constants.vams"
`include "disciplines.vams"
`include "smgDefinitions.vams"

module SMG1(inp,inn,vcc,gnd,out);
// Pin directions
inout  vcc,gnd;
input  inp,inn;
output out;
// Pin disciplines
electrical inp,inn,vcc,gnd,out;
// Variable Declarations
electrical node1_I0; 
electrical node2_I0;
real n_I0[0:0]; 
real d_I0[1:0];  
real outVal_I0; 

analog begin
// Analog Block
@ (initial_step) begin  
n_I0[0] = 1;
d_I0[0] = 1; 
d_I0[1] = 1 / (`SMG_M_TWO_PI * 100K);
end  
V(node1_I0) <+ 1e6 * V(inp, inn);
outVal_I0 = V(node1_I0); 
if (outVal_I0 > 5)   
outVal_I0 = 5;   
else if (outVal_I0 < 0)  
outVal_I0 = 0;   
V(node2_I0) <+ outVal_I0;
V(out) <+ laplace_nd(V(node2_I0), n_I0, d_I0);  
$bound_step(1);  
end

endmodule





综上步骤，我们没有输入任何一行代码，就自动生成了一个Verilog-AMS模块。 SMG工具不但可以自动生成Verilog-A/AMS代码，也可以生成WREAL代码。因此，有兴趣的话也可以是用这个工具生成的代码作为学习和书写WREAL代码的基础。

SMG工具的一些功能需要单独的License，作者受条件所限也没有办法尝试其所有的功能。希望有条件的读者进行尝试并和作者探讨。





第3章 详细语法介绍和模型范例

这一章主要讲解Verilog-A/AMS的详细语法。 Verilog-A/AMS参考了Verilog-HDL的很多语句。而Verilog-HDL语言又参考了C/C++以及一些Perl语言的语法。相比这几种语言，Verilog-A/AMS所使用的语法、所包含的函数都要少得多。因此，如果熟悉这些任何一种语言的语法，都应该可以很轻松地掌握Verilog-A/AMS。

Verilog-A/AMS排除了灵活多变的语法和多样性的函数，因此一般的语句只要写过几次，语法出错的概率并不大。但Verilog-A/AMS中的函数使用受到多种限制，而且Verilog-A/AMS语言更多地是依赖于仿真器而非依赖于代码解释器的。因此在书写的时候，更多的是要考虑到代码在仿真中的实际行为是否符合设计目标。

3.1 语法

语法表示代码的书写格式和特殊符号的使用。Verilog-A/AMS的语法简单而严格。

3.1.1 注释

Verilog-A/AMS中的注释分为两种：


	
单行注释


// comments content

// next comment line



单行注释用符号 // 开头来表示。






	
多行注释


/* this is a example of

Multiple comment line */



多行注释用前后的 / 和
 / 符号对进行表示。 和其他程序语言一样，在Verilog-A/AMS中，注释的语句不加入仿真。注释用于保留和解释语句信息，方便未来阅读和检查代码。





3.1.2 命名

在Verilog-A中，所有的网线，节点，模块，函数等等都需要命名。 命名的词可以包含：数字+字母+下划线，第一个字符必须是字母或者下划线。 或者，你也可以选择用反斜杠开头的命名，例如【\inputA】。当然，除了对代码加密外，没有任何理由使用如此古怪的命名。

命名模块或者端口的首要目的是表明对象功能。例如对一个模块起名PeakCurrentDetector或者adc_8bit就能够让这个模块的基本功能变得一目了然。而类似aaa，bbb这样子的命名显然无法让使用者了解模块的功能。

3.1.3 数字

Verilog-A/AMS中的数字不但包含0-9的字符，还包括一些常见的工程字符。


	整数
[+|-][0-9] 表示十进制数
100
9
01



或者
‘b[0-9|z|x] 表示二进制数
‘b011x11
‘b0010100010







表示格式时，文中的符号[]表示其中的内容是可选。 符号[a|b]表示可能出现a、或者b、或者空白。 符号[0-9]表示可能出现从0至9的字符。 下文的表示方法同此。


	浮点数
[+|-][0-9]e[+|-][0-9]
1e6 科学计数法
1.34e2 小数计数法



字符单位 很多时候工程师更习惯于使用字符来表示数量级。
T=1012
G=109
M=106
K=103
k=103
m=10-3
u=10-6
n=10-9
p=10-12
f=10-15
a=10-18



使用时可以写成
1K
10M
1p



数字是代码中的常量。



3.1.4 字符串

我们会在strobe或者print之类的调试语句内使用字符串。但在Verilog-A里一般变量是不能存储和修改字符串内容的。因此，模拟域中，所有的字符串都或者是常量，或者是读取来自文件。 如果使用Verilog-AMS，我们可以使用reg变量来保存字符串内容，比如：


    Reg[8*12:1] stringvar;
    Stringvar = “Hello World!”;



即使在数字域，操作字符串的情况也是比较少的。主要使用字符串的场合还是配合print或者strobe函数输出调试信息。 关于调试输出的使用技巧，我们会在后面进行详细描述。

3.1.5 换行和结束符

在Verilog-A中，每行都要用“;”来结束一行，如果这样写：


$strobe("bit error rate = %f%%",
100.0 * errors / bits ) ;




实际上被认为是同一行语句，换行会被自动忽略。 所有的语句都需要;符号结束此行。

换行代表了代码可读性的好坏。


if(V(out, gnd)>vth) begin xout=1; yout=V(out, gnd); end




这样的连续书写方式并不是一个好习惯。


if(V(out, gnd)>vth) begin 
  xout=1; 
  yout=V(out, gnd); 
end




上面代码里这样的书写习惯明显更加容易阅读。

3.1.6 空白

Verilog-A中所有的空白都被当作是变量、运算符等之间的间隔：


$strobe("bit error rate = %f%  d",100.0 * errors / bits ) ;




空白不能留在%d中间。这样书写是错的。

3.2 变量

3.2.1 整数型


integer variable_name[[array_size]]




整数类型变量使用integer关键词进行定义。其数值只能是整数，数值范围-231 -- 231。 [array_size]表示在实际书写中“array_size”是一个可选的参数。外部的[]符号则是如果定义为array，代码中必须有的标志符。

3.2.2 实数型


real variable_name[[array_size]]




浮点数类型变量用real关键词进行定义。其数值范围10-37 -- 1037。

整数类型的数值范围-231 -- 231。浮点数类型的数值范围10-37 -- 1037。虽然这两个数值范围看起来已经很大了，但在一些情况下依然可能出现溢出，特别是整数型的数值。因此，对于表征信号的数值变量，应该尽量使用浮点数类型。

3.2.3 参数


parameter [integer|real] variable_name [from|exclude]




参数是模型实例化的时候可以进行设置的变量值。用户可以对一个模型，使用不同参数进行实例化，从而生成性能不同的单元模型。

参数可以是整数型或者实数型。参数带有一个默认值。如果你直接使用而不做外部设置，这个默认值会起作用。可以使用from和exclude的组合来限定可以使用的参数值的范围。 例如：


parameter integer rin from [100K:10M]




这用来表示这个参数的取值必须大于100K而小于10M。


parameter integer rin from [100K:10M] exclude 200K




这用来表示这个参数的取值必须大于100K而小于10M，而且不能为200K。

有一个特殊的数值inf，它只能在这里使用。


inf




inf表示无穷，通常在参数的范围限定内使用。例如：


parameter real input_inpedance =100K from[1K:inf] exclude [90K:99K] exclude 12K;




表示参数，默认值为100K，取值范围从1K到无限大，但是不能取12K，以及从90K到99K的数值。

我们也可以定义参数阵列。


    parameter integer array_size = 4;
    parameter real
    I_gc[1:array_size] = ‘{4{0.00}};



3.2.4 动态参数


dynamicparam [integer|real] variable_name [from|exclude]




动态参数和一般参数的不同是，动态参数可以在代码内被修改，而一般参数不允许被修改。也就是说动态参数既是一个参数，也是一个变量。 这常常用于初始值需要用户在实例化的时候确定，而运行值需要在运行的时候不断被代码修改的情况。

需要注意的是，动态参数的实现很大程度取决于仿真器。比如，Cadence仿真器支持这个参数，但不能用于改变电路结构 -- 例如，你不可以用它来动态改变器件的个数。在数模混合描述中你也不能将此参数值用于数字域的语句。而很多其他仿真器则根本不支持这种参数。

3.2.5 局部参数


localparam [integer|real] variable_name [from|exclude]




局部参数的特点是，你不能用另一个参数给它赋值。


parameter interger param1 = 10;
localparameter interger param2=param1*2;
出错！




你也不可以对它使用defparam语句（defparam语句会在后面介绍）。

局部参数可以避免本模块的参数被一些顶层电路或者仿真中间使用的语句修改，从而保持本模块的独立性。这种独立性往往会给那些尝试进行全局修改的设计人员带来意料之外的结果，因此使用局部参数的情况最好有特别说明。

3.2.6 genvar


genvar variable_name;




genvar变量是一种特殊的整数变量。它存在的原因是，数值变量会在仿真之前被预先计算（因为SPICE仿真器不做代码解释的工作）。因此直接使用数值变量的循环语句去访问总线端口的写法会失效。我们用genvar变量作为一个循环访问总线的简便写法，它会被预先展开，在编译阶段就可以保证访问端口的每一个语句都是正确的。

genvar的数值只能是整数。genvar变量只能用于模拟事件代码内。

genvar最常见的用处是方便访问总线。如果没有genvar我们设置一个16总线的语句会是：


V(out[1])<+out[1];
V(out[2])<+out[2];
...
V(out[16])<+out[16]; //重复16遍




如果我们使用genvar，代码可以写成：


genvar i;
for(i=0;i<16;i+i+1) begin
  V(out[i]) <+ out[i];
end




四行代码显然比十六行不断重复的代码要简洁的多。如果是一些检测总线的语句，这种美化代码的效果会更加明显。

3.2.7 网线和节点


electrical net_name;




模块需要内部节点的时候就使用electrical语句。内部节点的使用和模块端口类似，你可以检测其电压电流值，也可以对其赋值。 Verilog-A中使用内部节点有这样一些优点：


	它可以保留内部计算的波形并显示，而如果使用数值变量则不行。




	一些函数的多次操作需要借助内部节点，比如ddt。




	在RF仿真中避免使用变量保存结果而产生Hidden State错误。



如果模块比较复杂，保留一些计算结果到内部节点上会是一个比较好的方式。虽然这样会增加一些仿真器的计算开销，但有限的几个节点不会影响总体的仿真速度。

在Verilog-AMS中也支持wire和reg作为数字域的内部节点。

3.2.8 地

默认情况下，如果你在Verilog-A中写V(out)，给出的结果是out的电压相对于0电压的值。如果你在某个模块内使用ground语句：


ground net_gnd;




这样，V(out)在这个模块内给出的值就是out相对于net_gnd的电压值，即


V(out) = V(out, net_gnd)




需要注意的是，如果你这么写了，你不可以再使用V(net_gnd)这样的写法，这样会报一个错误。

3.2.9 wreal


wreal net_name[array_size];




用来表示real数值的节点，适用于wreal类型的代码。 本文中不做wreal的介绍。

3.2.10 分支


branch (net_name1, net_name2) branch_name;




branch用于表示并联关系的多个分支。当多个节点之间有比较复杂的相互连接关系，或者节点名称比较复杂的时候，每次都写V(节点A，节点B)会使得代码相当难看。使用分支命名可以简化书写，特别是一些总线之间的分支。

使用branch的例子可以参看第2章。

3.3 运算符

3.3.1 一元运算符

一元运算符通常位于变量和数值的前面，对单一的变量和数值进行操作。


operator variable_name;




表格 3-1




	
运算符


	
优先级


	
定义


	
操作对象







	
+


	
1


	
正数


	
integer, real





	
-


	
1


	
负数


	
integer, real





	
！


	
1


	
逻辑反


	
integer, real





	
~


	
1


	
按位取反


	
integer





	
&


	
1


	
按位与


	
integer





	
~&


	
1


	
按位与非


	
integer





	
|


	
1


	
按位或


	
integer





	
~


	
1


	
按位或非


	
integer





	
^


	
1


	
按位异或


	
integer





	
^~


	
1


	
按位异或非


	
integer







+、-符号用于表示数值的正负。

！符号（逻辑反）的操作是判断变量的值是否为0，只要为0，返回1；其他情况，无论变量的具体数值是多少，返回都是0。


a=101;
!a is 0
a=0;
!a is 1




按位取反则是将变量视为2进制数值，对每一位进行取反，例如：


a=101
a equals 0b1100101
~a equals 0b0011010



其他按位运算符也是按照这样的形式进行计算。 对逻辑操作有疑问的读者可以查阅其他程序设计语言或者数字电路设计的书籍。

3.3.2 二元运算符


variableA_name operator variableB_name;




二元运算符通常位于两个变量或者数值之间，对两个变量和数值进行操作。如上表示则操作数为variableA，被操作数为variableB。

表格 3-2




	
运算符


	
优先级


	
定义


	
操作对象







	
+


	
3


	
加法


	
integer, real





	
-


	
3


	
减法


	
integer, real





	
*


	
2


	
乘法


	
integer, real





	
/


	
2


	
除法


	
integer, real





	
%


	
2


	
取余数


	
integer, real





	
<


	
5


	
小于，返回0或1


	
integer, real





	
>


	
5


	
大于，返回0或1


	
integer, real





	
<=


	
5


	
小于等于


	
integer, real





	
>=


	
5


	
大于等于


	
integer, real





	
==


	
6


	
等于，返回0或1


	
integer, real





	
!=


	
6


	
不等于，返回0或1


	
integer, real





	
===


	
6


	
恒等，返回0或1，判断包括X和Z值


	
integer





	
!==


	
6


	
恒不等，返回0或1，判断包括X和Z值


	
integer





	
&&


	
10


	
逻辑与，返回0或1


	
integer, real





	
||


	
11


	
逻辑或，返回0或1


	
integer, real





	
&


	
7


	
比特按位与


	
integer





	
|


	
9


	
比特按位或


	
integer





	
^


	
8


	
比特按位异或


	
integer





	
^~


	
8


	
比特按位异或非


	
integer





	
<<


	
4


	
按位左移


	
integer





	
>>


	
4


	
按位右移


	
integer





	
or


	
11


	
事件或


	
event







加减乘除大于小于计算比较直观，不做具体介绍。 取余计算用左侧算数除以右侧算数，得到不能被整除的部分。


a=10;
b=3;
a%b equal 1





等于符号==主要用于判断语句中，表示左右表达式或者变量是否相等。

恒等符号===和一般等于判断的差别在于，===符号的计算也判断Z和X值。表中的X和Z分别表示数字域中的“不确定态”和“高阻态”。由于模拟域中不存在这两个状态，如果你使用Verilog-A语法，则可以认为两者是一样的。

不等于号!=和恒不等号!==之间的差别也是一样的。

逻辑与&&和逻辑或||的结果都是一个非0即1的整数。它们的操作数具体数值同样被忽略，对于非0的数，都被认为是逻辑1。


a=102;
b=2;
a&&b equal 1




而按位逻辑操作，都将操作数视作2进制数，对每一位上的数值进行逻辑计算。


a=102
b=2
a&b equal 2





左移<<和右移>>运算符将左侧算数视为2进制数，然后左移（或者右移）右侧算数个bit，得到移位完成后的数值。


a=102
b=2
a>>b equal 25





‘事件或’or表示两个事件之间可以用或来公用相同的代码。比如：


@(cross(V(netA)-1, 1) or cross(V(netB)-1, 1)) begin
end




3.3.3 三元运算符

三元运算符只有一个。 表格 3-3




	
运算符


	
优先级


	
定义


	
操作对象







	
A?B:C


	
12


	
A是否为真，是则返回B，否则返回C


	
integer, real







使用方法：


dataout = datain>vth?V(VCC):V(GND);




判断，如果datain的值大于vth，则dataout为V(VCC),否则dataout为V(GND)。 很多时候，这个运算符可以用来替代简单的if-else语句。

3.3.4 其他运算符

另外还有一些必不可少的运算符。

表格 3-4




	
运算符


	
优先级


	
定义


	
操作对象







	
=


	



	
赋值


	
integer, real





	
<+


	



	
赋值


	
integer, real





	
()


	



	
分隔


	
ALL







如前面所述，=赋值用于变量，<+赋值用于信号。并且<+赋值符号具有叠加性。读者可以参考2.1.6节的详细解释。 括号（）则用于分隔语句。

3.3.5 运算符优先级

运算符优先级从1最高（最优先计算），到12（最后计算）。

表格 3-5




	
运算符优先级


	
运算符







	
1


	
+ - ! ~





	
2


	
* / %





	
3


	
+ -





	
4


	
>> <<





	
5


	
< <= > >=





	
6


	
== != === !==





	
7


	
&





	
8


	
^ ^~





	
9


	
|





	
10


	
&&





	
11


	
||





	
12


	
?:







和绝大多数程序语言一样，不建议记忆这些运算符的优先级。在实际书写中，尽量使用括号分隔运算符即可，以便于清晰表示操作优先顺序。

3.4 数值计算函数

数值计算函数顾名思义主要用于数学计算，例如计算次方、对数、三角函数等。这些函数的名称和使用方法也和其他语言中相当类似。

3.4.1 基础函数

表格 3-6




	
函数名称


	
定义


	
操作值范围


	
返回值







	
abs(x)


	
取绝对值


	



	
integer, real





	
ceil(x)


	
最小的大于等于x的整数


	



	
integer





	
exp(x)


	
指数


	



	
real





	
floor(x)


	
最大的小于等于x的整数


	



	
integer





	
ln(x)


	
自然对数


	
X>0


	
real





	
log(x)


	
对数


	
X>0


	
real





	
max(x, y)


	
最大值


	



	
integer, real





	
min(x, y)


	
最小值


	



	
integer, real





	
pow(x, y)


	
x的y次方


	



	
real





	
sqrt(x, y)


	
开根号


	
X>0


	
real







这些函数既可以用于数值变量，也可以用于信号值。但需要注意一些输入限制，例如log和ln不能输入负值。 对数值计算函数有疑问的读者可以查阅其他程序设计语言或者数字电路设计的书籍。

3.4.2 三角函数

表格 3-7




	
函数名称


	
定义


	
操作数范围







	
sin(x)


	
正弦


	






	
cos(x)


	
余弦


	






	
tan(x)


	
正切


	
x!=n*(pi/2)





	
asin(x)


	
反正弦


	
-1<=x<=1





	
acos(x)


	
反余弦


	
-1<=x<=1





	
atan(x)


	
反正切


	






	
atan2(x, y)


	
x/y的反正切


	
y!=0





	
hypot(x, y)


	
均方根，即sqrt(x2+y2)


	






	
sinh(x)


	
双曲正弦


	






	
cosh(x)


	
双曲余弦


	






	
tanh(x)


	
双曲正切


	






	
asinh(x)


	
反双曲正切


	






	
acosh(x)


	
反双曲余弦


	
x>=1





	
atan(x)


	
反双曲正切


	
-1<=x<=1







每个函数对应一个三角函数的运算。具体计算可以查阅三角函数资料。

3.5 模块定义与实例化

这一节主要讲述模块本身定义、申明和实例化的语法。

一个Verilog-A/AMS模块，无论其中含有（或者不含有任何语句）什么语句，其最基本的构造就是模块本身的定义。有了模块的定义，我们才能够在其中加入描述具体工作的语句。模块完成后，也必须实例化为电路中的某个或者多个单元，才能开始实际工作。

3.5.1 模块定义

一个完整的模块定义的基本格式为：


    modules module_name (pin_name_A, pin_name_B...)
    input pin_name_A;
    output pin_name_B;

    logic pin_name_A;
    electrical pin_name_B;

    analog begin    
    end
    endmodule



其中，module_name为模块名称，其命名规则见本章第一节。模块名称必须和模块对应的电路、标志的名称相同，并且在一个库（Cadence中称为lib）中是唯一的。

函数声明中的端口名称列表不需要说明类型。需要注意的是，如果一个模块既有电路，又有Verilog-A/AMS代码，代码中的端口声明顺序要和电路中的端口次序一致。否则在很多网表生成工具中会报错。在实际使用中，通常电路中实例化时（或者说模块图标上的）端口，会由软件自动生成，而避免这种错误。

端口类型需要在模块代码中声明。端口的类型包括：input, output, inout。根据端口的模拟和数字域类型，需要声明端口为：logic或者electrical。

最后，一个描述模拟或者混合信号的模块，必须有一个analog行为声明模块。所有的描述模拟电路行为的代码，都必须放在analog begin - end之间。

3.5.1.1声明变量


integer variable_A;
real variable_B;




Verilog-A/AMS支持整形和实数数值变量。变量不是构造模块的必需品，但使用变量存储中间计算结果能够优化代码。

定义变量的具体语法见本章第二节。

3.5.1.2声明参数


parameter par_name;




参数不是构造模块的必需品，但使用参数可以使得模块能够灵活的实例化为不同的单元。 参数定义的语法请详见本章第二节。

3.5.1.3声明内部节点


electrical node_name;




模拟信号的内部节点需要被声明为electrical。内部节点可以和模拟端口一样设置、获取电压电流值。 内部节点不是构造模块的必需品，使用内部节点可以在模块内部保留中间仿真结果。

3.5.1.4初始化行为描述


@(initial_step) begin
end




Verilog-AMS支持和数字行为描述同样的初始化行为描述。如果是Verilog-AMS则可以使用：


initial begin
end




3.5.1.5数字行为描述


always begin
end




Verilog-AMS支持混合信号模块，因此在模块内也可以声明数字信号的行为。数字行为的具体描述语句请参见数字设计书籍。

3.5.2 模块实例化

如果以生成电路单元为基础，最常见的实例化Verilog-A/AMS模块的方式是产生Verilog-A/AMS对应的图标（symbol），在电路中加入对应的模块symbol，并对相应的端口进行连接即可。这也是作者在本文中最常用的方式。它的优点就是直观，上下层次之间的连接关系一目了然。

另一种形式是在一个Verilog-A代码中实例化另一个Verilog-A模块，可以这样写：


module_name instance_name (pin_name_A, pin_name_B...)




参数module_name是已经被定义模块名称。参数instance_name是要被实例化的实例名称。端口列表必须是已经在上级模块中已经声明的端口或者内部节点，并且个数必须和模块声明列表中的相同。 模拟模块不能一次生成多个实例或者实例矩阵。

3.5.3 端口对应

模块生成实例时一个重要的内容是将实例化的模块的端口和模块定义的的端口进行对应。上面小节这种写法，模块的端口按照声明次序，和实例化语句中的各个端口对应。

也可以采用另一种写法实现端口对应：


module_name instance_name (.module_pin_A (pin_name_A), .module_pin_B (pin_name_B)...)




这种书写方法里，module_name是需要实例化的模块名称。module_pin_A表示被实例化的模块的端口，pin_name_A表示外部的端口或者节点。这里使用名称一一对应，书写次序并不重要。

3.5.4 设置实例模块的参数

模块生成实例另一个需要做的工作是设置参数。模块在声明的时候，有默认的参数值，但实例化的时候可以设定为新的参数值。我们这样做可以设置实例的参数：


module_name #(par_A, par_B...) instance_name (pin_name_A, pin_name_B...)




注意参数设置列表以符号#开始。 这个声明方式是按照参数的声明的前后顺序进行对应。也可以用如下方式用参数名称进行对应：


module_name #(.m_par_A(par_A), .m_par_A(par_B)...) instance_name (pin_name_A, pin_name_B...)




类似的，这里m_par_A表示模块内部声明的参数名称，par_A则是实例化时要赋给这个参数的值。对应列表的次序不重要。

对于参数，还可以在模块实例化之后，用defparam语句进行设置。其使用语法如下：


defparam instance_name.param_name = value;




参数instance_name是实例的名称。参数param_name是参数的名称。语句执行后，此单元的这个参数值被设置为value。

defparam语句支持多行连续书写：


defparam instance_name.param_name = value,
           instance_name.param_name2 = value2,
          instance_name.param_name3 = value3;




语句之间用“，”分隔，可以一次设置多个参数值。

因此，在比较复杂的情况，特别是几个实例化的模块互相之间使用的参数有关联的情况下，可以使用defparam，集中在一个地方设置多个实例的参数。

前面两种参数设置发生在模块实例化的时刻，而defparam的设置发生在语句执行的时候，因此如果几种设置参数都存在，最后的使用的是defparam语句内设置的参数。如果有多个defparam语句定义同一个参数，则根据代码先后最后出现的defparam语句生效。

3.5.5 实例化基本器件

除了可以实例化用户定义的模块，Verilog-A/AMS也支持所有SPICE预定义的器件，例如电阻，电容，电流源等。下面的例子显示了如何实例化几种常见的基本器件：


resistor #(.r(100K)) RL (out, pwr); //电阻
capacitor #(.c(1p), .ic(1)) C1 (vcc, out); //电容
inductor #(.l(1u)) L1 (vcc, out);  //电感
isource #(.dc(ibias)) Iin (gnd, in); //电流源




基本器件含有的参数可以用上面实例化讲到的同样的方法进行设置。由于一般来说不知道基本器件参数的定义次序，通常使用参数名称与参数值对应的方式。具体参数名称可以参考SPICE手册。 SPICE预定义的器件有时候支持一些特殊的参数，例如频率响应参数需要是一个数组的时候我们就这样写：


svcvs #(.gain(gain),.poles({-1M, 0, -1M, 0})) x_VF (vout, vgnd, vin, vgnd) ;




这里使用大括号｛｝定义一个数组。 如果参数是一个字符串的时候，我们可以这样写：


vsource #(.type("pwl"), .file("mydata.dat") X_VS(src,gnd);




这里，对于第一个type参数直接使用字符串常量。而后面的参数是时间点和电压值对应的波形矩阵。我们这里使用了先将数据存储在文件内，然后采用.file参数从文件中读取内容的方式。

实例化SPICE基本器件的功能在每个仿真软件中的实现都有所不同，有时候是名称变化（例如，电阻是r还是res），有时候则是参数支持的不同，有时候则是不支持一些基本的单元。读者需要具体查看自己使用的仿真平台的手册。

3.6 模拟事件

模拟仿真器进行的是连续时间的计算。一般的行为描述语句在每个时间点上都要进行计算。模拟事件语句则可以根据信号的某种行为是否发生，选择性的执行其内部的语句。它的作用像是数字电路仿真器中的事件驱动，只有当对应的事件发生了、条件满足了，内部的语句才进行执行。模拟事件简化了代码的书写，也在一定程度上减少计算量。

3.6.1 模拟事件触发


@(eventA or eventB) begin
end




Verilog-A中使用@符号来表示触发模拟事件。上面的代码表示：如果括号中的eventA或者eventB事件发生了，那么仿真器会执行begin和end之间的代码，否则跳过。你可以使用任意多个or来同时判断多个事件。 模拟事件的语句只能被放在模拟描述模块中，也就是analog begin和end之间。

3.6.2 initial_step 和 final_step


@(initial_step) begin
end
@(final_step) begin
end




我们在前面介绍过，initial_step 事件发生在仿真的开始阶段，而final_step只在仿真结束的时候被执行。

initial_step和final_step目前都只支持ac, dc, tran和noise仿真。也就是说，假如你使用别的仿真类型--比如RF--仿真，是不会触发这些事件的。

需要注意的是，如果手动结束仿真，比如一个100uS的tran仿真在90uS被手动结束，将不会触发final_step。

3.6.3 cross


@(cross(expr1[ ,direction [ ,time_tol [ ,expr_tol ] ] ] )) begin
end




cross事件表示当信号的值穿过0值（也就是从正值变为负值，或者反过来）的时候，其中的语句将会被执行。

参数expr1 表示检测的信号。参数direction 表示穿越方向，1表示正向（从负值变为正值），0表示双向，-1表示负向。direction 默认为0。

参数time_tol和expr_tol分别表示会被检测到的变化的时间和数值下限。比如将time_tol设置为1u，则所有持续时间小于1uS的毛刺将被忽略。即使这些毛刺的尖峰越0，也不会触发cross事件。

由于cross检测的是穿越0值这个动态的事件，在DC等静态仿真中它是不会被触发的。

代码示例：


@(cross(V(in)-vth,1)) begin
out = 1;
end




当输入电压超过vth的时候，设置内部变量out为1。

3.6.4 above


@(above (expr1[ ,time_tol [ ,expr_tol ] ] )) begin
end




above事件表示当信号值大于0的时候，其中的语句将会被执行。 参数expr1 表示检测的信号。参数time_tol和expr_tol分别表示会被检测到的变化的时间和数值下限。

和cross事件不同的在于，above事件是一个静态的检测。它在DC仿真，initial_step等静态仿真中也会被触发。

代码示例：


@(above(V(in)-vth)) begin
out = 1;
end




同样，当输入电压V(in)超过门限电压vth的时候，内部变量out被设置为1。

3.6.5 timer


@(timer ( start_time[ , period[ , timetol]] )) begin
end




timer事件被用于周期性的事件，比如产生时钟。 参数start_time表示开始时间。参数peroid表示周期时间。参数timetol表示在仿真中，最小允许的时间误差（也就是说，如果定义事件发生的时间是每10uS，timetol为1nS，则要求仿真器在每个10uS+/-1nS的时间点内必须触发事件）。

代码示例：


@(timer(0, 10u)) begin
clk = !clk;
end
V(out) <+ clk*V(vcc, gnd);




timer最常见的就是被用来产生时钟信号。

3.6.6 absdelta


@(absdelta( expr1, delta[ , time_tol [, expr_tol]])) begin
end




absdelta模拟事件发生在：当指定的信号，比上次absdelta事件发生时，其值变化超过delta值的时候。 参数time_tol和expr_tol分别表示会被检测到的变化的时间和数值下限。 absdelta事件会在这些情况下被触发：


	0秒时




	计算得到初始稳定状态




	变化量超过delta值



由于理论上来说在模拟域计算absdelta事件需要无限小的时间步长，它只能被用在always语句中。也就是说，它不能被放在analog块中。虽然它监测的是模拟信号，但却是一个工作在事件驱动仿真器下的代码。它的最小步长不是象SPICE仿真器那样根据收敛性变化，即使你定义了time_tol，事件的触发还是受到timescale精度的影响。abddelta函数只能被用于Verilog-AMS代码，并且仿真时需要混合仿真引擎。

需要注意的是absdelta事件如果设置不当，容易产生大量的计算。常见的问题是在absdelta事件内需要修改某个信号，而这个信号又会通过别的路径影响absdelta的指定信号。如果这中间涉及到模拟信号，就会产生类似震荡的效果而大大增加仿真量。最简单的避免这种情况的手段就是设置合理的time_tol和expr_tol值。

代码示例：


always@(absdelta(V(in),vth)) begin
out = out+1;
end




每当输入信号V(in)变化一个vth的时候，就会使得变量out加1。 看起来absdelta似乎很适合用于书写ADC的行为（当信号变化的时候修改输出码），但是在实际使用中由于很难添加一些非理想的因素，而且必须使用数字域描述，通常我们会选用其他方法。

3.7 模拟量计算函数

下面来介绍一些和模拟信号相关的函数。这类函数在使用上有一些限制，比如：


	它们的变量必须是信号变量，通常是电压或者电流。




	它们只能被放在使用genvar，常数，变量或者analysis的判断语句中。




	它们不能被用在循环语句中。




	它们不能被用在自定义函数中。




	它们不能被用在initial和always语句中。



因此，使用这些函数时要注意是否符合语法要求。

3.7.1 有限指数函数


limexp(expr);




limexp函数类似exp数值函数，得到常数e的expr的次方值。不同的地方是，limexp函数控制仿真时的迭代步长，比起exp函数有好的多的收敛性。

limexp最常见的就是用来描述二极管或三极管电压电流关系之类的指数增长信号关系。 代码示例：


analog begin
idio = IS * (limexp(V(a,c)/$vt) - 1);
I(a,c) <+ idio;
end




这段代码计算了一个二极管两端的电压电流曲线。其中变量IS是二极管的反向漏电流。

3.7.2 时间微分函数


ddt(expr [, abstol ]);




ddt函数把信号按照时间求导数。参数expr是输入信号。参数abstol是最大会被计算忽略的信号变化量。

ddt是一个时间相关函数，所以在DC仿真中它始终会输出0。

ddt函数的输入值是一个动态信号，但是它的返回值不是。所以假如你想获得二次微分，下面这样写是错误的：


V(out) <+ ddt(ddt(V(in))); //错误




你需要首先定义一个内部节点，利用这个内部节点间接的获得二次微分。二次微分的示例代码如下所示：


electrical internal_node;
V(internal_node) <+ ddt(V(in));
V(out) <+ ddt(V(internal_node));




V(out)的值是输入信号对时间的二次微分的结果。

3.7.3 时间积分函数


idt(expr [, ic [, assert [, abstol]]]);




idt函数将输入信号对时间进行积分。

参数ic是初始状态值。参数assert是一个整数型的变量值。当assert不为0的时候，idt从当前时间重新开始积分。参数abstol是最大会被忽略的信号变化量（被忽略表示当输入变化时，这个时间点上先不进行计算，其值依然会被积累）。

idt也是一个和时间相关的计算函数，但是对于不同的输入，它在DC或者AC这种静态点仿真下会给出不同的值。

Idt函数的积分计算和参数组合关系如下表：

表格 3-8




	
输入


	
函数返回值







	
expr


	
[image: 图像说明文字]
 从0时刻积分至t，其初始状态由DC仿真点决定





	
expr, ic


	
[image: 图像说明文字]
 从0时刻积分至t，其初始状态由ic值决定。如果仿真类型是DC，返回ic值。





	
expr, ic, assert


	
[image: 图像说明文字]
 从t0时刻积分至t，其初始状态由ic值定。T0是assert不为0的时刻。如果仿真类型是DC，assert不为0时返回ic值，否则返回0。





	
expr, ic, assert, abstol


	
[image: 图像说明文字]
 从t0时刻积分至t，其初始状态由ic值定。T0是assert不为0的时刻。







如果仿真类型是DC，assert不为0时返回ic值，否则返回0。|

需要注意的是，这里使用的ic值会影响仿真器对初始状态的计算。如果给一个不合理的初始值，可能导致收敛问题。 示例代码：


V(p, n) <+ R*I(p, n) + L*ddt(I(p, n)) + idt(I(p, n))/C;




这段代码在前面出现过。积分函数用于计算电容两端的电压值。

3.7.4 循环积分函数


idtmod(expr [, ic [, modules [, offset [, abstol]]]]);




idmodt函数将输入信号对时间进行循环积分。

循环积分和一般积分的差别是循环积分在达到某个值后会重新从0开始积分。

参数ic是初始状态值。参数modules是一个变量值，当积分达到modules值的时候，idtmod函数便重新开始积分。参数offset是为积分值添加的一个额外的值，它可以是一个变量。参数abstol是最大会被忽略的信号变化量。

modules和offset的值决定了idtmod函数的范围。


offset < idtmos() <= modules+offset




具体积分情况和几个参数之间的关系见下表：

表格 3-9




	
输入


	
函数返回值







	
expr


	
[image: 图像说明文字]
 从0时刻积分至t，其初始状态由DC仿真点决定





	
expr, ic


	
[image: 图像说明文字]
 从0时刻积分至t，其初始状态由ic值决定。如果仿真类型是DC，返回ic值。





	
expr, ic, modules


	
[image: 图像说明文字]
 从0时刻积分至t，其初始状态由ic值定。如果输出值大于modules或小于0，则对modules取余数。





	
expr, ic, modules, offset


	
[image: 图像说明文字]
 从0时刻积分至t，其初始状态由ic值定。如果输出值大于module+offsets或小于offset，则对modules+offset取余数。







示例代码：


phase = idtmod(fc + gain*V(in), 0, 1, 0);
V(OUT) <+ sin(2*‘M_PI*phase);




上面的代码使用循环积分计算相位从而产生一个正弦波波形。

3.7.5 信号微分函数


ddx(exprA , exprB);




ddx函数返回exprA对exprB求导的结果。在这里，对exprA和exprB有一些要求：


	exprA是一个信号值、变量、或者一个固定值。但不管怎样，如果exprA不随着exprB变化，函数总返回0。




	
exprB必须是一个模拟信号量，比如电压、电流。 可以这样使用ddx函数：

real volA; volA = V(inp, inn); V(out) <+ ddx(volA, I(senseA, gnd)); 对于exprA，有这样一些限制：






	它不能是一个动态的表达式的值。比如，它不能是ddt(...)表达式。




	它不能是一个嵌套值。比如，它不能是ddx(...)。




	它不能包含一个模拟函数的计算。比如，它不能是transition(...)。




	它不能是一个数组。比如，它不能是a[0]，即使a[0]始终被赋值为某个模拟量。




	它不能包含未定义值。




	它不能包含迭代计算。比如，I(a)<Gm*V(a); ddx(V(a), ...)会报错。 因此，使用ddx函数的时候要特别注意参数的属性。



3.7.6 延时函数


absdelay(expr , time_delay [, max_delay]);




延时函数将信号延时一段时间输出，同时信号的形状不发生变化。其延时是一个绝对的、时间上的延迟，而不对信号本身做任何变化，因此才被命名为abs(loute)delay。

参数expr是输入信号的表达式。参数time_delay是表示延时长短的一个变量，它必须大于等于0。如果你设置了参数max_delay，那么time_delay可以是一个动态的数值，只要它保持大于0并且小于max_delay。

参数max_delay表示允许的最大延时。max_delay必须是一个大于0的常数。

在DC等静态仿真里，信号通过absdelay函数后不发生变化。

3.7.7 瞬态函数


transition(expr [,delay [, rise_time [, fall_time [, time_tol]]]]);




transition是一个很常用的函数。它也被称为transition滤波器函数。因为它会减缓信号的变化，就像一个低通滤波器做的那样。

当有一个快速变化的信号输入到transition函数中时，它会给这个信号根据输入参数，增加延时，增加上升边沿和下降边沿时间，从而使信号变慢。而且transition函数也会告诉仿真器在增加的时间点上进行仿真，从而保证输出正确。

这里参数expr是输入变量。它必须是一个real型变量，来表示一个快速变化的信号值。

参数delay是一个大于等于0的变量，它表示信号的整体延时。

参数rise_time和fall_time是大于0的变量，他们表示最快允许的上升和下降边沿。如果设置了rise_time而没有设置fall_time，默认fall_time=rise_time。

参数time_tol是一个大于0的常数。它会告诉仿真器，以不大于这个值的时间步长进行transition信号的计算。

[image: 图像说明文字]
 图 3-1 信号经过transient函数 （摘自Cadence用户手册）

图3-1很好的解释了三个参数对与输入信号的影响。参数delay表示一开始的延迟时间。参数rise_time影响上升边沿的斜率，而参数fall_time影响下降边沿的斜率。

[image: 图像说明文字]


图 3-2 一个缓慢信号经过transient函数 （摘自Cadence用户手册）

假如由于延时或上升边沿时间，边沿还未达到最高点，但信号已经要开始下降了会产生什么样的结果。图3-2解释了这个问题。图中参数rise_time大于信号本身的宽度。因此上升边沿还未完成的时候就已经开始下降。这是实际的上升和下降斜率被自动计算以满足：波形在原定时间点恢复到结束值，即图中的“new destination”。

可以看到，一个很长的上升边沿会导致一个不完整的输出信号，这是我们在设置transition函数参数的时候需要考虑的。

另外需要注意的一点是，transition函数适用与快速变化的函数。比如数字的输出信号。假如将它用于一直缓慢变化的信号，会导致在大量时间点上进行延时和边沿变化计算，并且不断的发生如上图所示的边沿不完整情况，使得仿真时间变长。

3.7.8 摆率函数


slew(expr [,max_pos_rate [, max_neg_rate]]);




上面提到，限制缓慢连续变化的信号不适合使用transition函数，而使用slew函数可以很好的限制连续缓慢变化信号的速度。

slew函数的作用就是限制输入信号的变化速率。参数expr是表示输入信号的变量。参数max_pos_rate和max_neg_rate分别表示上升和下降的最大摆率。注意参数max_pos_rate是一个大于0的值，而参数max_neg_rate是一个小于0的值。

slew函数对于类似正弦波之类的信号特别有效，因为它可以在保证信号形状的同时进行速率的限制。

3.7.9 拉普拉斯变换函数

拉普拉斯变换是电路里非常基础的概念。拉普拉斯函数代表着时域中的电路所对应的频域响应。我们可以使用拉布拉斯函数来定义一个电路的频率响应，或者为信号增加一个滤波效果。

当我们为一个信号增加了拉普拉斯变换函数后，它的瞬态响应和频率响应都发生了变化，就像信号通过了一个真正的理想的滤波器那样。

拉普拉斯函数的主体是零极点。为了表示一个零极点函数，我们可以这样写 [image: 图像说明文字]


也可以这样写

[image: 图像说明文字]


所以也产生了好几种拉普拉斯函数的写法。

3.7.9.1 laplace_nd


laplace_nd(expr, nn, dd, [ep]);




laplace_nd函数用nn和dd定义两个数组分别表示极点和零点各个阶对应的参数，比如nn中的第一个数表示0阶的大小，第二个数表示1阶的大小。 比如：


V(pout, nout) <+ laplace_nd(V(pin,nin), {1,0,0}, {1,1,1});




定义的函数为：

[image: 图像说明文字]


需要注意的是，nn和dd参数假如在仿真中间被修改，laplace_nd函数仍然会使用其初始值。因为滤波器的形式在仿真一开始已经被确定了，不会随着变量改变而改变。 最后一个参数ep表示计算时的绝对值容差。是否会使用这个值取决于仿真器。

3.7.9.2 laplace_zp


laplace_zp(expr, zeta, rho [, ep]);




laplace_np函数直接用各个零极点的方式来表示s域函数。

ztea参数是一个实数对数组。每对实数确定一个零点。每对实数中的第一个数值表示零点的实部，第二个数表示零点的虚部，第三个数表示第二个零点的实部，第四个数表示第二个零点的虚部，依次类推。 rho参数用相同的方式表示一组极点。

zeta和rho的值都必须是确定的数值或者参数，不可以是变量。


laplace_zp(vin, {a,b}, {m,n,p,q});





用这种方式表示的s域函数等同于：

[image: 图像说明文字]


假如要在这里表示s而不是[image: 图像说明文字]
 ，则只需要将对应的a和b都设置为0即可。 参数ep表示计算时的绝对值容差。

3.7.9.3laplace_zd


laplace_zd(expr, zeta, d [, ep]);




这个函数被用来表示类似这样的s域函数

[image: 图像说明文字]


零点的变量a和b是zeta数组中的值。极点的变量则来自于数组参数d中。

参数ep表示计算时的绝对值容差。

3.7.9.4 laplace_np

laplace_np(expr, n, rho [, ep]);

这个函数被用来表示类似这样的s域函数

[image: 图像说明文字]


参数ep表示计算时的绝对值容差。

3.7.9.5 例子

比如说，我们要实现这样一个s域函数。 [image: 图像说明文字]


我们可以写成


V(out) <+ laplace_zp(V(in), {0,0},{1,2,1,-2});




或者


V(out) <+ laplace_nd(V(in), {0,1},{1,-0.4,0.2});



他们是等效的。

3.7.10 Z变换函数

我们知道z变换函数用于离散时间。因此，所有的z变换函数，都会需要一个参数t，用来表示采样时间间隔。一个单位z变换函数（也就是没有增益，没有零极点），等于一个间隔时间为t的采样保持电路。

另外，所有的z变换函数还都有参数tor和t0。其中，tor表示采样的边沿时间，它必须是一个大于等于0的数值。如果tor不为0，仿真器会根据你设定的上升下降边沿计算采样边沿。如果你设置tor为0，输出会在仿真中间加入一个不连续点。很明显，使用tor为0往往会使得仿真速度大大降低。t0则表示第一次采样（也就是实际z变换开始）发生的时间。

tor和t0会在仿真的初始阶段被计算并且存储，因此在仿真的过程中修改这两个变量的值不会影响结果。

类似拉普拉斯变换函数，z变换函数也有几种形式对应不同的函数书写方式。

3.7.10.1 zi_zp


zi_zp(expr, zeta, rho, T [, tor [, t0]]);




ztea参数是一组实数对。每对实数确定一个零点。每对实数中的第一个数值表示零点的实部，第二个数表示零点的虚部。 rho参数用相同的方式表示一组极点。

zeta和rho的值都必须是确定的数值或者参数，不可以是变量。


zi_zp(vin, {a,b}, {m,n,p,q});




用这种方式表示的z域函数等同于：

[image: 图像说明文字]


3.7.10.2 zi_zd


zi_zd(expr, zeta, d, T [, tor [, t0]]);




这个函数被用来表示类似这样的z域函数

[image: 图像说明文字]


3.7.10.3 zi_np


zi_zd(expr, n, rho, T [, tor [, t0]]);




这个函数被用来表示类似这样的z域函数

[image: 图像说明文字]


3.7.10.4 zi_nd


zi_nd(expr, n, d, T [, tor [, t0]]);




这个函数被用来表示类似这样的z域函数

[image: 图像说明文字]


3.8 仿真器函数

3.8.1 不连续点

我们并不希望在模型中出现不连续点。如果可以的话，我们可以用一个陡峭的变化曲线来替代真正的跳变点。但假如一个陡峭的曲线会导致我们损失仿真时间，我们就只能告知仿真器我们的模型中存在不连续点，以便仿真器收敛计算。


$discontinuity(expr);




这里的expr是一个大于0的整数，表示函数在多少阶数上存在不连续点。比如expr=0则表示输出函数本身存在不连续。如果expr=1则表示输出函数的导数存在不连续。

[image: 图像说明文字]


图 3-3 三角波形成的不连续点


上图摘自Cadence用户手册


discontinuity函数并不针对某个模块或者计算值。使用它的时候需要将它放置在可能出现不连续的点的计算的后一句。比如这个例子：


analog begin
@(timer(0, wavelength)) begin
Slope = 1;
start = $abstime;
$discontinuity(1);
end
@(timer(wavelength/2, wavelength)) begin
Slope = -1;
start = $abstime;
$discontinuity(1);
end

V(vout) <+slope*($abstime - start);
end




可以看见，这个例子里，输出每隔一段时间就反向，因此在一阶导数上存在不连续性。因此在写这个模型的时候，每当计算到输出的方向反向的时候，我们加上discontinuity函数。告诉仿真器这一步可能会出现不连续点。

一个使用discontinuity函数的特例是，当我们使用了limit函数产生不连续时，我们使用discontinuity（-1）来告知仿真器。详细解释见limit函数。

3.8.2 仿真步长


$bound_step(max_step)




bound_step用于设定仿真器的最大步长。通常我们会在仿真文件里面设置一个默认的仿真步长。在一个Verilog-A模型里，假如我们认为默认的仿真步长会导致误差，或者我们希望仿真步长会根据模型的情况（比如工作频率是一个可以修改的参数）变化而变化，我们就需要在仿真中间根据实际情况设定最大步长。

参数max_step表示，在未来max_step长的时间内，至少要进行一次计算。如果仿真器选择的步长小于max_step时长，则相当于此语句无效。

3.8.3 limit


$limit(exp);
$limit(exp, string, arg_list);
$limit(exp, function, arg_list);




limit函数同样是为了告诉仿真器可能产生不连续点。同时，limit函数还可以告诉仿真器可以使用的函数来限制计算中点到点的差值，从而帮助收敛。limit的名称来源就是告诉仿真器在这里计算中要进行一些限制。

如果只使用一个exp参数，则是表示：告诉仿真器，这里可能出现不连续点。如果仿真器收敛，则函数返回值是在允许误差之类的计算值。

第二个参数是告诉仿真器，建议使用来进行这里的计算的函数。

如果这个参数是string，这个字符串就代表仿真器内部计算函数的名称。最常见的两个计算函数是pnjlim和fetlim。不同的仿真器可能会支持不同的内建函数，请查阅具体的手册。

如果这个参数是function，则表示用户定义的函数名称。

如果最后系统依然没有办法收敛，或者给出的结果超过了前后一步的最大误差，可以使用discontinuity(-1)函数来表示这种情况。

这段代码很好的显示了如何使用limit和discontinuity(-1)。


module diode(a,c);
inout a, c;
electrical a, c;
parameter real IS = 1.0e-14;
parameter real CJO = 0.0;
analog function real spicepnjlim;
     input vnew, vold, vt, vcrit;
     real vnew, vold, vt, vcrit, vlimit, arg;
     begin
     vlimit=vnew;
     if ((vnew > vcrit) && (abs(vnew-vold) > (vt+vt))) begin
         if (vold > 0) begin
             arg = 1 + (vnew-vold) / vt;
             if (arg > 0)
                 vlimit = vold + vt * ln(arg);
             else
                 vlimit = vcrit;
         end else
             vlimit = vt * ln(vnew/vt);
        $discontinuity(-1);
     end
     spicepnjlim = vlimit;
end
endfunction

real vdio, idio, qdio, vcrit;
analog begin
vcrit=0.7;
vdio = $limit(V(a,c), spicepnjlim, $vt, vcrit);
idio = IS * (exp(vdio/$vt) - 1);
I(a,c) <+ idio;
if (vdio < 0.5) begin
     qdio = 0.5 * CJO * (1-sqrt(1-V(a,c)));
end else begin
     qdio = CJO* (2.0*(1.0-sqrt(0.5)) + sqrt(2.0)/2.0*(vdio*vdio+vdio-3.0/4.0));
end
     I(a,c) <+ ddt(qdio);
end
endmodule




在计算二极管压降的时候，使用了limit函数，并且告诉仿真器使用自定义的spicepnjlim函数进行计算。在自定义函数中，计算vlimit可能会产生不连续的情况，因此加上discontinuity(-1)。

3.8.4 过0检测


last_crossing(signal, direction)




我们使用cross或者above函数可以很方便的检测信号过0的事件。但有时候，我们需要知道前一次信号过0点是在什么时候。当然我们也可以通过cross或者above事件，存储每个时间点来实现这个功能。但在Cadence里提供了一个更加简单的函数last_crossing来帮助我们实现这个功能。

signal是检测是否过0的信号。direction=0表示任意方向过0都被检测，direction=1表示信号上升过0，direction=-1表示信号下降过0。 注意，last_crossing函数前面没有$符号，说明它不是一个系统函数，其他仿真器内不一定支持这个函数。

3.8.5 仿真参数

仿真器有很多默认的参数可以被Verilog-A代码使用。


$abstime()




获取瞬态仿真中的当前时间。


$realtime([time_scale])




同样获取瞬态仿真中的当前时间。差别在于可以用time_scale来表示时间单位。 timescale参数可以是1,10或者100加上时间单位。

表格 3-10




	
时间单位


	
时间长度







	
s


	
1秒





	
ms


	
1e-3秒





	
us


	
1e-6秒





	
ns


	
1e-9秒





	
ps


	
1e-12秒





	
fs


	
1e-15秒








$temperature()




获取仿真中的当前温度。返回值为绝对温度，也就是室温为300K。


$vt([temp])




获取仿真中的vt值。Vt=KT/q。你可以手动设置温度参数temp来获得不同于当前温度的vt值。


$simparam(“param” [, expr])




获取仿真器的参数。参数param可以是：

表格 3-11




	
参数


	
含义







	
scale


	
器件尺寸的倍数





	
gmin


	
仿真器会自动添加的最小跨导





	
iteration


	
本次迭代的次数







3.8.6 获得网表节点

随着仿真器的支持，Verilog-A代码不但能获取本模块的输入输出，它也能获得其他模块中的节点的值。比如你可以这样写：


V(out) <+ V(top.level1.level2.netA);




假定你仿真中顶层模块交top，依次往下为level1、level2。则你可以访问到level2这个模块内的节点netA的电压值。 如果level2不是一个真正的电路，而也是一个Verilog-A模型，这样写法需要netA是一个在level2的模型中明确定义为electrical的内部节点，否则会出错。

这样的书写方式，如果你指定的top.level1.level2有任何一层单元不存在，仿真会出错。

需要注意的是，在不同的仿真器内访问外部节点的电流值，会有很多限制，并不保证能正常工作。具体限制需要查看每个仿真器，甚至是不同版本之间的差异。

3.8.7 获得仿真类型


analysis(“analysis_type”);




我们可以用analysis函数判断当前仿真的类型是否在analysis_type中。 仿真类型如下表：

表格 3-12




	
仿真类型


	
含义







	
DC


	
静态或者工作点仿真





	
static


	
静态仿真，包括在AC或者NOISE仿真之前运行的DC工作点仿真，以及瞬态仿真的初始工作点仿真





	
tran


	
瞬态仿真







下表显示了不同仿真对仿真类型的返回值真假：

表格 3-13




	
类型


	
DC


	
Tran OP


	
Tran


	
AC OP


	
AC


	
Noise OP


	
Noise







	
static


	
1


	
1


	
0


	
1


	
0


	
1


	
0





	
ic


	
0


	
1


	
0


	
0


	
0


	
0


	
0





	
dc


	
1


	
0


	
0


	
0


	
0


	
0


	
0





	
tran


	
0


	
1


	
1


	
0


	
0


	
0


	
0





	
ac


	
0


	
0


	
0


	
1


	
1


	
0


	
0





	
noise


	
0


	
0


	
0


	
0


	
0


	
1


	
1







analysis函数常常被用于针对不同仿真类型给出不同的工作状态。例如：


If (analysis(“dc”, “ic”))
V(out) <+ 1;
else
V(out)<+ transition(outvalue, 5n, 1n, 1n);




这样在dc下保持电压为1伏，而在瞬态仿真中随着变量而变化。

3.8.8 AC信号源


ac_stim([“analysis_type” [, mag [, phase]]]);




ac_stim函数可以产生一个小信号的ac源。它只能作用于ac仿真，在大信号仿真中，其输出一直为0。 analysis_type默认为ac。虽然可以设置为其他值，但似乎没有意义，因为其他值会导致此函数输出一直为0。mag和phase参数分别表示信号源的幅度和相位。

3.8.9 噪声源

系统提供了三种噪声源。噪声源函数只在小信号噪声仿真中产生作用，其他仿真类型下输出都为0。


white_noise(pd [, name]);




产生白噪声。pd参数表示噪声谱密度，它以[image: 图像说明文字]
 或者[image: 图像说明文字]
 为单位。name参数表示噪声源的名称。在仿真后的噪声结果中可以看见噪声源的名称。


flicker_noise(power, exp [, name]);




产生闪烁噪声。power参数表示在1Hz频率上的谱密度。它以[image: 图像说明文字]
 或者[image: 图像说明文字]
 为单位。exp参数是一个整数表示噪声随着频率变化的阶数。


noise_table(vector[, name]);




用表格的方式记录对应频率点上的噪声谱密度，从而表示一个噪声源。vector参数是一个数组。第一个数字表示频率，第二个数字表示这个频率点上的噪声。频率必须是依次递增的。

3.8.10 随机数


$random([seed]);




产生一个伪随机数。使用不同的seed可以产生出不同的随机数序列。使用相同的seed会产生出相同的随机数序列。

seed参数必须是一个变量而非常量，原因是仿真的时候每次运行random函数会新产生一个种子值，这个值需要被存储在seed变量里，以在下一次运行random函数的时候使用。

系统提供的随机数随机性并不是很好，我们在后面章节介绍一些使用的技巧。

3.8.11 带分布函数的随机数

random函数生成的随机数理论上来讲是平均分布的。很多时候我们还需要生成依据一定分布函数的随机数。Verilog-A/AMS提供的函数可以生成这些主要的随机分布函数。


	Uniform

	Gaussian

	Exponential

	Poisson

	Chi-square

	Student’s T

	Erlang



使用对应的函数所生成的随机数，随着次数增加，逐渐逼近对应的分布曲线。这通常用于Mont-Carlo分布之类的随机量仿真。

所有的带分布函数随机数函数都是两个一组。一个以dist开头的函数表示这个函数生成的数值会是一个整数数值；而rdist开头的函数则会生成一个实数数值。dist开头的函数可以用于数字和模拟域的描述语句中。而rdist开头的函数只能用于模拟域的描述语句中。

3.8.11.1 Uniform


$dist_uniform(seed, start, end);
$rdist_uniform(seed, start, end);




参数seed表示生成随机数的种子。start是生成数值的最大值。end是生成数值的最小值。

seed参数必须是一个变量而非常量，原因是仿真的时候每次运行random函数会新产生一个种子值，这个值需要被存储在seed变量里，以在下一次运行random函数的时候使用。对下面所有的随机数函数均是如此。

3.8.11.2 Gaussian


$dist_normal(seed, mean, sdev);
$rdist_normal(seed, mean, sdev);




高斯分布，又称为正态分布，大概是最常见的分布类型。所以这个函数使用了normal字样。 高斯分布的表示方式是由平均值（或称为中间值）加上方差。因此，高斯函数的参数也是，除了生成随机数的种子值seed以外，参数mean表示平均值，参数sdev表示方差。

3.8.11.3 Exponential


$dist_exponential(seed, mean);
$rdist_exponential(seed, mean);




指数分布，随机数分布曲线在平均值两侧成指数衰减。参数和其他函数一样，seed用于生成随机数的种子，mean表示平均值。

3.8.11.4 Poisson


$dist_poisson(seed, mean);
$rdist_poisson(seed, mean);




泊松分布，特点是总平均数等于方差数。参数seed表示生成随机数的种子值。Mean表示平均值。

3.8.11.5 Chi-square


$dist_chi_square(seed, df);
$rdist_chi_square(seed, df);




中文叫卡方分布，也叫X2分布，更具体的信息读者可以上网查阅。 参数seed依然是种子值。参数df称为”degree of freedom”，表示分布的扩展范围。

3.8.11.6 Student’s T


$dist_t(seed, df);
$rdist_t(seed, df);




中文称为学生分布或者T分布。 参数seed依然是种子值。参数df称为”degree of freedom”，表示分布的扩展范围。

3.8.11.7 Erlang


$dist_erlang(seed, k, mean);
$rdist_erlang(seed, k, mean);




中文读作”爱尔郎分布”。参数seed依然是种子值。参数K是Erlang分布中使用的阶数。参数mean表示平均值。

3.8.12 查表

除了使用计算式给出输入输出关系，也可以使用查表的方式得到结果。


$table_model(variable, table_source [, ctrl_string]);




这里variable是一组输入信号，输入信号的个数可以是任意个，只要数据表的维数和输入信号个数能够对应即可。

table_source参数是数据表的来源。它可以是一个文件，也可以是多个数组。 如果table_source是一个文件，这个文件的每一行格式必须是：


Line_index X_number Y_number ...... Out_number




Line_index是从1开始依次递增的索引号。

X_number和Y_number是函数中所写的variable的各个变量的可能取值。 Out_number则是根据X和Y得到的函数对应输出值。

比如一个表格文件可能是这样的：


\# index  X    Y    Out

1        1      1     2

2         1     2     5

3        1     3     10

4        2     1     7



同样的，使用#符号来表示注释，用空格和分行来间隔各个数字。如果我们使用这个文件作为数据表的来源，则函数写为：


V(out) <+ $table_model(V(x), V(y), “ table_filename”);





如果table_source是一组数组，则上面这个例子会被写成：


    X={1,1,1,2};
    Y={1,2,3,1};
    FOut={2,5,10,7};
    V(out) <+ $table_model(V(x), V(y), X, Y, FOut);



最后，我们还可以加上可选的ctrl_string。Ctrl_string是由多个字符串组成的，每个字符串的组成形式是：


[I|D]+[1|2|3]+[C|L|S|E]




第一个字母表示对对应的数组的处理方式。如果这个字符是I，表示ignore，仿真中这个数组会被忽略。如果这个字符是D，表示这个数组的值在计算中不会被差值。以上面的数值表为例，假如X和Y的输入为1,1.5的时候，默认情况下仿真器会在点（1，1）和（1，2）之间自动计算插值。但假如我们设置了D字符，则系统不会计算插值，而直接报告一个错误，表示无法计算。

第二个数字表示对应的数组项。1表示第一个数组（这里就是X），2则表示第二个数组项（这里就是Y），依次类推。

第三个字母表示如果需要进行插值计算，用何种方式进行插值。C (clamp)表示插值使用最接近的点所生成的水平线。L（linear）表示使用正切函数延伸最接近的数据点进行插值。S（spline）表示使用多项式曲线延伸最接近的数据点进行插值。E(error)表示如果需要插值的点不在范围内，则会报告一个warning。

3.8.13 输出结果

像绝大多数程序语言一样，Verilog-A也支持文字信息输出。它提供了5条语句。


$strobe(xargs);
$display(xargs);
$monitor(xargs);
$write(xargs);
$debug(xargs);





参数xargs非常类似C语言等中的print语句的变量要求。它的组成大致上是：


“output string %format_parA %format_parB”, varA, varB





这些函数也支持输出一些特殊字符，如下表：

表格 3-14




	
特殊字符


	
含义







	
\n


	
换行





	
\t


	
Tab





	
\


	
字符





	
\”


	
字符





	
\ddd


	
由ddd数值表示的字符





	
%%


	
字符%





	
%m或者%M


	
当前模块的名称







字符%m是很有用的，因为你常常需要在大量输出信息中寻找是哪个模块输出了错误。

格式参数format_par的组成：


%[+/-][0-9].[0-9][format_char]




+-号表示输出添加正负符号。数字表示输出数值的位数。 Format_char字符见下表：

表格 3-15




	
字符


	
含义







	
B,b


	
二进制数





	
C,c


	
字符





	
D,d


	
整数





	
E,e


	
实数，科学计数法





	
F,f


	
实数，一般表示法





	
G,g


	
实数，选择一种最短的表示法





	
H,h


	
整数，16进制数





	
O,o


	
整数，8进制数





	
R,r


	
使用字符，例如M=1e6





	
S,s


	
字符串







下面来说说这几种输出函数的差异。

Strobe和Display完全等效。

Strobe和Display在最后自动输出一个换行符，而write不会。

Debug会在每次仿真机进行迭代的时候输出。注意它会有非常多的输出内容！

其他函数总是输出，而monitor只有在参数值变化的时候才输出。Monitor可以很好的控制输出的数量。

3.8.14 退出仿真

系统提供了两个语句以便于随时退出仿真。


$finish(code);
$stop(code);




finish函数和stop的差别是：finish退出仿真后直接返回操作系统；而stop会给出一个tcl命令行，用户可以再进行一些tcl操作仿真结果。 参数code表示退出后自动打印哪些仿真信息。

表格 3-16




	
字符


	
含义







	
0


	
不输出





	
1


	
输出仿真时间和仿真位置





	
2


	
仿真时间，仿真位置，统计信息，CPU和内存使用量







3.8.15 文件操作

3.8.15.1 打开文件


$fopen(“file_name”, open_mode);




参数file_name表示文件名称。参数open_mode表示打开模式，可以选择r，w或者a。参数r表示用读打开文件，w表示用写模式打开文件，a模式表示用增加模式打开文件，如果文件不存在，则创建文件。

fopen函数返回一个32bit的文件句柄。其他文件操作函数都需要使用这个文件句柄。

在参数file_name中我们可以使用一些特殊字符来组成内建的一些文件（文件路径）名称。

表格 3-17




	
字符


	
含义







	
%C


	
当前设计文件名称，比如input.scs





	
%D


	
当前年月日





	
%H


	
当前主机名称





	
%S


	
仿真器名称，比如spectre





	
%P


	
仿真进程号





	
%T


	
当前时间，时分秒





	
%I


	
当前模块名称





	
%A


	
仿真类型名称







每一个这类特殊字符后面可以跟一个或者多个修饰符，用来提取Unix文件名中的一部分。

表格 3-18




	
字符


	
含义







	
:r


	
文件所在的根目录





	
:e


	
文件所在的完整路径





	
:h


	
路径头





	
:s


	
路径唯





	
::


	
字符：







使用这些特殊字符和修饰符，可以针对不同的模块，不同的仿真，打开不同的文件。

3.8.15.2 读文件


$fscanf(fd, format[, storage_arg]);




fscanf用于读文件。参数fd是前面打开文件时得到的文件句柄。参数format是一个正则表达式，fscanf函数会不断的读文件以匹配这段正则表达式。在参数format中用%d、%e等特殊字符匹配文件中的数值（见$strobe命令）。匹配到的数值会被存储到storage_arg所表示的变量中去。

由于Verilog-A没有用于保存字符串的变量，所以读文件的时候也不能保存字符串。

例子：


$fscanf(fd, "parameter = %d", dbl);




3.8.15.3 写文件

有三个函数可以用来写入文件。


$fstrobe(fd, xargs);
$fdisplay(fd, xargs);
$fwrite(fd, xargs);




参数fd是打开文件时得到的文件句柄。参数xargs和$strobe函数的解释相同。

3.8.15.4 关闭文件


$fclose(fd);




最后，我们需要关闭文件。很简单，我只需要传入打开的文件的句柄即可。

3.9 自定义函数

除了使用系统提供的函数，我们还可以自己定义函数。


analog function [type] function_name
endfunction




参数type表示这个自定义函数的返回值，默认为real。 自定义函数受到不少的限制，有很多操作不能放在自定义函数内进行。


	不能包含ddt操作

	不能包含idt操作

	不能包含idtmod操作

	不能包含模拟事件



每个函数至少需要有一个输入变量，有一个返回变量。输入变量如此定义：


input varA;
real varA;




而返回变量直接使用自定义函数的函数名称。例如：


analog function real maxvalue
input a, b;
real a,b;
     maxvalue = MAX(a, b);
endfunction





当需要使用自定义函数的时候，直接调用函数名称即可。


V(out) <+ maxvalue(V(in), V(ip));




3.10 编译器语句

以下这些语句是为了控制编译器的一些行为。如果读者熟悉C语言，就应该了解宏定义的概念。这里的语句非常类似宏定义，它们在编译阶段工作，但对模型的计算和执行方式不起决定作用。 这些语句都以字符` 开头，注意区分字符 ` 和 ’ 。同时，这些语句不属于Verilog-A/AMS的主体代码，因此不用字符 ; 结尾。

3.10.1宏定义


`define macro_name macro_text
`undef macro_name




第一个就是Verilog-A/AMS中的宏定义。经过define宏定义的一段代码macro_text，可以直接被下面的代码用macro_name替代，直到代码中使用了undef语句。 同时，和C语言中一样，Verilog-A/AMS的宏定义也支持参数。 例如：


`define sum(a,b) ((a)+(b))




使用的时候


    If(`sum(p+q) < 5) begin
      c = 0;
    end



注意，使用宏的时候也要加上字符`。 如果不使用undef，宏定义有效范围为整个代码文件。

3.10.2 宏定义判断语句


`ifdef macro_name
`else
`endif




宏定义的判断语句也和C语言中的意义一样，判断是否已经定义了某个宏。通常宏判断语句是为了根据不同的情况，使用不同的常量（定义为宏）、执行不同的宏语句、或者包含不同的文件。

3.10.3 包含文件


`include file_name




include语句我们已经见过很多此了，包含某个文件到当前文件里来。实际上和其他语言的操作一样，这个文件的原文会被添加到include语句的位置上。

3.10.4 Timescale


`timescale time_period/time_precision




timescale语句用于定义模块的时间间隔和时间精度。Time_period实际上只对数字域的语句起作用。具体可以参考verilog的书籍。 Time_precision的值必须小于time_period。Time_precision值对数字域的仿真有影响。同时，time_precision也对transition函数，Z域函数以及一些延时函数有影响 -- time_precision值会作为这些函数的一些默认精度参数。

time_unit可以是数字1、10、100加上时间单位。时间单位列表如下:

表格 3-19




	
单位


	
含义







	
s


	
秒





	
ms


	
10-3秒





	
us


	
10-6秒





	
ns


	
10-9秒





	
ps


	
10-12秒





	
fs


	
10-15秒







3.10.5 设置默认限定类型


`default_discipline discipline qualifier scope




这个语句的作用是设定默认的网线类型。参数discipline可以使用：


reg
wire
tri
wand
triand
wor
wreal
trior
trireg
tri0
tri1
supply0
supply1




这些网线定义都是给数字域描述使用的，具体含义可以参考Verilog-HDL的书籍。模拟域的网线必须写上electrical。 也可以使用参数qualifier和scope来限定指定默认网线类型的范围。参数qualifier是需要设置默认类型的信号类型。而scope参数表示使用默认指定的模块。 例如这么写：


 `default_discipline logic trireg x_bfm.io




表示将x_bfm下的io模块内部所有连接到trireg信号的未定义网线类型设置为logic类型。 大多数情况下，我们都建议明确指出网线的类型，而不是依赖于宏定义。

3.10.6 默认上升和下降时间


`default_transition time




用于设置默认的上升和下降时间。transition和一些其他函数中，假如没有明确指明上升和下降时间，并且设置了这个宏定义，就会使用这个值。

如果没有定义这个值，而且在使用函数的时候也没有设置上升和下降的时间，默认参数值为0。但仿真器会根据它的仿真时间步长来进行实际运行。

3.10.7 重置设定


`resetall




这条语句会恢复所有的默认设置。你通常会在代码的第一行使用这个语句，以保证你的其他设置都是系统默认值。但大多数情况下，不使用这条语句，仿真器也会为每个模块使用默认设置。





第4章 Verilog-AMS的混合信号仿真

Verilog-AMS用于将数字信号的行为描述和模拟信号的行为放在同一个模块内实现。在Verilog-AMS中使用数子信号的描述需要读者了解Verilg-HDL的一些语法。这在很多讲解数字设计的书籍中都有阐述，本文不做介绍。这一章主要是绍一下混合信号仿真的概念和原理，以及一些简单的书写方法。

4.1 混合信号仿真基础

首先我们要了解一些概念。


域


在混合信号仿真中，模块被划分为不同的域。我们通常会有模拟域，数字域和混合信号域的称呼。在模拟域中，信号用连续的方式表示自身的属性，比如电压，电流值。在数字域中，信号用离散的方式表示自身属性，比如H，L，Z。

不同的域也会使用不同的仿真步长。模拟域使用迭代计算方式，时间步长根据计算需要而改变。数字域中计算步长固定，所有信号的变换都需要跟随此固定步长。

不同域之中，数值变量的计算并无差别，有差别的只是信号的计算。


上下文


Verilog-AMS的语句可以放在analog语句块中，也可以放在initial或者always语句块中。放在analog语句块中的语句被称为在连续时间上下文中。放在其他语句块中的语句被称为在离散时间上下文中。变量可以存在于任何一种上下文中。但一个变量不能存在于两个上下文中。


网线、节点、端口和信号


端口指的是模块实例化形成更高一级电路的时候，所产生的物理连接。每个端口包含两个连接，一个是向上一层连接到父电路的某个位置，另一个是向下连接到模块内的某个节点。

一根网线可以是在模拟域中，也可以在数字域中。多根连接在一起的网线形成一个信号。如果信号的全部网线都在模拟域中，则此信号为模拟信号；如果信号的全部网线都在数字域中，则此信号为数字信号；否则，此信号为混合信号。

类似的，端口也被分为模拟端口，数字端口和混合信号端口。

如果一些网线或者信号是模拟的，那么它必然对应一个节点。这个节点的电压在任何时候表征了些个网线的模拟部分分量的大小。


混合信号节点的计算


混合信号节点的值的计算基本上跟随模拟部分的计算方式。假如某个混合信号节点是由多个模拟网线组成的，那么计算的时候总是依据要求精度最高的那个网线。

4.2 模拟与数字系统的互动

Verilog-AMS支持在同一个模块内用数字和模拟的语句进行描述。也支持在同一个模块内，不同上下文之间互相访问网线和变量，以及不同域相互检测事件。一个模块内部可以包含多个always语句块，但是，在一个模块内，只能使用一个analog语句块。

4.2.1 从模拟上下文中访问数字

下表显示了数字上下文中的量在模拟上下文中的转换。

表格 4-1




	
数字域变量


	
模拟域变量


	
含义







	
Real


	
Real


	
数值不变





	
Interger


	
Interger


	
数值不变





	
bit


	
Integer


	
比特值高和低被转化为1和0， Z和X值无法转化，会在仿真中报错





	
bit[n:0]


	
Integer


	
bit[n]总被认为0，数值大小=bit[0]20+bit[1]
 21+bit[2]*22...







示例代码：


module dac_1bit(in, out);
input in;
inout out;
wire in;
logic in;
electircal out;
real vout;
analog begin
     if(in == 0)
     vout =0.0;
     else
     vout = 3.0;
     V(out) <+ vout;
end






4.2.2 从数字上下文中访问模拟

所有模拟上下文的值都可以直接被数字上下文语句读取。电压电流等信号默认为real变量。

示例代码：


module sampler (in, clk, out);
inout in;
inout clk;
output out;
electrical in;
wire clk;
reg out;
always @(posedge clk) 
begin
     out = V(in);
end
endmodule




上面的例子里，进行了一次默认的数值转换，只要V(in)对应的值大于0，则赋值为1。

4.2.3 在模拟上下文中检测数字事件

Verilog-AMS支持在模拟上下文中检测数字事件。事件的参数是数字上下文的一部分。

示例代码：


module sampler(in, clk, out);
input in, clk;
output out;
wire clk;
real vout;
electrical in, clk, out;
analog begin
     @(posedge clk, 1) begin
     vout = V(in）；
     end
     V(out) <+ vout;
end
endmodule




4.2.4 在数字上下文中检测模拟事件

同样的，Verilog-AMS也支持在数字上下文中检测模拟事件。


module sampler2 (in, clk, out);
input in, clk;
output out;
wire in;
reg out;
electrical clk;
     always @(cross(V(clk) - 2.5, 1)) 
     out = in;
endmodule




4.3 连接模块

当模拟域的网线和数字域的网线连接在一起时，它们不能直接进行计算。每个跨域连接点上都会自动或者手动的插入一个连接模块，用以将不同域的仿真值进行相互转换。

一般来说，连接模块会定义一个门限值，高于门限值的模拟域电压等于数字域的1，而低于门限值则会等于0。

了解连接模块有助于了解混合信号仿真的限制和可能产生的问题。

4.3.1 定义连接模块


connectmodule module_name(port) ;
endmodule



我们用connectmodule来定义连接模块。port为输入输出端口列表。端口属性类似一般模块中的端口定义，也有input,output,logic和electrical等。由于连接模块用于连接数字域和模拟域，所以其端口必定含有两种不同的类型。

我们用端口的输入输出类型将连接模块分为三种：


	d2a -顾名思义，工作的类似DAC，数字域输入，模拟域输出

	a2d -工作类似ADC，模拟域输入，数字域输出

	bidir -双向工作



一个双向的连接模块的示例如下：


connectmodule bidir(out, in) ;
  inout out ;
  inout in ;
  logic out ;
  electrical in ;
endmodule




上面这个例子的连接模块什么都没做。一个可以工作的连接模块需要详细定义模拟与数字之间的信号转换。


connectmodule bidir(anax, digx) ;
  inout anax;
  inout digx;
  logic digx;
  electrical anax;
  reg tmp;

  assign digx = tmp;

  analog begin
V(anax) <+ transition( digx == 1 ? 5.0 : 0.0, 3n, 3n);
  end
       always @( cross ( V(anax) - 2.5, +1 ) )
     tmp = 1`b1; 
       always @( cross ( V(anax) - 2.5, -1 ) )
     tmp = 1`b0;
endmodule




在这个例子里，我们定义了一个内部寄存器，将数字域的端口连接到这个内部寄存器上。然后通过检测模拟域和数字域的事件，设置对应侧的信号。我们也可以设置信号跳变的门限，加入延时，加入模拟函数等等。

4.3.2 自动插入连接模块

Verilog-AMS支持在代码中设置自动插入连接模块。

首先，你需要定义connectmodule。我们已经在上面看见如何书写自己的connectmodule。

其次，在代码中，定义connectrules。connetcrules包括两部分， 一种是表示如何插入连接模块（insertiong），一种是表示连接模块以何种信号为计算基础（resolution）。其中resolution主要用于定义混合信号线连接时用到的discipline，用的比较少。

定义connectrule的语句如下：


     connectrules connectrule_name;
     connect_insertion | connect_resolution
     endconnectrules




定义connect_insertion的语句如下：


 connect module_name [param]




参数module_name就是已经定义的连接模块的名称。

param参数可以包含多种参数。它可以包含connect_mode，connectmodule的参数和端口方向。

connectmodule的参数指的是你可以在使用连接模块的时候重新设置其参数，就像使用一般的Verilog-A模块那样。例如：


connect d2a #(.delay(3u));






端口方向指的是，你可以在使用连接模块的时候重新设置其端口方向。例如：


connect d2a output logic, input electrical ;




系统在插入连接模块的时候会根据你设置的端口类型。在这个声明里，这个d2a的连接模块会被自动插入在从模拟域网线到数字域网线的连接之间。

connect_mode参数用于表示，当有多个相同类型网线连接到同一个节点上时，系统是为每个网线都分派一个单独的连接模块，或是合并相同类型的网线，为他们使用同一个连接模块。connect_mode可以使用merge或者split。split表示为每个网线单独分派连接模块；merge表示为所有同类型网线使用同一个连接模块。

图 4-1 连接模块的放置


上图摘自Cadence用户手册


上图描述了不同mode下系统插入连接模块的方式。左边表示merge模式，右边表示split模式。

4.3.3 连接模块工作原理

有了连接模块，有了自动插入连接模块的规则，系统就会自动在模拟域与数字域之间插入连接模块。

[image: 图像说明文字]


图 4-2

最简单的情况，数字域和模拟域内部信号之间的连接，系统不会插入任何连接模块，所有仿真在域内进行。

[image: 图像说明文字]


图 4-3

如果是多个数字输出连接至模拟域的输入端，则插入一个连接模块。这个连接模块作用的象一个1bitDAC，将逻辑高和低转换为Vsupply或Vgnd。

如果反过来，多个模拟输出接至一个数字的输入端，则插入一个类似1bitADC的连接模块。

[image: 图像说明文字]


图 4-4

如果一个数字的输出，连接到一个数字的输入以及一个模拟端口，则有两种情况。上图所示，n1为数字域的内部节点。由于连接到了C1，信号变成混合信号。这时候系统会在与C1的连接点上加入一个连接模块，n1依旧是数字域的内部网线。此模块将数字信号转换为模拟信号。

另一种情况是：

[image: 图像说明文字]


图 4-5

假如n1并不是数字域的内部节点，而是外部连接，则n1变为一混合信号网线。这时候系统会如图加入两个连接模块。分别进行DA转换和AD转换。

与上图还有个最大的不同，上一图中d1到d2的信号并不延迟，因为n1的计算完全不经过模拟仿真器。而下图中，由于n1节点的计算经过模拟仿真器，d1到d2的信号天然带有延迟。

如果是一个模拟输出同时驱动多个模拟和数字输入，也有类似的情况。读者可以依据上面的例子画出系统会插入连接模块的位置。





第5章 系统划分与验证

到这里，我们已经了解了绝大多数的Verilog-A语句使用方法。我们也了解了Verilog-A/AMS是如何在系统中与数字模块相互作用，进行仿真的。接下来，我们就要开始思考构建系统中会遇到的问题了。

在进行系统工作之前，设计者首先应该在脑海中对需要实现的功能，需要达到的性能，产品的应用范围，可以采用的工艺等信息有完整的理解。这个阶段的信息可能来源于产品定义的文档，市场工作人员的反馈，以往的工作经验和生产厂的工艺技术。

在这个阶段，往往对如何实现系统的功能和参数已经有了一个大概的了解。但是，很多细节设计还存在不确定性，有一些备选的方案。一些参数的实现大致上认为可行，但具体实现的代价还不明确。一些功能的实现方式需要进行验证。这时候是最需要使用Verilog-A/AMS这样的系统级别建模语言的时候。

我们第一步，需要将我们构想中那个“大概”“朦胧”的系统价格进行细分，分化为可描述的单元模块。

5.1 系统划分

5.1.1 顶层划分

现在，构思一个数模混合的系统。系统中的90%以上的数字模块是为了实现某个独立功能而设计的。例如数字通信接口，数字信号处理，存储等等，都是规模较大而且功能独立的数字模块。也就是说，在系统最高层划分的时候，绝大多数的系统都可以简单的被分为数字模块和模拟模块。

这种高层的划分是明显有好处的。

第一，高层的功能划分是最清晰的，便于理解和分析系统细节。

第二，便于分工合作。在大规模项目中，通常数字电路的设计工程师和模拟电路的设计工程师不会是同一个人。工作方式，使用的仿真环境和分析方法也很不同。在顶层进行划分可以明确各自的工作范围以及目标。

第三，模拟与数字之间的接口变得清晰，这样不但利于定义模拟数字接口，也能够在仿真验证中够获得最高的效率。假如一个信号，既经过数个数字网线，又经过数个模拟网线，就需要多个、复杂的连接模块。设计者不但需要很好的理解这些连接模块之间的相互转换是否正确，而且会浪费大量的仿真时间。

5.1.2 子模块划分

通常我们不会用一个超大的Verilog-A文件描述全部子模块的功能。子模块同样需要进行进一步的划分。

一般来说子模块的划分基于以下这些原则。

5.1.2.1 功能独立性

这是最显著的划分依据。我们划分系统的最基础的目的就是分割功能，这样我们才能分析各个功能模块的需求和实现方式。因此，系统中的比较独立的单元应该被单独分在子模块中。比如系统中使用的ADC，DAC通常我们都会很自然地把他们分在单独的模块。

5.1.2.2 分工合作

系统划分也提供了团队分工的基础。将子模块进行划分，才能将整个系统化整为零，团队成员每个人完成相对独立的模块设计。因此，在系统设计阶段就要考虑到模块划分对后续工作分派的影响。

这里需要考虑的还有后端的设计工作量。全定制的模拟设计后端，依旧依赖于人工绘制。有些模块，也许电路本身设计并不复杂，却需要大量的后端绘制时间。如果在系统划分的时候将几个这种模块合并在一起，将很不利于后端工作。

例如，通常一个ADC我们认为是一个基本模块。但有时候这个ADC本身非常关键因而需要比较复杂的设计。我们在系统划分的时候，就可以考虑将这个ADC进一步划分为，例如它内部的控制逻辑，内部的DAC，采样保持和放大电路等等独立模块。

5.1.2.3 强调关键模块

一个在系统划分时常会遇到的问题是，系统在不断划分为关键模块后，随着层次增加，无法强调关键路径和关键模块。比如用于一个信号处理的芯片，要经过几级模拟电路和数字电路的信号处理，随着系统划分碎片化，关键信号经过的模块，有些在顶层，有些被藏到了很深的位置。

这样的缺点是关键路径和关键模块很难突出。如果在设计过程中发现某些模块需要单独建模，会很难划分出来。而且在系统验证中，往往希望对关键路径进行着重验证。而层次不一，分散的系统划分很难将整个关键路径拿出来单独做仿真验证。

因此，我们希望在系统划分的时候，尽量让关键路径放在自己的一个大模块中。比如，一个系统顶层被划分为数字模块和模拟模块之后，我们在两个顶层模块内，分别划分一个模块，将系统的关键路径的模拟和数字部分放在这两个模块内。其余的模块，包含比如基准电压，振荡器，过压过流保护，温度检测等等，都可以放在别的模块里。这样，所有关键的模块都被放在同一个层次里，而且要单独拿出来验证会相当方便。

5.1.2.4 层次化

前面已经提到，过多的层次会导致很难看懂系统架构。而且过多的层次也使得问题的查找变得很困难，所有的问题都需要不断的进入到一层又一层的子单元中进行调试。大的系统有可能会出现10层以上的分层，试想一下从一个最底层单元跳到另一个最底层单元需要经过20级，光是操作可能就让人厌倦了。而且，系统设计过程中的修改不可避免，一个连接的修改就可能变成7-8层模块的变动，很繁琐。

通常我们已经有了一个顶层模块和数字模拟两个基本划分，考虑到数字和模拟的模块内部势必有子模块，我们已经有了3层分层。一般最深的层次是实际的器件，那么通常在器件之上会有另一层器件组成的小单元。最简单的分层就是这些小单元组成子模块。这样，一个简单的系统模型至少需要5层。

图 5-1 典型的层次图

某些比较复杂的模块可能会需要更多层次。也就是说，比较好的系统层次应该可以控制在5-7层。

如果不注意，可能会出现例如一个电源Power模块包含LDO，LDO又包含Bandgap，Bandgap又包含运放，运放又包含输入输出单元，输入单元又包含Trim电路......层次瞬间就会变得无法控制。

图 5-2 不好的层次划分

这个例子，合理的做法是：Bandgap里使用的运放完全可以扁平放置；Trim电路尽量放置到Bandgap这一层；Bandgap则应该和LDO并列放置到Power模块下。

图 5-3 修改后的层次划分

这样划分，层次更加简洁，而且单元功能独立性更好。

5.2 关键路径和关键模块

当我们有了系统划分之后，下一步要考虑的就是书写模型，进行仿真和验证。

我们的系统里，不会全都是挑战性的、不确定的模块。一个比较稳妥的系统，可能卖的就是一两个亮点。如果是一个开创型的项目，风险比较高，也不太会超过30%的电路是不确定的。

也就是说，在系统设计的初期，应该有70%以上的模块是我们有把握，对其实现方式和细节都比较了解的。对于这些模块，我称之为成熟模块，我们的模型也应该是比较成熟的，可能也是可以从以前的项目复用的。唯一要注意的是，要考虑到某些未描述的特性是否对其余30%的挑战型模块会造成影响。

我们的主要力量将会花在为余下的30%的挑战型模块的建模上面。这些模型可能需要大量的调整，尝试不同的结构，依据实际器件的特性进行修改。

5.2.1 整体建模

假如对于如何实现关键模块还不是很有信心，可以先对其进行整体建模。因为，通常系统的输入输出是已经确定的。需求也是确定的。内部实现方式可能有很多种。为了避免一开始就不断的在几种方案中修改，先对整个输入输出关系建立模型，避开实际实现的一些问题，有利于及早完成整体系统。

实际上，在整体模型的书写中，就可以分析何种实现方式是最适合的。举个例子，需要精确放大某个信号，又要解决小输入电流和高共模电压的问题。一开始可能这样书写：


// comments and module declare
V(outp, outn) <+ gain*V(inp, inn);
I(inp) <+ V(inp)/Rin;
I(inn)<+ V(inn)/Rin;






可能还有一些别的语句，这里省略。

这样的写法可以反映输入和输出的关系，但没有很好的显示如何实现指标。比较有指导意义的也许是，根据目前的写法，输入电流是一个和电压成正比的值，就像一个电阻。那么，需要有多大的Rin值，才能满足我们的指标，就能看出我们是否能够直接用电阻解决输入电压的问题。

最常见的方式是，我们使用电阻比例来实现增益。为了耐住高共模输入电压，我们可能需要两级，第一级提供个较小的增益来保证器件上分压不会看见很高的电压，第二级提供放大。那么，这个整体的描述可能会变成是这样写：


// comments and module declare
electrical nodeinp, nodeinn;
V(nodeinp, nodeinn) <+ V(inp, inn)*Rin1/Rfb1;
V(outp, outn) <+ V(nodeinp, nodeinn)*Rin2/Rfb2;

I(inp) <+ V(inp, vref)/(Rin1+Rfb1);
I(inn)<+ V(inn, vref)/(Rin1+Rfb1);






基于这种描述，我们可能再进一步考虑，两级电阻的匹配有多好，放大器的输入失调、速度等等需要多好，选用多大的Rin1、Rfb1和增益的设置。

假如我们认为这些指标是可以达到的，也就是说这种描述方式可以实现我们需要的性能，接下来我们就可以将整体模块细化。细化可能单独分出来的就有两级分别的运放的设计，电阻网络的设计，增益的修正电路等等模块。

5.2.2 使用参数

任何一个开发阶段电路的模型，都会被不断的修改。对于关键路径上的模块，修改的可能性更大。如果每次尝试的过程都需要修改模型，更新电路，再重新仿真是非常费事费力的。

常见的做法是在书写模型的时候尽量使用参数。参数让你可以在调试阶段很容易的修改模型。

在仿真阶段，以Caence环境为例，你还可以将模型的参数写成pPar(“para_name”)格式。这类参数会直接被传递到仿真环境中，作为仿真本身的参数。你可以在ADE界面或者仿真网表直接修改参数，无需进入电路进行手动修改。

5.2.3 分离指标

模拟电路和系统的设计的思想有一条很重要，Trade off。大致上是说，如果设计者希望提高一个指标，必定会有别的某个或者某些指标下降。

对于系统关键的指标的实现，如何选择最适合的电路或系统结构是第一步，对指标之间的权衡则是第二步。有很多时候，并不存在什么特别优秀的架构能够让你的设计远超市场上的竞争对手，那么除了期待工艺水平的提高，就只能依赖于设计者对Trade off的精确把握了。

有经验的设计者会在面对一大张系统设计需求的时候，挑选出那些最难实现的，然后针对这些比较难实现的指标和与之相关的指标，进行权衡。很多时候，为了达到最优化的参数组合，需要反复的尝试。这种尝试不单存在于真实器件的电路搭建过程中，也存在于系统级别的设计中。

如果说将一个关键参数和其他几个参数进行Trade off是考验设计者的经验和能力的话，把3-4个关键参数和其他几十个参数放在一起希望进行优化那复杂度就过高了。

因此，我们在系统设计的时候就需要将重要的参数和指标分散开。在前面讨论整体建模的时候，那个例子里的写法就是类似的思想：将一个系统分为两部分，第一级处理了高共模输入电压的指标，第二级处理了增益的指标。

类似的，如图5-4所示系统，电源电压噪声很大，如果用这种方式来划分，每个功能模块都需要实现很好的电源抑制。假如A，B，O这几个模块自身也有重要的参数，在系统设计阶段，就将这些参数都混合在一起考虑了，增加了复杂度。

图 5-4 不好的划分

也许，我们都会这样做。

图 5-5 修改后的划分

随着增加一个模块，电源抑制这个参数就基本上被独立了，其他模块在设计的时候就得到了简化。当然，这样做我们也需要考虑：参数分离出去是否能够保证整体的复杂度还在可控范围内？额外的设计时间和实际开销是否值得？是不是所有模块都能这样做，或者都需要这样做？一般来说，如果是关键的参数指标，我们对它的实现把握不大的，而且对其他参数影响又比较大的，尝试将它独立实现是可以减轻系统设计难度的。

很多时候，我们下意识的将一些东西分离出去，或是为了分离独立的功能，或是为了分离指标，达到了很好的效果。但在新的系统里，并不完全是类似的情况了，也许依旧这么做反而是破坏了系统设计。牵一发而动全身这种事情，在模拟系统设计里常常会发生，因为模拟系统很多时候就是为了几个甚至一个指标而存在的。因此，带着目的性的划分模块，是很有必要的。

5.2.4 理解混合信号仿真的限制

在前面讲Verilog-A/AMS的连接模块的时候就已经提到，在混合仿真中，需要使用类似AD、DA的连接模块。信号在两个域之间传递的时候，通过连接模块转换。连接模块通常使用类似1bit ADC、DAC的工作方式，它尽可能的提高了仿真速度，但是信号传递的过程中收到了很大的限制。在系统级别划分，建模和验证的时候，我们需要考虑到这种转换的限制。

在数字域转模拟域过程中可能会有这些限制：


忽略了电容，没有转换过程


连接模块的输出是一个理想的源，即使考虑到一些负载特性，通常也不考虑电容负载对转换过程的影响。假如在模拟域这一侧有比较大的电容，而系统又比较关心转换过程的特性的话，就无法很好的验证对应的指标。


固定的延迟


连接模块通常不设置延时。一些比较好的连接模块也可能只是加了一个transition的函数。这样设置的延迟和数字域输出的能力，以及模拟域侧的负载大小都没有联系。也就是说，如果系统关系延迟，这样的仿真验证将无法给出正确的结果。


绝对的VCC或GND


数字域给出的是绝对的0和1。连接模块通常会将之转换为VCC和GND。但有些时候，我们要考虑到数字域输出的最终值并不那么理想，特别是带一定负载，在很短的时间内切换的情况。

例如，数字域输出一个时钟，这个用于驱动Charge Pump的充电电容。很自然，我们既关心这个时钟信号带负载之后的建立情况，也关心实际输出的电压值。假如时钟是直接来自数字域的模型，我们会得到一个很理想的驱动信号从而认为没有问题，但实际上有可能是错误的。


在模拟域转数字域过程中可能会有这些限制：



翻转门限


由于数字域的理想特性，在模拟域转数字域中同样会丢失一些信息。最终实现的数字电路同样具有模拟特性。在模型的仿真中丢失的诸如翻转门限电压，上升下降速率，驱动能力等参数也可能对最终结果产生影响。

通常在一个混合信号芯片中，数字部分对最终产品的指标不起决定作用。但随着数字设计的复杂度增加，数字处理的频率提高，数字部分的工作特性也越来越接近模拟电路。例如，一个Vth为0.2伏的低压数字电路被错误触发，而在模型验证阶段如果将转换模块门限默认设为0.4V则无法发现这个问题。。


噪声


另一个会在转化中丢失的便是噪声。这里的噪声不光是指热噪声这类随机噪声，也包括例如信号耦合的噪声，脉冲电流带来的地噪声等。

所有的噪声仿真验证都只在模拟域中存在。当信号从数字域到模拟域时，由于信号基本上处于或强上拉或强下拉的饱和状态，对模拟域的噪声不会起什么影响。而且模拟域本身可以通过仿真验证诸如非理想信号，数字部分脉冲电流等的影响。但当一个模拟域的噪声信号输入至数字域时，由于噪声会在转换中丢失，就无法检测到噪声信号的影响。

5.3 子模块模型

5.3.1 模型书写的一般流程

图 5-6 模型的书写流程

通常书写一个模型会经过如上图的几个步骤。

首先，根据对模型的构思，用公式描述各个端口各个信号之间的关系。然后，有了这些公式，就可以直接在Verilog-A文件中书写这些公式，作为第一版的模型文件。第三步，由于纯数学的公式化模型常常需要转化为描述实际实现的架构，我们有时候还需要将纯计算公式的模型语句，更换为实际实现的结构化描述。最后，在模块阶段也需要经过反复的仿真验证，以证明在各种情况下模型工作的符合预计。如果模块仿真验证发现问题，可能是前面任何一个步骤出现的问题，都需要回溯回去修改。

5.3.2 简单的模型最好

我们书写模型的目的是，用模型来替代复杂的真实器件模型，从而获得早期的、快捷的仿真和验证结果。因此，单元模型书写的第一条准则就是必须要简单。

设计者必须了解什么细节是这个模型文件需要反映的，哪些是不需要关心的。比如，很多时候我们确实不需要考虑模块的功耗，因此电流尖峰等等不需要被建模。一个单纯的电压驱动模块很明显会加快仿真速度。但假如电流是系统所关心的，那么即使牺牲一部分速度我们也不能将电流的建模简化掉。

5.3.3 让模型象模拟电路

不理想的波形，延迟，信号畸变，这些都是模拟信号的特点。给出一个完全理想的信号，工作的象一个数字电路一样，往往蒙蔽了仿真验证中间能看见的很多问题，同时也大大的增加了仿真器的收敛性能。前面提到很多语句，例如transition，都是用来帮助你的模型工作的更像模拟电路的。尽可能的使用这些语句并不会增加很多设计和仿真时间，但却可以帮助你省去很多调试的时间。

5.4 电路与模型验证

5.4.1 验证类型

图 5-7 模型交叉验证

仿真和验证贯穿于整个设计阶段。仿真验证主要目的就是对比仿真波形或计算得到的指标参数，以保证不同系统设计、架构、电路实现之间可以达到某种程度的一致性。


模型对模型


在系统设计初期，绝大多数的验证是模型对模型（理想模型）的验证。设计和书写模型的时候，我们与之比对的是客户的要求、理想的时序、产品的指标。可以说我们是在用脑海中的一个理想模型的输出曲线在和我们搭建的模型的仿真结果进行验证。

另一种情况则是数字模块的验证，它的输入输出结果和时序会需要和详细定义的波形进行比对。

模型对模型的验证，通常会需要很高的误差精度。因为模型本身也具有“理想工作”的特性，实现比较高的匹配度并不会很难。这个阶段残留的容差，将无法被电路实现所弥补。


电路对电路


电路对电路的验证主要用于修改和简化现有的电路或者IP。模拟设计里，会出于对面积，功耗等的考虑而不能直接复用现有的电路设计。修改后的模块的关键指标是否依然满足要求，就需要对比验证。

电路对电路的验证是为了验证，除了某几个指标之外，修改后的设计和原设计偏差在一定范围内。电路对电路的验证可容忍的误差取决于各个模块的指标的要求。


模型对电路


模型与电路之间的验证大多用于测试实际器件搭建而成的电路是否符合系统设计的要求。有一些情况下，我们是根据某个IP进行建模（例如，为了加快仿真速度），我们需要将模型和现有电路进行对比以保证模型代入系统后工作正常。

通常这个阶段会产生一些预计之外的误差，因为实际电路受到各种参数变化的影响，诸如供电电压、温度、工艺变化等。这些误差需要被加入到系统模型中验证是否影响最终结果。假如有较大的影响则需要修改电路/模型以保证系统实现。

5.4.2 验证方式

验证主要是通过仿真，获得仿真波形，或者计算得到指标，进行对比。有些情况下，验证新设计和已有的电路、芯片，也会进行实际测量，进行对比。

数字电路的验证对比已经有了很多现成的工具，对电路的工作覆盖率也比较高。

模拟电路的验证基本上还是要依赖于设计者的经验。一些指标是通过计算仿真结果所得到的。这类指标的验证比较直观，只需要对比不同仿真情况下的结果数值即可。一些指标比较难以通过计算得到，比如波形的纹波，尖峰，噪声等，就需要肉眼观测对比。还有一些指标，例如过压、最大电流等则需要在仿真中加入特定的检测功能才能看到。

5.4.3 验证工具

验证的最基础的工具是仿真器。除了仿真器，还需要有一个能够进行结果对比，自动化的工具。

对于数值比较的验证，比较简单的方法是使用Excel这类表格软件，即可以进行一定的后期处理和计算，也可以制造图表，直接可以生成最终文档。现在，有一些在线工具，更加可以在线处理和存储，方便地和小组成员分享结果，也是很不错的。

对于波形文件的比较，大多数仿真器都有自带的看波形工具。但最大的问题是很多波形工具不支持同时加载多个仿真结果。下面介绍两个Cadence商用平台下的工具，以作参考。

5.4.3.1 SimVision与Visualization&Analysis

在Cadence6系列中可以使用被称为Visualization&Analysis的新的波形观测工具。它可以同时加载多个结果文件，自然也可以将不同的波形放在一起进行比较。

图 5-8 主界面

如上图所示，我们在Visualization&Analysis中加载了两个需要进行比对的电路的仿真结果。

图 5-9 加载多个结果

同样，在混合信号仿真中使用的SimVision工具，也可以同时加载多个结果。如上图所示。

图 5-10 对比波形

不同结果的波形可以放在一起进行比对。

SimVision也可以连接至多个正在进行中的仿真（connect to simulator），可以边仿真边进行对比。

5.4.3.2 AmsDmv

直接使用波形观测软件可以比较自由灵活的对比，但是很难做到自动化和对大批量的指标直观的做出验证是否通过的判断。在Cadence平台下，还有一个叫做AmsDmv的工具，可以对两个仿真结果进行自动的验证。

图 5-11 主界面

打开AmsDmv界面，可以选择需要对比的仿真类型。Cadence系统里的组织形式是Library →Cell→View，选择相应的仿真View即可。

图 5-12 选择需要对比的测试平台

如上图，我们选中了两个需要对比的仿真，View即是仿真设置。

图 5-13 可以对比的波形

切换到Waveform页中，已经出现可以对比的所有波形。

图 5-14 端口对比

在仿真之前，需要进行端口的对比。如果端口有不同，如上图中所示红色Failed字样。

图 5-15 出现测量结果的对比

我们忽略端口的差异，运行仿真。这时候切换到Measure Results页面，就可以看见我们在仿真中进行的测量已经被自动计算并且得到了验证结果。页面下方可以设定对比验证的容差等。

图 5-16 波形对比

这时候，在Waveform页面就可以看见我们想要比对的信号之间的差异。

5.4.4 小结

到这里，我们已经讲完了从分析系统，划分系统，构建模型，具体Verilog-A模型的语法，功能的书写，到模型的验证的整个流程。系统建模可以给我们的设计带来很多好处，也会在一开始带来很多困扰。不同的情况，也会有不同的设计方法和流程，孰优孰劣，必须要实际动手做过才会真正的有体会。

接下去的几章，会讲述一些零散的内容。这些内容很难被包括到前面的章节中，也并不是系统模型构建中总需要用到的。有一些只是作者个人实际使用的经验和感受，受能力所限，也无法保证是最好的解决方案，供读者在遇到一些困难的时候参考。





第6章 高级应用

6.1 网表和模块实例

Verilog-A/AMS的实例最终要加入到仿真的网表内才能进行仿真。通常我们的操作都是基于某个图形化的编辑界面，将实例化的模块的symbol进行连接。这样生成网表的工具就会自动将模块加入到仿真的网表内。

有一些时候，我们喜欢手动修改网表，以便进行自定义的仿真。因此我们需要了解一下Verilog-A/AMS实例在网表内的表示情况。

首先，实例化的模块在电路里的表示和一般的电路模块相同：


instance_name pin_A pin_B ... instance_type param_A param_B...




我们可以根据实例名称搜索，手动修改参数等。实例化的语法规则在前面章节已经讲了。

这里instance_type就是模块的名称。

6.1.1 切换代码文件

模块的定义文件.v或者.vams文件通过ahdl_include加入到仿真网表里。


ahdl_include “file_name.va”;






一个简单的技巧是，可以通过修改这个va文件名，在几个不同版本的Verilog-A/AMS代码之间切换对比。这样就不需要回到图形界面修改电路，再做生成网表这个步骤。

比如我们的仿真中的默认文件为：


ahdl_include “/home/swu/project/vatest-1/invx0/veriloga/verilog.va”;




我们用//符号将这一行注释，然后加入一行替代：


//ahdl_include “/home/swu/project/vatest-1/code/invx0small-verilog.va”;
ahdl_include “/home/swu/project/vatest-1/code/invx0mid-verilog.va”;
//ahdl_include “/home/swu/project/vatest-1/code/invx0large-verilog.va”;




我们希望尝试中等延时的模型。在命令行中运行仿真：


spectre input.scs ../psf/spectr.out -format psfxl -raw ../psf




命令行中运行spectre的命令只需要从以往运行仿真的log文件中拷贝稍加修改即可，一般不需要记忆。

这样可以直接用文本编辑器和命令行进行仿真，通常要比回到图形界面修改电路要便捷。

6.1.2 模型的cornor

仿真中有时候希望模型也能反映出随着工艺cornor变化的特性。通常模型文件会带有参数。我们下面演示，可以用设置不同参数的方式，在不同的工艺角（cornor）下面，设置不同的延迟时间。

在cornor仿真中，假如模型文件被认为是一个单独的cornor -- 也就是说不随着MOS管、RC的变化，这样子很简单。最直接的方法就是随着仿真修改实例化的模块的传入参数。这个步骤既可以手动完成（假如需要修改的模块不是很多的时候），也可以使用类似perl之类的语言强大的正则表达式，自动替换。

比较复杂的情况是，我们的仿真中，希望模型的cornor跟随MOS管、RC的工艺角变化。例如，我们认为模型应该在MOS为slow，RC为slow的时候有大于10n秒的延时；在MOS为fast，RC为slow的时候有5n秒左右的延时；在MOS为fast，RC也为fast的时候则只有小于1n秒的延时。在仿真中，我们就希望在设置MOS管和RC的仿真cornor的时候，模型能够自动体现这种变化。


第一种实现方式


首先，SPICE仿真器通常设置仿真cornor是通过选择SPICE模型文件的section来实现的。在section中，写入不同的参数值，这样器件的模型在不同cornor下仿真也就会有不同的表现了。这是一个典型的section例子：


section m_tntp
parameters param_A=0.019e-6
...
endsection m_tntp




这种做法本身简单而且直观，但很不幸，SPICE里的parameters不能直接传递给实例化的Verilog-A/AMS模型。原因是，前面看到，Verilog-A/AMS参数的是在实例化的时候被重新设置的。在网表中实例化的单元，可以接受仿真参数，但不能接受SPICE模型所定义的参数。而仿真器也通常不支持cornor文件修改仿真参数。

因此，实例化只能在SPICE模型文件中进行。我们可以在模型中，将实例化的模型作为某种器件的子电路。


ahdl_include “verilog.va”
subckt device_m (in, out)
ex (in, out) model_name param1=param_A
ends device_m




首先，使用ahdl_include将Verilog-A/AMS文件包含进来。然后建立单元，在单元内部实例化一个模型文件所定义的模块。这里的实例化语法使用的是SPICE语法，也即是:


实例名 （端口） 模型名 参数=数值




这样子，在不同的section中填写param_A的对应数值，就可以得到一个随着MOS管cornor变化而变化的device_m。而device_m的实际实现使用的是一个Verilog-A/AMS文件。


第二种实现方法


上面的实现方式需要修改cornor文件，还需要定义新的单元来替代Verilog-A/AMS单元。假如在设计初期没有预计到这些步骤，就需要修改系统模块图。因此，这种做法通常用于规范化的单元模块设计，比如EDA工程师提供完整的模型库。对于简单的使用cornor，还有另一个方法。

上面已经看到，SPICE的cornor文件中是可以使用ahdl_include来引入模型文件。因此，我们也可以不通过修改参数的方法来实现cornor。

例如，我们先做三个va文件，以表示三个不同的cornor下的工作状况。在cornor文件里，我们在不同的section里引用不同的va文件。


section m_tntp
ahdl_include “/home/swu/project/vatest-1/code/invx0mid-verilog.va”;
...
endsection m_tntp

section m_wnwp
ahdl_include “/home/swu/project/vatest-1/code/invx0small-verilog.va”;
...
endsection m_wnwp

section m_snsp
ahdl_include “/home/swu/project/vatest-1/code/invx0large-verilog.va”;
...
endsection m_snsp




为了避免重复定义，我们要在仿真网表里将那里对应的ahdl_include语句删除。

这样，选择不同的MOS管cornor的时候，也选择了不同的模型文件。

这种做法的缺点是不够灵活，增加一个cornor就需要增加一个va文件，而且无法对应多种器件的cornor组合。但考虑到实际仿真中间，对模型文件并不会象器件模型那样，遍历所有的cornor组合，因此这种做法还是比较实用的。

6.2 用Verilog-A/AMS做验证平台

现代的模拟设计环境都包括图形化的电路绘制界面。你在里面绘制电路，连线，然后工具根据你绘制的电路图为你生成网表，进行仿真。

有些时候，我们也更愿意使用文本和命令行形式进行仿真。特别是在项目结尾阶段，大量的大规模的仿真，会进入仿真服务器进行排队，或者在开发者自己的工作站上彻夜排队仿真。这种时候，使用命令行形式进行排队，要比在图形界面下一个一个点击要快的多。

如果有多个测试平台，要在图形界面下，为每个测试电路生成网表，然后书写仿真排队命令。如果有测试平台的电路需要修改，则需要回到图形界面修改，重新生成仿真网表。如果对仿真器的仿真网表格式比较熟悉，也可以手动修改网表而不需要回到图形界面。

修改仿真网表的主要问题是仿真网表是一个很长的文件。比如在Cadence里，生成的input.scs可能是这样子的：


simulator lang=spectre
global 0 GSUB!
parameters v1=0 ......
include "model_rc.scs"
.......

subckt BlcokTop (pinlist......)
all devices .......
ends BlockTop

subckt BlcokA (pinlist......)
all devices .......
ends BlockA

subckt BlcokB (pinlist......)
all devices .......
ends BlockB
Many blocks ......
simulatorOptions options reltol=1e-3 vabstol=1e-6 iabstol=1e-12 temp=27.0 \
some simulation options......
......
tran tran stop=150u......\
more simulation setting......





可以看到，顶层仿真网表需要列举所有的模块和器件、它们之间的连接关系、仿真的设定和参数等等。一个顶层的仿真网表文件大小几十M，长度几十万行是很正常的。

在这几十万行里找到需要的模块，修改其连接并不容易，而且容易出错。比如手动增加一个电容C1，首先你得检查整个网表是否已经使用了这个名称。我遇到过一个错误就是手动增加的一个电容命名为C1，和一根时钟线名称C1重名，最后出现古怪的结果。

一个替代的方法是，可以将测试平台的顶层连接信息放在一个Verilog-A模块中。

图 7-1 测试平台结构

如图所示，整个测试平台顶层只包含由Verilog-A/AMS书写的测试电路。顶层单元“Ciruit Top”依然会显示在仿真网表内。但其连接关系会在testbench.va中。

仿真网表依然会是个很长的文件，最后会增加类似这样一行：


ahdl_include "/path/veriloga.va"






而对应的Verilog-A文件可能会是这个样子：


module module_name ()
CircuitTop I1(net_vdd, net2, net3......);
vsource #(.dc(5)) I2(net_vdd, gnd!);
vsource #(.type(pwl), .dc(0), .file(“input-signal.txt”)) I3(net2, net3);
......
end module





这个模块不需要任何输入也没有输出。在这个模块内部实例化顶层电路，并且加入了理想源作为外部的偏置和激励。书写实例化语句会花一些时间，但是后续的修改会相对容易。对于连接关系，由于代码数量小，出错的概率也会大大降低。

一旦这个测试平台运行，以后任何修改都只需要修改这个Verilog-A文件，而不需要回到图形界面重新生成网表。而且测试平台代码在不同项目之间都往往是可以复用的。

使用这个方式需要注意的一点是，你可能要修改你的测量语句。原本测量的I0.netA，需要被修改为Itop.I1.netA(假设这个va模块被实例化为Itop)。

使用Verilog-A/AMS模块作为顶层仿真单元需要一些额外的工作。如果顶层验证比较多，而且采用命令行形式比较多，还是值得一试的。对习惯图形界面操作的设计师来说，可能是个噩梦。但对习惯文本模式的设计师来说，这种模式会更加简便。

6.3 瞬态，S域与Z域

在Verilog-A/AMS既可以使用瞬态描述系统的频率响应，即使用ddt等函数表示微分积分行为；也可以使用S域和Z域描述传递函数。我们下面来演示一下频域函数的使用，以及时域频域描述的差别。

6.3.1 瞬态模型和S域函数模型

为了进行比较，我们设计这样一个滤波器电路：

图 7-2 滤波器电路图

根据计算可以知道，它的等效传递函数为：

根据这两个式子，我们分别书写两个模型。第一个使用标准器件，搭建成图表一的样子。


`include "constants.vams"
`include "disciplines.vams"

module filter1(in,out,gnd);

input in;
electrical in;

output out;
electrical out;

inout gnd;
electrical gnd;

cap #(.cp(11.25p)) cap1 (out, gnd);
inductor #(.l(22.51u)) ind1 (in, out);
resistor #(.r(1K)) res2 (out, gnd);

endmodule






第二个模型则使用lplace函数直接表示频率响应。


`include "constants.vams"
`include "disciplines.vams"

module filter2(in,out,gnd);

input in;
electrical in;

output out;
electrical out;

inout gnd;
electrical gnd;

analog begin
V(out, gnd) <+ laplace_nd(V(in, gnd), {3.948e15,0,0}, {3.948e15,8.886e7,1});
end
endmodule






搭建仿真平台如图:

图 7-3 进行对比仿真

首先我们看AC仿真的结果：

图 7-4 AC仿真结果

可以看到，两个模块的输出的AC曲线是完全相同的。这也证实了我们进行比对的模型在频率响应上是一样的。

接下来，我们对这两个模型进行阶跃响应的瞬态仿真对比。

[image: 图像说明文字]


图 7-5 瞬态阶跃响应

两者的瞬态阶跃响应也是完全一致的（图下方两条曲线完全重合）。如果再具体看会发现一点差别，用瞬态器件搭建的模型会消耗输入电流，而使用S域函数搭建的模型则不会。

接下来，我们将上面的语句


inductor #(.l(22.51u)) ind1 (in, out);




修改为


inductor #(.l(22.51u), .ic(0.001)) ind1 (in, out);

图 7-6 初始值对比

可以看见输出波形的变化。注意初始值和稳定的过程。

对比两个例子的代码，使用S域函数描述模型的频率响应是很便捷的。我们可以认为对于线性系统响应，它完全可以替代瞬态模型的书写。但即使只从书写方式考虑也可以看出，S域函数只描述了小信号的响应，不可能包含DC工作点，大信号的非线性响应和初始状态。当描述的模块既有复杂的频率响应，又有非线性的输入输出特性的时候，我们通常可以在输入输出使用瞬态描述非线性特性，而在内部使用S域函数，简洁的描述小信号频率响应。

6.3.2 Z域函数的使用

我们用一个理想的积分器模型来演示Z域函数使用的例子。


`include "constants.vams"
`include "disciplines.vams"

module filterz(in,out,gnd);

input in;
electrical in;

output out;
electrical out;

inout gnd;
electrical gnd;

analog begin
V(out, gnd) <+ zi_nd(V(in, gnd), {0,1}, {1,-1}, 50u);
end
endmodule






可以看到，我们使用zi_nd函数实现了一个传递函数为[image: 图像说明文字]
 的理想积分器。[image: 图像说明文字]
 对应于连续域的[image: 图像说明文字]
 。采样频率使用了50u。

这个模型对正弦波输入的响应如下图。

图 7-7 正弦波输出

可以看到50uS周期的采样波形。和模拟积分器的差别在于，积分峰值大小不但和输入信号有关，还和半周期的采样次数有关系。读者可以尝试将采样周期增加，得到的峰值会大致增加。这也符合采样-积分器的特点。

图 7-8 细节

放大波形可以看到，积分使用的是采样尖峰的值。

Z域函数模型可以很好的描述对连续信号的采样保持这类电路的传递函数。用于ADC、DAC和一些开关电源设计中。这也是Z域函数最常见的使用方法。

于Z域函数同样不能包含大信号特性，往往我们同样会在Z域函数的前端加上一些描述输入的大信号特性的描述。这时候，假如Z域函数看见的输入信号已经是经过时钟的，会有一些不同。

在测试中，我们使用相同的Z域模块，而将输入变为跟随时钟的正弦波。

图 7-9 间隔信号的采样

可以看到，输入信号已经变成了随着时钟信号变化的正弦波。输出信号基本上没有变化。

这里，事实上在时钟信号上增加了一个1uS的延迟，也就是说Z域参数采样的时间点和信号时钟是不对齐的。我们尝试一下去掉这个延时。

图 7-10 延时造成的误差

橘黄色输出曲线是新的输出结果。对比以前的仿真结果，可以看到积分峰值变小了。我们放大输入波形进行比较。

图 7-11 延时导致边沿未对齐

可以看到，由于上升下降边沿严格对齐Z域函数的采样边沿，采样到的值是正在变化中的斜坡值，导致采样到的信号会比理想的信号小。时间点的误差取决于仿真器的时间步长，因此这种情况下采样到0或者峰都是可能的。

因此，对离散信号使用Z域函数会遇到问题就是Z域函数的采样时钟应该对应到输入信号的稳定周期而不能进行时钟对齐。这种情况类似于数字设计中，上升边沿采样信号就会在下降边沿改变信号，反之亦然。

对于一个设计好的模块，这需要根据外部时钟修改Z域函数的延时值。Z域函数的延时值只能通过参数来修改（而不能依赖于某个输入信号值），在时钟信号比较复杂的电路，或者使用了较多Z域函数的模块，特别需要注意这个问题。

6.4 随机数

6.4.1 瞬态仿真中的随机性

我们使用标准的Verilog-A/AMS函数书写一个产生随机输出电压的模块。


`include "constants.vams"
`include "disciplines.vams"

module random(out,gnd);

output out;
electrical out;

inout gnd;
electrical gnd;

integer seed;
real vos_number;

analog begin
@(initial_step) begin
      seed = 123;
     end

@(timer(0, 10u))
      vos_number = rdist_normal(seed, 0, 1);

V(out, gnd) <+ vos_number;

end
endmodule






这个模型每隔10uS生成一个新的随机数。

但是，在仿真中就可以发现，不管多少个实例，不管运行几次，输出波形都是同样的。

图 7-12 仿真结果

这里10条曲线重合了，看上去像是只有一条曲线。

这并不符合我们的希望。我们希望的是，每次仿真出现的随机值都是不同的。最常见的思路就是使用时间值作为产生随机数的种子，这样就可以保证每次运行，随着时间不同，产生的随机值也是不同的。但是很不幸，Verilog-A/AMS语言内部无法得到真实时间。

一种解决方式是，在Linux和Unix系统下，有两个设备文件/dev/random和/dev/urandom。进程可以读取这两个文件，获得一个随机值。在Verilog-A/AMS中，我们可以用$fopen系统函数打开并读取其中的随机值，使用系统生成的随机值作为种子，就可以得到很好的随机性。


Windows平台似乎无法实现此功能。不过主要商业仿真器都运行于Linux平台，因此影响并不大。


以下是修改后的模型代码。


`include "constants.vams"
`include "disciplines.vams"

module random(out,gnd);

output out;
electrical out;

inout gnd;
electrical gnd;

integer seed;
integer fhandle;
real vos_number;

analog begin
@(initial_step) begin
      fhandle = $fopen(“/dev/random”, “r”);
      $fscanf(fhandle, “%c”, seed);
     end

@(timer(0, 10u))
      vos_number = rdist_normal(seed, 0, 1);

V(out, gnd) <+ vos_number;

end
endmodule






我们在初始化阶段打开文件，并读取一个随机数种子，接下来使用这个种子值生成随机数分布。在读取的时候，我们使用了一个技巧：由于random文件内是随机的二进制数，无法使用%d或者%g这类格式符去匹配数字，我们只能使用%c去匹配一个字符，接着将这个字符存入一个整数变量，这个整形变量的值将会是这个字符的ASC值。从语法要求上这是一个错误，所以在仿真中会出现一个告警，但这不影响代码执行。

我们不建议每次都使用系统函数重新获得随机数种子，因为系统函数的调用会消耗比较多的资源，拖慢仿真。设备文件/dev/random本身是一个阻塞型的文件，当很多别的程序在使用这个随机值的时候，读取可能被阻塞而导致仿真一直在等待读取文件。替代的方法是可以使用/dev/urandom。

仿真结果如下：

[image: 图像说明文字]


图 7-13 新的仿真结果

可以看到，多次仿真，每次出现的随机值都是不同的。

6.4.2 Monte Carlo 仿真中的随机数

上面这种方法实现了在任意次仿真中都能生成不同的随机数。在Cadence环境中，系统提供了一个可以在Monte Carlo仿真中生成真正随机种子的函数。


$cds_get_mc_trial_number();




在运行Monte Carlo仿真的时候，每次仿真都会生成一个随机数，系统将这个随机数代入带有随机分布的模型中，得到带有一定随机分布的仿真结果。使用这个函数，我们的Verilog-A/AMS模型也可以使用这个随机数。

代码：


`include "constants.vams"
`include "disciplines.vams"

module random(out, seedout, gnd);

output out;
electrical out;

output seedout;
electrical seedout;

inout gnd;
electrical gnd;

integer seed;
real vos_number;

analog begin
@(initial_step) begin
seed = $cds_get_mc_trial_number();
     end

@(timer(0, 10u))
      vos_number = rdist_normal(seed, 0, 1);

V(out, gnd) <+ vos_number;
V(seedout, gnd) <+ seed;
end
endmodule




在这里我们使用$cds_get_mc_trial_number()函数得到每次运行Monte Carlo仿真的随机种子。我们还增加了一个输出seedout。我们用这个输出电压来演示我们从仿真系统得到的随机值seed。目的是为了演示$cds_get_mc_trial_number()这个函数输出的分布。

仿真结果如下：

[image: 图像说明文字]


图 7-14 在MC仿真中的随机结果

可以看到，上面的代码在Monte Carlo仿真中，输出电压有很好的随机性。而$cds_get_mc_trial_number()函数输出的随机数是均匀分布，也即是类似白噪声。

需要注意的是，这个函数只在运行Monte Carlo仿真的时候才会返回不为0的值，因此如果希望写一个在普通瞬态仿真和Monte Carlo仿真中都会产生随机效果的模型，你需要结合上面那段代码。当然，你很可能还需要增加一个参数，以保证在不需要随机性的时候关闭这些函数，读者可以自己尝试。

6.5 如何使用调试与输出

Verilog-A/AMS语言本身比较简单，语法上的问题并不多。绝大多数问题都是在仿真中间出现的。仿真中出现的问题很难进行调试。而语言本身提供的调试能力基本上可以说就只有打印语句。这里通过几个例子来展示一下对Verilog-A/AMS代码的调试手段。

6.5.1 查看参数

最常见的调试过程就是，通过尝试模型的参数，得到所需要的输出，从而了解系统对模块的性能和功能要求。这种事情，我们通常会在initial_step的时候，输出模块所使用的参数和一些内部变量。

这样做可能带来两个不好的效果：

第一，每个模块都输出模型参数的话，仿真日志文件会变得混乱而无法查看。

第二，如果一个模块被多次使用，而且使用的地方所采用的参数又不相同的话，需要能够区分模块实例的方式。

为了避免问题二，我们可以在输出调试信息的语句里使用参数%m。这个参数可以直接给出实例化的模块的层次名称。代码如下：


`include "constants.vams"
`include "disciplines.vams"

module Fdebug1(out, gnd);

output out;
electrical out;

input gnd;
electrical gnd;

parameter integer DEBUG = 0;
parameter integer par1 = 100;
parameter real par2 = 20.1;

analog begin
@(initial_step) begin
$strobe("Instance is %m, par1 = %d, par2 = %g", par1, par2);
end
V(out) <+ 1;
end
endmodule






为了展示如何输出实例名和层次名，我们另外制作了一个模块F_container。F_container包含了两个Fdebug单元。

下面是仿真输出：

[image: 图像说明文字]


图 7-15 仿真日志

可以看到，在F_container里的单元，其仿真输出了完整的层次名称。

6.5.2 关闭调试信息

只需要将前面的代码做一个修改：


`include "constants.vams"
`include "disciplines.vams"

module Fdebug1(out, gnd);

output out;
electrical out;

input gnd;
electrical gnd;

parameter integer DEBUG = 0;
parameter integer par1 = 100;
parameter real par2 = 20.1;

analog begin
@(initial_step) begin
      if(DEBUG > 0)
$strobe("Instance is %m, par1 = %d, par2 = %g", par1, par2);
end
V(out) <+ 1;
end
endmodule






我们将F_container中的一个模块的DEBUG参数设置为1，仿真日志如下：

[image: 图像说明文字]


图 7-16 仿真日志

这段代码，采用了一个额外的参数DEBUG。通过判断这个参数是否被设置为1，决定是否使用打印语句。这样，对于不需要输出调试信息的模块，只需要在其实例化的时候将DEBUG参数设置为0，就可以避免打印大量的调试信息。

而且，在仿真中，还可以将实例化时的DEBUG参数值，也作为一个仿真参数传递到仿真文件中。这样，即使有几个模块需要调试，也不需要修改仿真网表，只需要修改仿真参数就可以了。

6.5.3 运行时输出信息

如果问题出现在代码运行中，就希望随着仿真输出调试信息。也许你会这么做：


analog begin
.......
V(out) <+ xout;
$strobe("Instance is %m, xout = %d”, xout);
end




这样的输出语句会在每个收敛的时间点上输出一次。你无法将输出的调试信息和仿真点进行对应。

这样修改会好一点：


analog begin
.......
V(out) <+ xout;
$strobe("Sim Time is %g, Instance is %m, xout = %d”, $abstime, xout);
end




有了仿真时间，你可以知道输出的信息对应的是哪个时间点。但是，这样还是有太多的输出了。

通常我们有可能希望根据以下情况输出调试信息：


	间隔一定时间

	某个信号超过某个阈值

	某个信号的变化超过某个阈值



对于希望间隔一定时间输出调试信息，如果是希望跟随时钟信号周期性的输出，可以使用cross事件，在时钟边沿输出信息。如果只是大致的一定时间刷新一次，也可以使用timer事件。

代码：


`include "constants.vams"
`include "disciplines.vams"

module Fdebug3(out, gnd);

output out;
electrical out;

input gnd;
electrical gnd;

parameter integer DEBUG = 0;
parameter integer par1 = 100;
parameter real par2 = 20.1;

integer seed;
real vos_random;

analog begin
@(initial_step) begin
seed=par1;
vos_random = 0.0;
end
@(timer(0, 10u)) begin
vos_random = $rdist_normal(seed, 0, 1);
if(DEBUG > 0)
  $strobe("Instance is %m, time=%g, par1 = %d, par2 = %g, output = %g", $abstime, par1, par2, vos_random);
end

V(out) <+ vos_random;
end
endmodule






仿真日志（部分）如下：


tran: time = 1.55 us  (3.1 %), step = 800 ns(1.6 %)
tran: time = 4.55 us  (9.1 %), step = 1 us(2 %)
tran: time = 6.55 us (13.1 %), step = 1 us(2 %)
tran: time = 9.275 us(18.6 %), step = 725 ns   (1.45 %)
Instance is I0, time=1e-05, par1 = 100, par2 = 20.1, output = -0.727366
tran: time = 12.02 us  (24 %), step = 1 us(2 %)
tran: time = 14.02 us  (28 %), step = 1 us(2 %)
tran: time = 17.02 us  (34 %), step = 1 us(2 %)
tran: time = 19.02 us  (38 %), step = 1 us(2 %)
Instance is I0, time=2e-05, par1 = 100, par2 = 20.1, output = 0.73992
tran: time = 22.11 us(44.2 %), step = 1 us(2 %)
tran: time = 24.11 us(48.2 %), step = 1 us(2 %)
tran: time = 27.11 us(54.2 %), step = 1 us(2 %)
tran: time = 29.11 us(58.2 %), step = 1 us(2 %)
Instance is I0, time=3e-05, par1 = 100, par2 = 20.1, output = -0.373864
tran: time = 31.25 us(62.5 %), step = 612.4 ns (1.22 %)
tran: time = 34.25 us(68.5 %), step = 1 us(2 %)
tran: time = 36.25 us(72.5 %), step = 1 us(2 %)
tran: time = 39.25 us(78.5 %), step = 1 us(2 %)
Instance is I0, time=4e-05, par1 = 100, par2 = 20.1, output = -0.136814
tran: time = 41.52 us  (83 %), step = 725.3 ns (1.45 %)
tran: time = 44.52 us  (89 %), step = 1 us(2 %)
tran: time = 46.52 us  (93 %), step = 1 us(2 %)
tran: time = 49.26 us(98.5 %), step = 741.5 ns (1.48 %)
Instance is I0, time=5e-05, par1 = 100, par2 = 20.1, output = 0.999913




这样，每间隔10uS，当重新生成随机输出的时候，输出的日志信息显示了内部参数的情况。

如果希望检测某个信号或者变量超过某个值，可以使用cross模拟事件。如果是检测变量，可以在模块内部增加一个内部节点，将变量值赋给此节点，然后使用corss事件。

如果希望检测某个信号或者变量的变化超过某个值，有两种情况。

第一种是，每间隔一定时间的变化量，也就是一定时间内的平均斜率不超过某个值。这种情况下，依然可以使用timer事件，用两个变量，一个保存前一次的值，另一个保存现在的值，进行比较。

另一种情况是，希望信号变化超过一定值，就马上输出信息。这种时候可以使用absdelta事件。

示例：


`timescale 1us/100ps

`include "constants.vams"
`include "disciplines.vams"

module Fdebug4 ( in, out, gnd );

input in;
electrical in;

output out;
electrical out;

input gnd;
electrical gnd;

parameter real DEBUG = 0 ;
parameter real par1 = 100 ;
parameter real par2 = 20.1 ;

real old_value;
real current_value;

analog begin
@(initial_step) begin
old_value = 0;
current_value = 0;
end
@(timer(0, 10u)) begin
current_value = V(in);
if(current_value - old_value > 0.1) begin
if(DEBUG > 0)
      $strobe("TIMER EVENT! Instance is %m, time=%g, output = %g", $abstime, current_value);
end
old_value = current_value;
end

V(out) <+ current_value;
end

always@(absdelta(V(in), 0.1, 1n, 1m)) begin
$strobe("ABSDELTA EVENT! Instance is %m, time=%g, output = %g", $abstime, V(in));
end

endmodule






得到的仿真日志文件（部分）：


ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=0, output = 0
tran: time = 1.55 us  (3.1 %), step = 800 ns(1.6 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=1.59437e-06, output = 0
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=3.20611e-06, output = 0
tran: time = 4.55 us  (9.1 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=4.85147e-06, output = 0
tran: time = 6.55 us (13.1 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=6.55e-06, output = 0
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=8.33685e-06, output = 0
tran: time = 9.275 us(18.6 %), step = 725 ns   (1.45 %)
TIMER EVENT! Instance is AMS_test_debug.I0._cds_internal_Fdebug4_, time=1e-05,
output = 0.587785
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=1.02424e-05, output = 0.587785
tran: time = 12.03 us(24.1 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=1.23482e-05, output = 0.587785
tran: time = 14.03 us(28.1 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=1.47673e-05, output = 0.587785
tran: time = 17.03 us(34.1 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=1.78331e-05, output = 0.587785
tran: time = 19.03 us(38.1 %), step = 1 us(2 %)
TIMER EVENT! Instance is AMS_test_debug.I0._cds_internal_Fdebug4_, time=2e-05,
output = 0.951057
tran: time = 21.43 us(42.9 %), step = 687.1 ns (1.37 %)
tran: time = 24.43 us(48.9 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=2.54292e-05, output = 0.951057
tran: time = 26.43 us(52.9 %), step = 1 us(2 %)
tran: time = 29.21 us(58.4 %), step = 785.4 ns (1.57 %)
tran: time = 32 us (64 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=3.2182e-05, output = 0.951057
tran: time = 34 us (68 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=3.52436e-05, output = 0.951057
tran: time = 37 us (74 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=3.76603e-05, output = 0.951057
tran: time = 39 us (78 %), step = 1 us(2 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=3.97643e-05, output = 0.771855
tran: time = 41.49 us  (83 %), step = 721 ns   (1.44 %)
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=4.16707e-05, output = 0.587785
ABSDELTA EVENT! Instance is AMS_test_debug.I0, time=4.34564e-05, output = 0.587785
tran: time = 44.49 us  (89 %), step = 1 us(2 %)






仿真中，输入信号是一个幅度为1V, 频率为10KHz的正弦波输入。

可以看到，由于输入信号变化率大，absdelta事件远远多于timer事件。同时，我们没有比较反向的（减小的）的值，因此在反半轴没有发生timer事件。而absdelta依然输出。

6.5.4 输出信息至文件

当输出调试信息数量很大，或者这些数据需要被其他程序处理，就可以将调试信息输出至文件。

输出至文件的函数在前面已经提及了，使用上也没有太多变化。这里值得一提的是，一种很便利的输出文件是.csv文件格式。它可以被绝大多数的办公表格软件打开，很多matlab程序也支持直接处理其格式，因此如果是为了后续处理，输出成.csv格式文件是比较便利的。

那么.csv文件的格式是什么样的呢？


	
每条记录占一行



	
以逗号为分隔符



	
逗号前后的空格会被忽略



	
字段中包含有逗号，该字段必须用双引号括起来



	
字段中包含有换行符，该字段必须用双引号括起来



	
字段前后包含有空格，该字段必须用双引号括起来



	
字段中的双引号用两个双引号表示



	
字段中如果有双引号，该字段必须用双引号括起来



	
第一条记录，可以是字段名





这9条里，如果输出结果是为了后续处理，不会含有特殊字符，因此其实3-8条都可以被忽略。

下面是一个输出为.csv格式文件的示例：


`include "constants.vams"
`include "disciplines.vams"

module Fdebug5(in, out, gnd);

input in;
electrical in;

output out;
electrical out;

input gnd;
electrical gnd;

parameter integer DEBUG = 0;
parameter integer par1 = 100;
parameter real par2 = 20.1;

integer fhandle;

analog begin
@(initial_step) begin
if(DEBUG > 0) begin
fhandle = $fopen("/home/swu/debug.csv","w");
$fstrobe(fhandle, "This_is_a_debug_data_table");
end
end
@(timer(0, 10u)) begin
if(DEBUG > 0)
  $fstrobe(fhandle, "%m, %g, %g", $abstime, V(in));
end

V(out) <+ 0;
@(final_step) begin
if(DEBUG > 0)
$fclose(fhandle);
end
end
endmodule






比较方便的是$fstrobe函数已经自动加入了回车符。如果希望自己控制，使得每一行的长度能够不同，可以使用$fwrite函数。

输出的文件内容如下：


This is a debug data table
AMS_test_debug.I0._cds_internal_Fdebug5_, 0, 0
AMS_test_debug.I0._cds_internal_Fdebug5_, 1e-05, 0.587785
AMS_test_debug.I0._cds_internal_Fdebug5_, 2e-05, 0.951057
AMS_test_debug.I0._cds_internal_Fdebug5_, 3e-05, 0.951057
AMS_test_debug.I0._cds_internal_Fdebug5_, 4e-05, 0.587785
AMS_test_debug.I0._cds_internal_Fdebug5_, 5e-05, 2.787e-15




可以直接在excel等表格软件内处理。

6.6 收敛与稳定性问题

大概每个人都在仿真中遇到过不收敛和不稳定的问题。也可以说，没有哪一种方法能够保证你的模型和电路不出现不收敛或者不稳定的现象。这里，举几个例子，当这些烦人的情况发生的时候，我们可以如何去分析和修改模型，以帮助仿真器。

6.6.1 排除错误

大多数情况下，我们会尝试修改仿真器的参数、修改模块参数、尝试别的理想单元、简化结构等等，也许最终会沮丧的觉得，这东西就是莫名其妙地难收敛。但也许，问题只是你的代码压根就写错了！

出现问题后的第一步就是要审视自己的代码，特别是和出现问题有关节点的代码。有很多时候代码虽然是错误的，却不违反语法规则。可以进行仿真，却无法给出正确的结果。


if(V(mode[1]) > 1) begin
  V(out) <+ V(VREF, VSS)*(code&16)>>4;
end
else if(V(mode[2]) > 1) begin
  V(out) <+ V(VREF, VSS)*(code&8)>>3;
end
else if(V(mode[3]) > 1) begin
  V(out) <+ V(VREF, VSS)*(code&4)>>2;
end
else if(V(mode[4]) > 1) begin
  V(out) <+ V(VREF, VSS)*(code&2)>>1;
end
else if(V(mode[5]) > 1) begin
  V(out) <+ V(VREF, VSS)*(code&1);
end
else
  V(sdmode) <+ V(VDD);
  I(out) <+ 0;






上面这段代码看起来很合理，也完全可以执行，但却是一系列错误的大集合。

第一，判断条件从mode[1]开始，移位数量却从4开始。很多时候，这就是一个信号线net[4:0]和net[0:4]定义次序的问题。总线高低位定义的混乱，往往会在系统中带来一个莫名的错误。而且由于往往信号出现在循环和判断分支内，不容易被发现。

第二，移位计算和乘法的先后次序问题。连续书写的计算符会有计算次序问题，有时候次序不会产生问题，例如A*B/C，如何计算结果都是相同的；但也有时候，特别是会发生real变量和integer变量转换的计算，则容易发生错误。

第三，最后的else忘记了begin-end对。忘记begin-end对笔者来说是个经常发生的错误。糟糕的是，这种错误的书写完全符合语法，而且会在仿真中产生很难捕捉的错误。例如这个例子，前面设定了输出电压，最后却变成设定了输出电流为0，也就是输出浮空。浮空的输出会在系统中有任意值，也许很多时候它的值看上去都会是正确的！

第四，如果说前面的错误还比较容易看出来，那结构的问题就需要仔细思考了。这个模块，输出由变量code和输入的mode总线信号控制。但假如，有多个mode控制输入都为高，则输出会变成多个电压的叠加，也许是这样：


V(out) <+ V(VREF, VSS)*(code&16)>>4;
V(out) <+ V(VREF, VSS)*(code&2)>>1;




等效于


V(out) <+ V(VREF, VSS)*((code&16)>>4 + (code&2)>>1);




这种结果对仿真器来说完全正确。问题是，这是否是设计者的本意？设计者是希望利用模拟变量输出叠加的特性，还是忘记了这个特性而导致错误的输出？

另一种会出现的代码本身的问题是自己形成了循环。


module
analog begin
@(cross(V(outA)-vrefA))) begin
     xoutb = ! xoutb;
end

@(cross(V(outB)-vrefB))) begin
     xouta = ! xouta;
end

V(outA) <+ xouta*V(VCC, VSS);
V(outB) <+ xoutb*V(VCC, VSS);

end
end module




这两个信号互相检测，导致系统在每个time step震荡。当然这个例子过于明显了，更隐蔽的例子可能是发生在几个信号之间，可能是在数字信号的事件和模拟信号事件之间。当你发现仿真器步长变得非常小，而且输出（如果你能实时观测的话）锯齿状波形，你可以分析一下是否存在这种类型的错误。

Verilog-A/AMS的强类型结构和简单的语法使得代码一般都能轻松通过编译器的检查。但大多数真正的计算工作留给了仿真器。现代强大的仿真器也许会为你错误但合法的代码猜测到一个结果，也许直接告诉你无法收敛。假如你不仔细审视你的代码，你会花上几倍的时间反过来猜测你的系统哪里出了问题。

6.6.2 初始态收敛问题

大多数仿真软件都会在仿真伊始进行初始状态的计算。很多仿真不收敛问题都会出现在这里。初始态不收敛的最大困难是没有任何输出和数据可以作为参考。

解决初始状态不收敛的问题最直接的方法就是在模型中使用initial_step事件。在initial_step中初始化内部节点、内部变量和输出值。模块设定的初始状态的原则是，根据估算，找到一个整个系统达的稳定状态，设定的初始值能够使得系统进入这个稳定状态。更进一步的估计是，这个状态应该处于系统的连续非突变工作点上。


module A
...
paramter integer initout = 1;
analog begin
  @(initial_step) begin
      xout = initout ;
  end
...
end
end module

module B
...
analog begin
  @(initial_step) begin
      xoutB = V(VDD, VSS)/(V(ip, in) - 1);
  end
end
end module






有时候几个模块的初始状态和后续计算如果不能互相匹配，会导致初始状态不收敛的问题。例如两个互相连接的模块，两个模块都初始设定输出为0，而第二个模块的执行代码如果认为输入为0会产生无穷大的输出，常常会在初始状态计算中得到诸如“step larger than xxx”之类的错误。这实际就是选择了一个突变的系统工作点。仿真器在计算初始化状态的时候同样会经过反复迭代的过程，也就是在你设定的工作点周围进行计算，假如一个微小的偏离就会极大的误差，仿真器不会认为这是一个允许的工作点。

另一些导致初始点不收敛的问题也可能是来源于系统错误。还记得前面使用系统文件/dev/ramdon么？这里就有个例子，不合适的书写方法会导致仿真器报告一个不收敛错误。


@(initial_step) begin
      fhandle = $fopen(“/dev/urandom”, “r”);
      $fscanf(fhandle, “%d”, seed);
     end




在作者的机器上，如果使用整数去获得系统的一个随机数，会有两种情况发生问题。第一个情况是，假如使用的是/dev/random文件，当系统随机值不足而无法给出一个随机量的时候，仿真器会认为初始化失败。第二种情况是，如果使用/dev/urandom文件，但尝试用%d或者%g去获取值的时候，由于/dev/urandom文件是个非阻塞的文件，一直无法匹配到数值的话，也会导致仿真器报告初始化失败。

这种情况下，仿真器无法告知用户是什么地方发生了错误。在Cadence环境下它只是简单报告初始化计算失败。因此解决这种问题基本只能靠审视代码。

很多仿真器也会输出一些有帮助的信息。例如spectre仿真器就可以打开diagnose选项，当计算出现异样多次迭代的时候，他就会告诉用户主要的迭代步骤和特别大的迭代差异发生在哪些点上。

6.6.3 上电或不上电

一般，如果仿真开始直接设定电源电压（也就是没有上电过程），有些电路会比较难收敛。我们会设定一个电源电压逐渐上升的过程，这样整个系统会从0状态进行初始化计算，逐渐进入正常工作状态。很多时候这样会容易收敛很多。

但对于自己书写的模型，有时候并非如此。

原因是，一般仿真电路使用的器件模型（诸如BSP，PSP等），都是多年积累的完善而且标准化的模型。这种模型很复杂，但是它对各个工作点都会有完备的描述。也就是说，当系统直接“跳入”某个状态的时候，由于其复杂性，可能会导致收敛问题；但由于0状态是最简单的一个状态（可以认为这时候所有节点都应该为0），由0状态上电直至正常工作状态，每个点它都会有描述，因此能够很迅速的进入收敛状态。

而我们自己书写的Verilog-A/AMS模型，很多时候根本没有考虑这么多状态，或者在不同的状态之间容易产生不连续。这时候，0状态对于简单的Verilog-A/AMS模型反而会更难计算。

假如你在为一个IP书写模型，合理的描述上电次序是必须的。你的客户几乎100%会运行上电这个仿真。你的模型需要考虑的东西会多一些，合理的不合理的输入、正确的和错误的控制都在考虑之列。

假如你为自己的系统做一个模型，目的只是模拟一下正常情况下的工作或者参数，省掉在模型中考虑诸多异常情况也是合理的。这种时候，你会自己知道，这个模型并不适合于运行上电过程。

6.6.4 设定电压而非电流

在电路中使用理想源的时候我们会有个感觉，使用理想电压源总是很安全，而使用理想电流源时不时会带来一些问题。如果仿真器发现一个电压源短接到另一个电压源，也就是会造成无穷大电流的情况，会在一开始就告诉使用者这个错误。而一个开路的电流源，也就是会浮空节点导致无穷大的电压的情况，则不会。很有可能在仿真中间导致收敛问题。

仿真器的这种特性，鼓励Verilog-A/AMS模型尽可能用电压限定来替代电流限定的书写方法。

对于电压输出型的端口，尽量使用输出电压和阻抗限定，例如：


V(out) <+ xout;




或者


V(out) <+ iout*rout;






对于电流输出型的断口，最好限定电流随着输出电压达到一定幅度而减小，例如：


I(out) <+ iconst*(1-1/100^V（out）);




两端限制


I(out) <+ iconst*(1-1/100^V（out）-1/100^（5-V（out））);




可以利用指数或者高阶函数，产生对在中间电压不产生影响而在接近门限电压处快速减小的函数。虽然看上去式子有点复杂，但并不会影响模型计算速度。

需要注意的是保证所产生的函数连续，比较简单的方法是在matlab等数学工具中验证一下函数。

对于信号描述（比如频率相应，延时等），一般都可以使用电压描述。如果是确实是描述电流的，可以在模块内部使用电压计算，在输出做一个VI转换，例如：


electrical xout;
analog begin
V(xout, gnd) <+ laplace_nd(V(in, gnd), {1,0,0}, {1,1,0});
     I(out) <+ V(xout, gnd)*(1-1/100^V（out）-1/100（5-V（out））)/Rideal;
end




这个例子，也加入了输出电流随着输出电压的限制。

对于微分，积分运算，同样尽量使用电压描述。对电流的微分计算会造成特别糟糕的结果。很多时候仿真电路都会遇到这种情况，输出的电压曲线比较连续，而输出的电流曲线成锯齿状。这是因为仿真器为了保证电压连续性而牺牲了电流的连续性，特别是在IC设计中常见电容器件而少见电感器件，因为电压不能跳变而电流跳变。这时候如果遇到电流的微分计算，不连续点会大大增加。

6.6.5 输入输出阻抗

遇到不收敛的问题，有时候修改一下输入输出阻抗也会起到效果。

对于电流限定的节点来说，如果使用了类似上面这种限制电压范围的函数，也就限制了输出阻抗。

对于电压限定的节点来说一般没有什么问题。但在一些高频应用里，需要考虑的可能不止DC阻抗，有时是需要增加一些LC负载，也有可能需要通过调整Q值来使得系统工作得如你所计划。

6.6.6 内部限制

对于很多开环电路，我们会这样写：


V(out) <+ gain*(V(sigin, sigref));




在诸如比较器或者运放这样的模型中，gain会是一个很大的值，比如10e6。

对于电路工作特性来讲，这样写没有错。正常工作的负反馈也会将输出拉到合理的电位上。

问题是，在仿真器收敛过程中，也许它会选择一个不那么合理的值，比如输入偏离了1mV。结果是，你得到了一个1000V的输出。这可能产生很多结果：无法收敛，因为发现巨大的跳变；其他模块同样没有限定输入输出，它处理了这个1000V，系统停留在一个实际上不可能的状态下；也有可能，其他模块认为器件击穿等，停留在一个大电流的器件损毁状态下。

前面的例子使用过这样一个函数：


V(sigout) <+ 0.5 * (out_high - out_low)* tanh(gain*(V(sigin, sigref)- sigin_offset))




tanh是用于限定的一个非常有用的函数。原因是，它中间点线性度很好，任意点连续可导，两端的衰减趋势和实际电路的情况往往非常吻合。这也是为什么Cadence环境下提供的现场模块大多数使用这个函数。

6.6.7 反馈

反馈是常见的系统结构，也是产生振荡、仿真速度慢的主要原因。在模型中，我们一般建议把带有反馈的部分放在电路结构中，而不要在模型内部描述反馈。

举个最常见的运放作为缓冲的例子。电路结构类似下图：

[image: 图像说明文字]


图 7-17 带反馈单元

最直观的方法当然是，将图中的运放设计为一个单元：


module amp(ip, in, out)
......
end module




然后如图连接输入和输出。这样，模型只需要描述运放单元的开环特性，其闭环特性通过仿真器的计算可以得到准确的展示。

假如你希望自己描述反馈，也就是将上面这个结构描述为：

[image: 图像说明文字]


图 7-18 描述为不带反馈结构

你需要这样描述：


module amp(in, out)
......
analog begin
      V(out) <+ V(in);
end
end module




这样写，当然是太简单了。如果你想把非理想特性在这里考虑进去，就不是那么容易了，因为你得自己计算反馈的效果。

比如，小信号响应，也许需要这样一个计算过程：

[image: 图像说明文字]


这看起来有点复杂了。如果希望考虑到负载的变化，或者加入建立时间的大信号描述的语句，那就更加复杂一些了。如果你对反馈的计算有问题，不稳定或者奇怪的输出是常见的结果。而最重要的是，你的计算设定了系统的稳定性等运行状态，你无法在仿真器中验证这一切了。

在实际中，当面对明显的反馈系统时，这种将反馈写入自己的代码的尝试几乎不会出现。这种情况往往出现在一些小的、内部的反馈电路上，比如current-boost电路，一个内部的regulator等。基本的准则是，如果不需要，那就简单化、理想化这些模块；如果需要，描述开环特性，让仿真器去处理反馈后的结果。

6.7 高级仿真中的Verilog-A/AMS

Cadence的spectre有spectreRF的特性。Hspice也有HspiceRF。这些高级仿真主要是针对：周期小信号、高频应用，它关心的可能是：周期小信号稳定性、周期小信号噪声、相位噪声、S参数等。

这些高级仿真与一般性仿真的核心差别是，周期性信号的平均化。

图 7-19 RF仿真器的计算流程

如图所示，周期性小信号的计算大致就是每次计算一个周期内的平均值，然后估计电路周期性信号是否收敛。不收敛则再计算一个周期。如果收敛，基于这个状态下得到的值，再进行其他仿真计算。

由于这种周期时间平均化的计算方式，我们可以想到，这些功能是无法工作的：


	Z变换函数。离散周期性信号显然无法正常计算平均值。

	absdelay()。由于计算的是稳定后的周期平均值，绝对的延时没有效果。

	absdelta()。需要混合信号仿真，不支持。



这些功能可以使用，但会受到影响：


	$abstime。可以获得仿真绝对时间，但由于绝对时间会随着收敛周期数增加而增加，不建议用此数值修改仿真内部变量。

	analysis。目前此函数不支持RF仿真类型，也就是说输出总是0。

	random。如果生成随着时间变化的随机数，很显然周期性小信号无法收敛。



除此之外，在功能的书写上也有一些讲究。周期性计算的核心是计算每个周期开始点和结束点的电路状态，对这些状态计算平均值。这种“状态”对于Verilog-A/AMS代码来讲，不仅包括物理端口的输入输出值，也包括内部存储的变量。如果内部的变量，在每个周期状态，都有确定值（周期信号为高、为低、为中间态），那么系统可以得到它的平均值。否则，系统内部存在一个未知的状态，则不能使用RF仿真器。

Cadence文档内有这样一个例子来解释，什么样的是未知的状态。


module (clk, in, out)
input clk;
electrical clk;
input in;
electircal in;
output out;
electrical out;

integer flag;
parameter real vth=1;

analog begin
     @(initial_step) begin
     flag = 0;
     end

     @(cross(V(clk)-vth)) 
     flag = !flag;

      V(out) <+ idt(0, V(in), flag);
end
end module






这段代码，在不同的时钟周期，设置不同的flag值。flag值被用于选择进行积分，还是reset积分函数。

这段代码不能在RF仿真器内仿真。原文的解释是：它没有确定在时钟跳变的时候，模块的行为。我更加直白的解释是，如果仿真一个周期性，计算变量flag的平均值（假定时钟占空比为50%），应该是0.5。但是输出电压值是什么呢？idt函数是无法接收这个平均后的0.5的。也因此，这个模块无法计算平均状态的。

对于Verilog-A/AMS代码来说，稳定的状态意味着，内部状态平均化后生成的依然是可以被仿真器计算的一系列算式。

一个替代的计算方法：


module (clk, in, out)
input clk;
electrical clk;
input in;
electircal in;
output out;
electrical out;

integer flag;
parameter real vth=1;

analog begin
     @(initial_step) begin
     flag = 0;
     end

     @(cross(V(clk)-vth)) 
     flag = !flag;

      V(out) <+ idt(flag*(V(in)-V(out)));
end
end module





仿真器就会很容易得到输出的平均状态为：


V(out) <+ idt(0.5*(V(in)-V(out)));




这样的书写方式，使用迭代计算的公式，避免了离散的输入参数，从而保证了仿真器平均化的结果都能够转化为稳定的状态。

因此在使用RF仿真器的时候，特别要注意的是内部变量状态和避免出现离散的参数。





附录A Verilog-A/AMS 关键词


above
abs
absdelay
acos
acosh
ac_stim
aliasparam
always
analog
analysis
and
asin
asinh
assign
atan
atan2
atanh
begin
bound_step
branch
buf
bufif0
bufif1
case
casex
casez
ceil
cmos
connectrules
cos
cosh
cross
ddt
ddx
deassign
default
defparam
delay
disable
discipline
discontinuity
driver_update
edge
else
end
endcase
endconnectrules
enddiscipline
endfunction
endmodule
endnature
endparamset
endprimitive
endspecify
endtable
endtask
event
exclude
exp
final_step
flicker_noise
floor
flow
for
force
forever
fork
from
function
generate
genvar
ground
highz0
highz1
hypot
idt
idtmod
if
ifnone
inf
initial
initial_step
inout
input
integer
join
laplace_nd
laplace_np
laplace_zd
laplace_zp
large
last_crossing
limexp
ln
localparam
log
macromodule
max
medium
min
module
nand
nature
negedge
net_resolution
nmos
noise_table
nor
not
notif0
notif1
or
output
parameter
pmos
posedge
potential
pow
primitive
pull0
pull1
pullup
pulldown
pwr
rcmos
real
realtime
reg
release
repeat
rnmos
rpmos
rtran
rtranif0
rtranif1
scalared
sin
sinh
slew
small
specify
specparam
sqrt
strobe
strong0
strong1
supply0
supply1
table
table_model
tan
tanh

task
temperature
time
timer
tran
tranif0
tranif1
transition
tri
tri0
tri1
triand
trior
trireg
vectored
vt
wait
wand
weak0
weak1
while
white_noise
wire
wor
wreal
xnor
xor
zi_nd
zi_np
zi_zd
zi_zp







附录B Symica

模拟电子学习过程中的一个重要困难是，绝大多数仿真开发平台都是商业平台，需要价格不菲的许可证费用。作为一个学生或者业余学习者，没有办法支付高额的费用。学习的东西往往没有很好的软件平台可以尝试。

现在，这种情况有所好转。有很多商业的仿真平台开始提供免费版本。很多开源的仿真器平台也能够完成复杂系统的建模与仿真。

我会演示symica这个商业IC设计软件的免费版本和qucs这个开源的仿真设计平台的使用。他们都在一定程度上支持Verilog-A模型。


Symica


Symica是一个运行于windows和Linux平台的商业模拟与混合信号设计的软件。它包括：SymicaDE, SymicaLayout和SymicaProbe。分别用于电路设计，版图设计和波形仿真。

总体来讲Symica的界面和操作都非常类似Cadence。它也支持Cadence软件的Skill脚本语言。因此对于熟悉Cadence的用户来说上手十分简单。

Symica是一个商业软件，但是它的免费版本包含了商业版本的全部功能，而且除去一些限制之外（比如节点数目），全部功能都可以无限期使用，这对学习者来说非常可贵。

图表 0-1

打开Symica，界面十分简洁。左侧是器件库，右侧是文件内容的显示窗口。

图表 0-2

选择文件-> 创建电路图。可以从左侧器件库里拖曳器件到电路图中。

图表 0-3

也可以点击insert，从窗口中选择器件。 [image: 图像说明文字]


图表 0-4

点击仿真按钮，弹出仿真界面。可以点击仿真类型进行选择和修改。这里将仿真设置为

transient，时长100uS。

[image: 图像说明文字]


图表 0-5

仿真结束后弹出波形界面，点击insert键选择已经保存的节点波形。

Symica对Verilog-A的支持相当好。生成Verilog-A模块的步骤也很简单。

图表 0-6

首先选择创建文件，在对话框中选择Verilog-A类型。

[image: 图像说明文字]


图表 0-7

然后就可以在文本编辑框中写入代码。代码完成后，可以选择Create Cell From，在弹出窗口中选择从Verilog-A代码创建symbol。

[image: 图像说明文字]


图表 0-8

如果代码没有错误，Symica会自动生成对应的端口和默认的图标。你会在左侧库列表中直接看见新产生的模块图标。直接将此图标拖进电路图即可进行仿真。

需要注意的是，如果生成模块图标的时候发现少了几个端口，或者名称错误，这多半意味着你的代码存在问题，或者你使用了Symica目前还不支持的语句。即使你手动添加端口，也会在仿真阶段失败。因此，必须要保证所有的端口被Symica自动产生。

[image: 图像说明文字]


图表 0-9

在仿真中，就可以看到Verilog-A模块产生的波形。

Symica对Verilog-A语法的支持比较好。在作者的试验中，绝大多数前面提到的语句都可以正常运行。





附录 C QUCS

qucs是一个开源的电路设计仿真平台，可以在http://qucs.sourceforge.net/ 找到它的下载安装文件。它支持Linux和windows以及其他操作系统。

[image: 图像说明文字]


图表 0-10

下载后点击程序即可开始安装。

[image: 图像说明文字]


图表 0-11

qucs的安装附带数个附加程序包，建议为qucs的安装选择一个单独的目录。

图表 0-12

介绍一下安装时附带的几个安装包。

Mingw是windows下的编译和运行库环境。基本上属于必装的一项。

FreeHDL是用于解释和仿真VHDL代码的程序。

iVerilog是用于解释和仿真Verilog代码的程序。

Octave是一个数值计算的库。可以理解为一个简化版的Matlab程序。

[image: 图像说明文字]


图表 0-13

接下来就按步骤完成安装。

[image: 图像说明文字]


图表 0-14

运行qucs程序，默认出现的是电路编辑界面。在左侧可以打开项目、内容、器件和库页面。

项目页面可以管理当前的项目。

内容页面会显示当前项目下的所有文件。

器件页面会显示当前可使用的器件。

库页面可以管理qucs加载的库文件。

[image: 图像说明文字]


图表 0-15

使用的时候首先，新建项目。

[image: 图像说明文字]


图表 0-16

进入器件页面，可以向电路图中添加已有的器件。

图表 0-17

在加入的器件上点击邮件，可以编辑器件属性。比如，电容器件的属性包括器件的值，初始状态等。

图表 0-18

绘制如图所示的这样一个电路，并保存。

[image: 图像说明文字]


图表 0-19

在器件页面，选择“simulation”，从里面可以添加诸如transient仿真，DC仿真等。这种将仿真类型直接添加在电路图上的风格，十分类似pSpice。

图表 0-20

设置如图所示的两个仿真类型。需要注意的是，如果需要在仿真结果中显示相应节点，必须命名此节点。

图表 0-21

点击F2或者在菜单里选择simulation即可运行仿真。

[image: 图像说明文字]


图表 0-22

仿真成功的话会自动跳转到结果显示界面。

从器件页面添加diagram（图表）。可以为图表添加仿真中保存的节点。

图表 0-23

结果显示出来了。

图表 0-24

右键点击图表，可以修改其显示属性。比如这里我们将XY轴都变成指数形式，可以看到曲线变成我们常见的20dB下降形式。

接下来，我们就开始尝试在qucs中创建Verilog-A模型，并进行仿真。

很多仿真器编译和仿真Verilog-A模块，都使用一个叫做ADMS的程序。有兴趣的读者可以在这里http://mot-adms.sourceforge.net/找到这个程序相关的信息。

ADMS的作用是将Verilog-A模型的代码转换为仿真器可以读取的C/C++代码。它需要正确配置运行环境，并且完全是一个命令行工具，用户手动配置有一些难度。因此，在类似qucs这样的软件里，集成了ADMS的功能，并且支持通过简单的图形界面操作即可以生成可使用的Verilog-A模块。图形化生成有时候会存在一些bug，因此有兴趣研究的读者可以阅读一下ADMS的手册，研究一下手动使用ADMS生成Verilog-A模块的形式。

另外要注意的是，ADMS支持所有的Verilog-A的语法，能够将这些语句转化为C/C++代码。但相应的代码是否能够在仿真器内正常运行取决于仿真器的实现。也就是说，虽然ADMS支持绝大多数语句，但不同的仿真器会有不同的尚未被支持的语句。具体了解仿真器的限制，最佳方式就是阅读仿真器自带的Verilog-A模型的代码。这些预编译的模型代码包括了绝大多数此仿真器支持的语句，同时也可以帮助你学习在有一定限制的条件下如何实现不同的功能。

图表 0-25

在创建Veirlog-A模型之前，我们需要设置ADMS程序的路径。不知为何，默认安装后的设置并没有指向qucs自带的ADMS程序。我们可以如图中修改AdmsXml路径。

图表 0-26

首先，打开文本编辑，在编辑窗口写入Verilog-A代码并保存。

图表 0-27

选择Project->Build Verilog-A module。

图表 0-28

正常的话，下方会出现编译窗口。如果所有图标都是绿色，表示代码编译正常。

图表 0-29

选择File->Edit Text Synbol。这一步是为了生成可以加入到电路图中的模块符号。

[image: 图像说明文字]


图表 0-30

出现如图窗口，修改自动生成的图标，保存。

图表 0-31

选择Project->Load Verilog-A module。这一步将生成的图标加入当前的项目。

[image: 图像说明文字]


图表 0-32

为了在左侧库列表中正常显示，你需要为模块的图标选择一个图片。这个图片文件必须存放于当前项目目录下。

图表 0-33

完成后你就可以看到在左侧的库中多了一项：Verilog-A User Device。你可以将你生成的模块图标加入到电路图中。

图表 0-34

完成整个测试电路。

图表 0-35

和前面一样运行仿真，可以看到通过Verilog-A模块被放大的信号。





看完了

如果您对本书内容有疑问，可发邮件至contact@turingbook.com，会有编辑或作译者协助答疑。也可访问图灵社区，参与本书讨论。

如果是有关电子书的建议或问题，请联系专用客服邮箱：ebook@turingbook.com。

在这里可以找到我们：


	微博 @图灵教育 : 好书、活动每日播报

	微博 @图灵社区 : 电子书和好文章的消息

	微博 @图灵新知 : 图灵教育的科普小组

OEBPS/Image00101.jpg





OEBPS/Image00100.jpg





OEBPS/Image00103.jpg
v )

e )

&





OEBPS/Image00102.jpg





OEBPS/Image00105.jpg
B S
B b ® D

4
il

i

T






OEBPS/Image00104.jpg
L





OEBPS/Image00106.jpg
Y
S TSR






OEBPS/Image00021.jpg
De EM vew Bob 1

| et pa | cokn

e
|e ey e e Wty s cassns |
= e —
ey ]
Ver vetogs ]
@ns P [

P






OEBPS/Image00022.jpg
®

‘Schematic model generation completed:
Generate Verilog AViS model(s) produced the following warnings:
WARNING (SMG-61074): Pin ‘gnd’ has only one conneclion.
WARNING (SMG-61074): Pin vec: has only one conneciion.

‘Successfuly generated the following model(s):
Verilog-AVS: YSMG] veriogams

‘Successfully generated the following symbol view:
‘Generated new symbol view: === & H/SMG/symbol
Generated models can be viewed from the Model Targel tab

TE: To highiight the instances, nets or pins on the Model Schematic,
Click on the fine in the warning message text above.






OEBPS/Image00019.jpg





OEBPS/Image00020.jpg





OEBPS/Image00017.jpg
cadence

S s e e

Wooe Schemat ||| Mode Torget | | Buldng Blocks | | Cabrate |
_ Create New Intial Model Schematic [NEW USERS START HERE]

¥ Location of Existing Model Schematic

ey ([
ca au B
[






OEBPS/Image00097.jpg
T e e g





OEBPS/Image00018.jpg
Model Pins

inp inn voe g ot Qe

Model Schematic Pin nformation






OEBPS/Image00099.jpg





OEBPS/Image00098.jpg





OEBPS/Image00025.jpg
[roare





OEBPS/Image00026.jpg
[roare





OEBPS/Image00023.jpg
[ roa





OEBPS/Image00024.jpg
[ o vic





OEBPS/Image00002.jpg
Cellview res symbol does not exist
Do you want to create it?






OEBPS/Image00001.jpg





OEBPS/Image00004.jpg
cadence






OEBPS/Image00003.jpg
Library Name Cell Name View Name
res synbol
Pin Specifications

LetPins

Right Pins

TopPins  vp wn

Bottom Pins
Exclude Inherited Connection Pins:

© None U All L Only the

Load/save L EditAtributes . Edit Labels L Edit Propetties

@D concel ponly Help






OEBPS/Image00006.jpg
L
epomm





OEBPS/Image00005.jpg





OEBPS/Image00112.jpg
CapRrar/zmieony

o






OEBPS/Image00111.jpg





OEBPS/Image00114.jpg





OEBPS/Image00113.jpg





OEBPS/Image00116.jpg





OEBPS/Image00115.jpg





OEBPS/Image00010.jpg





OEBPS/Image00011.jpg
Modelwriter 2.31






OEBPS/Image00008.jpg





OEBPS/Image00096.jpg
B T S P
e 9 o ©® 88






OEBPS/Image00009.jpg





OEBPS/Image00131.jpg
VERILOG-A/AMS
RRERITS5HE

SYSTEM LEVEL DESIGN
AND SIMULATION






OEBPS/Image00094.jpg
a5+

IR )

e
a+ 20+





OEBPS/Image00108.jpg
€ Inn

°
!
.
|
J

e





OEBPS/Image00007.jpg





OEBPS/Image00095.jpg





OEBPS/Image00107.jpg
Y
S TSR






OEBPS/Image00092.jpg





OEBPS/Image00110.jpg





OEBPS/Image00093.jpg





OEBPS/Image00109.jpg





OEBPS/Image00016.jpg





OEBPS/Image00014.jpg
- Cadence Modelwriter 2.31

pos omveng 5w =] @
g Cotvetoan 100 w0 | @
[——— Y
g1 vooe B v =@
O o I [ )
sovrmmcnown v~
Por Sewnan 00— ws | @
g Swwsen [0 ws =] @






OEBPS/Image00015.jpg
e

vin isiag +

A
oy

it

(A

e |

pocoetn setngr oeniog - 0
ot v, m. otton, balys, s

s ) oin

omatin by 1) begn

S

ot nag ¢ g i st < o

Modelwriter 2.31

Eybcticrty

e - viin ) < ot

SRR R g g s o ot bt v 3 By
L S AR M a0 2 /R

R g g e, s s Lo g0 o By
LA






OEBPS/Image00000.jpg
Verilog-AMS

Verilog-HDL Verilog-A





OEBPS/Image00012.jpg
- Cadence Modelwriter 2.31 - Ox

ncto: OrsrVotag Angoe
v Carcs oo Sy,
iy
Vs 4o 27
Lt gt A 14 204720 57208
r——
pipeeme

ot 80 v, rovs v
et ity st 8 skl






OEBPS/Image00013.jpg
Cadence Modelwrit

2.31

[
ey
Lt gt T g 14 20470 5T 200
e —
sty g varsr .
[i ey ——

1 0 g, v o s o
[N —






OEBPS/Image00090.jpg
WoTEithieading Enabled. 4 theads on systen vith 17 svailable procy

: R R BT
i L B

TNARNING (BECIRE-1670T) | Gnly tean supports pafd farmat, cesult o

o2 dasireis, ale.

ectaat prrmatac nlues
et it
e T

e
[
S T

e scciulafid G 0"k, ahilea ey G

econoto_veseing epeacing posn sntoonason o covize

e vy ik 1%-5 oo to il
esEiranionials. Vit rovkioe parsmecees o rovie
SR G R

sgcecnce susse (2.96.3 o ved sex 6. 2010)
02 0es "Gy BT o, elapand + 201 5, wtal = 1231

[E52 Sine ricingsny fhapece T 888 2 pekcuntage of sota1 « 7134
i e o

SimiSEien stareed 3¢ 2 0116 . ved ace 6, 2014 ondd st 2.06.36
oot Seiotes et B occotor | vemann. o bortone






OEBPS/Image00091.jpg
[Wotice Fiom spectie during amatasl setup.
s Enbina
dtithresding Enabled. 4 theeads on systen vith 12 svailsble proc)

rine for pacsing. co0 - 7.999 ns, elapasd = 1.75222 5
[Tine sciukiiated: Go0 © 220964 ke, siapsed - 15,3152 s
[ Zecient Memory weed = 552 Mgtes

vocning fcon spectce
A (SECIRE-16T07) - only tran suppores pfel format, resul of

o6 ansiyis - deop
artant prcsmetee values
o Thtecte
Satolh '+ T v
B3 T i

S 30 0
k3
thedteess - an
gunde - 15

Loy - 2000 s, elapsed =
oot =t o9 2
yies

e fesiiont nenory uoed = 557

PR ol E )
R R PRI T
E R T
L i R e
i T I

T e

[pgocegate susie (2.54:14 2. ved oct 5, 2014)
et Bl et M T e

[Fise Shene R 1icenasng: S135000 - 115 3 percentage of total « 771
e 2oy oed T S Soyen g

Simuiacion stareed at. 263,60 P, Ved 0ct 8, 2014, ended st 2:54:14
[Fpectre’conpioces vith D eecors, I vernang, snd 3 Rotsces






OEBPS/Image00088.jpg
o N
| {
foret
i s = =






OEBPS/Image00089.jpg





OEBPS/Image00087.jpg





OEBPS/Image00085.jpg
w00

g il

il

i *5\1\ {1

55

B






OEBPS/Image00086.jpg
e e —

mn]

wn | o mex | mes | wn s s isis
i






OEBPS/Image00083.jpg
i 2050754 2656870

%
e T i
P

s 200m 5 0565






OEBPS/Image00084.jpg
o 2






OEBPS/Image00081.jpg





OEBPS/Image00082.jpg





OEBPS/Image00079.jpg





OEBPS/Image00080.jpg





OEBPS/Image00077.jpg





OEBPS/Image00078.jpg





OEBPS/Image00076.jpg





OEBPS/Image00074.jpg
3948
574888605 +3 948"






OEBPS/Image00075.jpg





OEBPS/Image00072.jpg
Fez=





OEBPS/Image00073.jpg
2s15r [0k





OEBPS/Image00070.jpg
cadence

o nms )1 1 e arey

e e R 5 o
€3 GRKT ety sy v, v on o
s W et ey s e i o v o)

TP (s g i 1 ngt e lovety 0 ikt (st e






OEBPS/Image00071.jpg





OEBPS/Image00068.jpg
Erol e
23 Vet St St From ot of 8

s  Conra e s oo,
e

o
£
b

cadence

[ s e

L ——






OEBPS/Image00069.jpg
e et o ver cadence

e

o

% 1=

F e

1 s

M s
]

=
e

e vttt CSTANS Sttt O AN St S et
oot o conppns s

[T ————— P





OEBPS/Image00067.jpg
cadence

© e o el Caoex <o © oo






OEBPS/Image00065.jpg





OEBPS/Image00066.jpg





OEBPS/Image00063.jpg





OEBPS/Image00064.jpg





OEBPS/Image00061.jpg
il





OEBPS/Image00062.jpg
[





OEBPS/Image00059.jpg





OEBPS/Image00060.jpg





OEBPS/Image00057.jpg





OEBPS/Image00058.jpg





OEBPS/Image00054.jpg
nt

at






OEBPS/Image00055.jpg





OEBPS/Image00052.jpg
a





OEBPS/Image00053.jpg
m

a1






OEBPS/Image00050.jpg
ol

™

nputs

1

@





OEBPS/Image00051.jpg
e :z electrical
e g =

" :: @a
=

bar





OEBPS/Image00048.jpg
v ez





OEBPS/Image00049.jpg
A8z





OEBPS/Image00056.jpg
| [ -]






OEBPS/Image00123.jpg





OEBPS/Image00047.jpg
A8z





OEBPS/Image00122.jpg





OEBPS/Image00125.jpg





OEBPS/Image00124.jpg





OEBPS/Image00126.jpg





OEBPS/Image00043.jpg





OEBPS/Image00044.jpg





OEBPS/Image00041.jpg





OEBPS/Image00117.jpg





OEBPS/Image00042.jpg





OEBPS/Image00039.jpg
Fe)=

lins 4ns’

X+





OEBPS/Image00119.jpg





OEBPS/Image00040.jpg





OEBPS/Image00118.jpg
anasassamic0zy






OEBPS/Image00037.jpg





OEBPS/Image00121.jpg





OEBPS/Image00038.jpg





OEBPS/Image00120.jpg





OEBPS/Image00045.jpg
Triangular Wave

Triangular Wave First Derivative






OEBPS/Image00046.jpg
v ez





OEBPS/Image00032.jpg
Neworigin -

Interruption

output_expression(t)

rise_time

 fall_time





OEBPS/Image00033.jpg
P g
I





OEBPS/Image00030.jpg
[ roavie





OEBPS/Image00128.jpg





OEBPS/Image00031.jpg
expr(t)

output_expr(t)

delay fall_time

rise_time





OEBPS/Image00127.jpg





OEBPS/Image00028.jpg
[ o vic





OEBPS/Image00130.jpg





OEBPS/Image00029.jpg
[ roavie





OEBPS/Image00129.jpg





OEBPS/Image00133.jpg
VERILOG-A/AMS
RIERITSHR






OEBPS/Image00027.jpg
[ roa





OEBPS/Image00036.jpg





OEBPS/Image00034.jpg
R =

Q- )1 - bs)(1 - cx)
50— )1 - w)1-z)





OEBPS/Image00035.jpg
F(s)=

14545





